MATHÉMATIQUES sur MICRO-ORDINATEUR

2. ALGÈBRE

Alain REVERCHON
Marc DUCAMP



EYROLLES

MATHÉMATIQUES
SUR MICRO-ORDINATEUR
2. ALGÈBRE

CHEZ LE MEME EDITEUR

Des mêmes auteurs

Tome 1 - Analyse - 1984.

Autres ouvrages

PICARD et DUCAMP	- *Jeux d'action, de hasard et de réflexion sur MSX* - 1985.
STUMM	- *Jeux de robotique en BASIC* - 1985.
PITRAT	- *Textes, Ordinateurs et compréhension* - 1985.
JAMES	- *Introduction à l'intelligence artificielle sur Micro-ordinateur* - 1985.
MONTEIL et SCHOMBERG	- *Programmes d'intelligence artificielle en Basic* - 1985.
ANGELL et JONES	- *ZX Spectrum et Spectrum +* : Techniques graphiques avancées - 1985.
O'MALLEY	- *25 programmes graphiques en basic microsoft* - 1985.
DELAHAYE	- *Dessins géométriques et artistiques avec votre micro-ordinateur* - 1985.
VEY	- *Apprentissage et utilisation du Bus I.EEE 488/CEI 625* - 1985.
JAULENT et BATICLE	- *Circuits périphériques de la famille 68 000* - 1985.
DARDANNE avec la collaboration de BOULESTEIX	- *Microprocesseur 6809.* Ses périphériques et le processeur graphique 9365-66 - 1984.

MATHÉMATIQUES SUR MICRO-ORDINATEUR

2. ALGÈBRE

par

Alain REVERCHON
Marc DUCAMP

EYROLLES
61, boulevard Saint-Germain — 75005 Paris
1985

Avant-propos

Faisant suite au premier tome consacré à l'Analyse, cette deuxième partie des "Mathématiques sur micro-ordinateur" s'intéresse à l'Algèbre.

Cet ouvrage s'adresse à tous ceux qui souhaitent utiliser leur ordinateur dans un but scientifique, et permet l'initiation aux méthodes du calcul numérique ainsi que l'illustration concrète des théories mathématiques.

Nous avons conservé la démarche du premier tome : les bases théoriques indispensables sont rappelées sans être développées, et de nombreux exemples commentés illustrent l'utilisation des programmes proposés, dans le but de favoriser une compréhension plus intuitive.

Les programmes d'un même chapitre sont destinés à être utilisés simultanément, ce qui permet par exemple de chaîner des opérations sur les matrices ou les fractions rationnelles, mais ils peuvent également être introduits séparément, par exemple comme sous-programmes d'une réalisation plus importante.

Les chapitres sont indépendants les uns des autres, mais sont classés par difficulté mathématique croissante.
Les programmes sont de manière générale plus complexes que ceux du premier tome. En Algèbre, les opérations à effectuer sont en effet plus compliquées que

celles de l'Analyse, et font souvent appel à des organigrammes relativement complexes. Le stockage des programmes et des variables nécessite en particulier une mémoire importante; il faut disposer d'un minimum de 16 Koctets pour permettre le fonctionnement simultané des programmes d'un même chapitre. Naturellement, chaque module pris séparément se contente d'une mémoire plus faible.

Précisons enfin que les sujets abordés dans les deux tomes sont en rapport étroit avec les programmes de mathématiques des classes préparatoires aux grandes écoles. L'informatique devant être prochainement introduite dans ces classes, l'étude et l'utilisation des programmes proposés permettront aux étudiants de se familiariser aux méthodes numériques et à la programmation en Basic, tout en vérifiant avec rapidité et sûreté les résultats des calculs rencontrés dans les problèmes.

Nous espérons que cet ouvrage vous apportera entière satisfaction, et vous souhaitons de longues heures passionnantes aux commandes de votre ordinateur.

Table des matières

1 Arithmétique

1. Introduction

L'arithmétique est un domaine des mathématiques qui suscite en général beaucoup de curiosité : en effet, la formulation des problèmes est le plus souvent très claire et les principaux résultats théoriques se comprennent aisément.

Nous étudierons tout d'abord les systèmes de numération : nous effectuerons des conversions d'entiers d'une base dans une autre, puis des calculs élémentaires dans une base quelconque.

Nous nous intéresserons ensuite à la recherche des nombres premiers et à la décomposition des entiers en facteurs premiers.

L'algorithme d'Euclide nous permettra de calculer le PGCD et le PPCM de plusieurs nombres.

Nous résoudrons ensuite dans Z les équations diophantiennes, c'est-à-dire de la forme $ax + by = c$. Enfin, nous aborderons les problèmes de dénombrement avec un programme d'analyse combinatoire.

Pour utiliser les six programmes simultanément, il faut remplacer chaque instruction END par l'instruction GOTO 1ØØ et ajouter le menu :

```
100 CLS
110  PRINT  TAB( 5);"ARITHMETIQUE": PRINT
120  PRINT : PRINT "1- CHANGEMENT DE BASE"
130  PRINT : PRINT "2- OPERATIONS ELEMENTAIRES DANS UNE"

140  PRINT "   BASE QUELCONQUE"
150  PRINT : PRINT "3- NOMBRES PREMIERS"
160  PRINT : PRINT "4- DECOMPOSITION EN FACTEURS PREMIER
    S"
170  PRINT : PRINT "5- PGCD ET PPCM DE N NOMBRES"
180  PRINT : PRINT "6- RESOLUTION DANS Z DE A*X+B*Y=C"
190  PRINT : PRINT "7- ANALYSE COMBINATOIRE"
200  PRINT : PRINT "8- FIN"
210  PRINT : INPUT "VOTRE CHOIX ";E
220  ON E GOTO 1000,2000,3000,4000,5000,6000,7000,8000
230  GOTO 210
8000  END
```

L'ensemble nécessite environ 5 Koctets de mémoire.

2. Changement de base

a. Les systèmes de numération

Notre système de numération usuel est le système décimal, c'est-à-dire que nous représentons les nombres à l'aide de dix chiffres. L'origine de ce choix provient probablement du fait que nous possédons dix doigts.

Nous utilisons également les bases douze et soixante lorsque nous mesurons des durées : un cadran d'horloge possède douze graduations symbolisant les heures, et il faut soixante secondes pour faire une minute.

Les ordinateurs, ne possédant que des systèmes logiques à deux états, Ø ou 1, codent leurs données et effectuent leurs calculs dans le système binaire.

b. Programme

Le programme fournit la représentation en base b d'un nombre exprimé en base a ; il suffit pour cela de lui introduire les valeurs a et b, puis les nombres à convertir.

Il est possible d'utiliser toutes les bases comprises entre 2 et 36, les nombres 1Ø, 11, ..., 36 étant symbolisés par les lettres A, B, ..., Z.

Toute introduction incorrecte sera détectée : par exemple, tout nombre exprimé en base 3 et contenant un 4 sera refusé par le programme.

La conversion s'effectue en deux étapes ; le nombre de départ est d'abord converti en base 1Ø, puis écrit en base b.

Soit un nombre n de N chiffres s'écrivant $\alpha_{N-1}\ \alpha_{N-2}\ ...\ \alpha_2\ \alpha_1\ \alpha_0$ en base a ; sa valeur en base dix est alors :

$$n = \alpha_{N-1} . a^{N-1} + \alpha_{N-2} . a^{N-2} + ... + \alpha_2 . a^2 + \alpha_1 a + \alpha_0$$

Divisons ce nombre n par b :

$$n = b\, q_0 + \beta_0$$

Le chiffre β_0 obtenu est celui des unités du nombre n écrit en base b.

Poursuivons le calcul en divisant cette fois q_0 par b :

$$q_0 = bq_1 + \beta_1$$

β^1 est cette fois le chiffre des dizaines de n en base b.

De proche en proche, tous les chiffres sont ainsi déterminés.

Ce sont les sous-programmes 28ØØ et 291Ø qui effectuent les conversions base a → base 1Ø et base 1Ø → base b ; ils serviront également au programme suivant : "opérations élémentaires dans une base quelconque".

```
1000 CLS
1010  PRINT  TAB( 5);"CHANGEMENT DE BASE": PRINT : PRINT

1020  INPUT "BASE DE DEPART:";BD
1030  INPUT "BASE D'ARRIVEE:";BA
1040 BD =  INT ( ABS (BD))
1050 BA =  INT ( ABS (BA))
1060  IF BD < 2 OR BD > 36 OR BA < 2 OR BA > 36 THEN 10
    20
1070 ER = 0
1080  PRINT
1090  INPUT "VOTRE NOMBRE:";NB$
1100  IF NB$ = "0" THEN 1160
```

```
1110  GOSUB 2800
1120  IF ER = 1 THEN 1070
1130  GOSUB 2910
1140  PRINT : PRINT "EN BASE ";BA;": ";RE$
1150  GOTO 1070
1160  PRINT : PRINT
1170  INPUT "VOULEZ-VOUS REUTILISER LE PROGRAMME?:";Z$
1180  IF Z$ = "O" THEN 1000
1190  END
2800 LG =  LEN (NB$)
2810 NB = 0
2820  FOR I = 0 TO LG - 1
2830 NI =  ASC ( MID$ (NB$,LG - I,1))
2840  IF NI < 48 OR NI > 90 OR (NI > 57 AND NI < 65) THEN
    ER = 1:I = LG - 1: GOTO 2890
2850  IF NI < 58 THEN NI = NI - 48
2860  IF NI > 64 THEN NI = NI - 55
2870  IF NI >  = BD THEN ER = 1:I = LG - 1: GOTO 2890
2880 NB = NB + NI * BD ^ I
2890  NEXT I
2900  RETURN
2910 RE$ = ""
2920 RI = NB - BA *  INT (NB / BA)
2930  IF RI < 10 THEN RI$ =  CHR$ (RI + 48)
2940  IF RI > 9 THEN RI$ =  CHR$ (RI + 55)
2950 RE$ = RI$ + RE$
2960  IF  INT (NB / BA) > 0 THEN NB =  INT (NB / BA): GOTO
    2920
2970  RETURN
```

c. Exemples commentés

● ***Exemple 1*** :

```
    CHANGEMENT DE BASE


BASE DE DEPART:5
BASE D'ARRIVEE:3

VOTRE NOMBRE:3421

EN BASE 3: 200000
```

```
VOTRE NOMBRE:3420

EN BASE 3: 122222

VOTRE NOMBRE:3333

EN BASE 3: 122100

VOTRE NOMBRE:0


VOULEZ-VOUS REUTILISER LE PROGRAMME?:O


    CHANGEMENT DE BASE


BASE DE DEPART:3
BASE D'ARRIVEE:5

VOTRE NOMBRE:200000

EN BASE 5: 3421

VOTRE NOMBRE:1211221

EN BASE 5: 20343

VOTRE NOMBRE:0


VOULEZ-VOUS REUTILISER LE PROGRAMME?:O
```

● ***Exemple 2*** :

```
    CHANGEMENT DE BASE


BASE DE DEPART:10
BASE D'ARRIVEE:16

VOTRE NOMBRE:65535
```

```
EN BASE 16: FFFF

VOTRE NOMBRE:256

EN BASE 16: 100

VOTRE NOMBRE:100

EN BASE 16: 64

VOTRE NOMBRE:1024

EN BASE 16: 400

VOTRE NOMBRE:0


VOULEZ-VOUS REUTILISER LE PROGRAMME?:O


    CHANGEMENT DE BASE


BASE DE DEPART:16
BASE D'ARRIVEE:2

VOTRE NOMBRE:FF

EN BASE 2: 11111111

VOTRE NOMBRE:3A

EN BASE 2: 111010

VOTRE NOMBRE:E2

EN BASE 2: 11100010

VOTRE NOMBRE:0


VOULEZ-VOUS REUTILISER LE PROGRAMME?:N
```

Pour les utilisateurs du langage machine, les bases 2 et 16 jouent un rôle particulier.

En effet, l'introduction de programmes écrits dans ce langage s'effectue en hexadécimal, et les différentes conversions entre bases 2, 10 et 16 sont très fréquentes.

3. *Opérations élémentaires dans une base quelconque*

a. Principe

Nous allons effectuer les différentes opérations :

- → addition +
- → soustraction −
- → multiplication *
- → division entière /

sur des nombres exprimés dans une base quelconque.

b. Programme

Après avoir choisi la base, il faut introduire les deux nombres puis l'opérateur à leur appliquer, de la même manière que sur les calculatrices utilisant la notation polonaise inversée.

Les calculs peuvent ensuite s'enchaîner, le résultat précédent jouant le rôle de premier opérande.

Le programme traduit en fait les nombres en base 10 avant d'effectuer l'opération, puis convertit le résultat dans la base de départ.

Il n'est pas nécessaire de réintroduire les sous-programmes 28ØØ et 291Ø utilisés par le programme précédent : "changement de base" si celui-ci est toujours en mémoire.

```
2000 CLS
2010  PRINT  TAB( 5);"OPERATIONS ELEMENTAIRES DANS UNE"
    : PRINT
2020  PRINT  TAB( 13);"BASE QUELCONQUE": PRINT : PRINT

2030  INPUT "BASE:";BD
```

```
2040 BD =  INT ( ABS (BD))
2050  IF BD < 2 OR BD > 36 THEN 2030
2060 BA = BD
2070 ER = 0
2080  PRINT
2090  INPUT "NOMBRE DE DEPART:";NB$
2100  PRINT
2110  GOSUB 2800
2120  IF ER = 1 THEN 2070
2130 N1 = NB
2140  INPUT "NOMBRE SUIVANT:";NB$
2150  IF NB$ = "0" THEN 2330
2160  GOSUB 2800
2170  IF ER = 1 THEN 2070
2180 N2 = NB
2190  PRINT
2200  INPUT "OPERATEUR:";OP$
2210 OP = 0
2220  IF OP$ = "+" THEN OP = 1
2230  IF OP$ = "-" THEN OP = 2
2240  IF OP$ = "*" THEN OP = 3
2250  IF OP$ = "/" THEN OP = 4
2260  IF OP = 0 THEN 2200
2270  ON OP GOSUB 2400,2500,2600,2700
2280  GOSUB 2910
2290  PRINT
2300  PRINT "RESULTAT : ";RE$
2310  PRINT
2320  GOTO 2140
2330  PRINT : PRINT
2340  INPUT "VOULEZ-VOUS REUTILISER LE PROGRAMME?:";Z$
2350  IF Z$ = "O" THEN 2000
2360  END
2400 N1 = N1 + N2
2410 NB = N1
2420  RETURN
2500 N1 =  ABS (N1 - N2)
2510 NB = N1
2520  RETURN
2600 N1 = N1 * N2
2610 NB = N1
2620  RETURN
2700 N1 =  INT (N1 / N2)
2710 NB = N1
2720  RETURN
```

```
2800 LG =  LEN (NB$)
2810 NB = 0
2820  FOR I = 0 TO LG - 1
2830 NI =  ASC ( MID$ (NB$,LG - I,1))
2840  IF NI < 48 OR NI > 90 OR (NI > 57 AND NI < 65) THEN
    ER = 1:I = LG - 1: GOTO 2890
2850  IF NI < 58 THEN NI = NI - 48
2860  IF NI > 64 THEN NI = NI - 55
2870  IF NI >  = BD THEN ER = 1:I = LG - 1: GOTO 2890
2880 NB = NB + NI * BD ^ I
2890  NEXT I
2900  RETURN
2910 RE$ = ""
2920 RI = NB - BA *  INT (NB / BA)
2930  IF RI < 10 THEN RI$ =  CHR$ (RI + 48)
2940  IF RI > 9 THEN RI$ =  CHR$ (RI + 55)
2950 RE$ = RI$ + RE$
2960  IF  INT (NB / BA) > 0 THEN NB =  INT (NB / BA): GOTO
    2920
2970  RETURN
```

c. Exemples commentés

- ***Exemple 1*** :

```
    OPERATIONS ELEMENTAIRES DANS UNE

            BASE QUELCONQUE


BASE:25

NOMBRE DE DEPART:AM3G

NOMBRE SUIVANT:28

OPERATEUR:+

RESULTAT : AM50

NOMBRE SUIVANT:J

OPERATEUR:*
```

```
RESULTAT : 86MD6

NOMBRE SUIVANT:2B

OPERATEUR:/

RESULTAT : 39JM

NOMBRE SUIVANT:0

VOULEZ-VOUS REUTILISER LE PROGRAMME?:N
```

- ***Exemple 2*** :

```
    OPERATIONS ELEMENTAIRES DANS UNE

            BASE QUELCONQUE

BASE:2

NOMBRE DE DEPART:11001010

NOMBRE SUIVANT:1100

OPERATEUR:-

RESULTAT : 10111110

NOMBRE SUIVANT:1110

OPERATEUR:-

RESULTAT : 10110000

NOMBRE SUIVANT:10101

OPERATEUR:+

RESULTAT : 11000101

NOMBRE SUIVANT:10
```

```
OPERATEUR:/

RESULTAT : 1100010

NOMBRE SUIVANT:11

OPERATEUR:*

RESULTAT : 100100110

NOMBRE SUIVANT:0


VOULEZ-VOUS REUTILISER LE PROGRAMME?:O


    OPERATIONS ELEMENTAIRES DANS UNE

            BASE QUELCONQUE

BASE:16

NOMBRE DE DEPART:FEAA

NOMBRE SUIVANT:FF

OPERATEUR:+

RESULTAT : FFA9

NOMBRE SUIVANT:1799

OPERATEUR:-

RESULTAT : E810

NOMBRE SUIVANT:2

OPERATEUR:/

RESULTAT : 7408

NOMBRE SUIVANT:0

VOULEZ-VOUS REUTILISER LE PROGRAMME?:O
```

Toujours lors de la programmation en langage machine, les calculs en base 2 ou 16 sont très fréquents; calculs d'adresses, de masques,...

4. *Nombres premiers*

a. Principe de la recherche

Rappelons qu'un nombre premier est un entier naturel dont les seuls diviseurs sont 1 et lui-même, et que la suite des nombres premiers est infinie.

Pour tester la primarité d'un nombre n, nous pouvons utiliser l'algorithme suivant: divisons n successivement par tous les entiers qui lui sont inférieurs, et si aucune de ces divisions ne fournit un reste nul, le nombre est premier.

Il est possible d'améliorer facilement cet algorithme, dans le but d'augmenter la rapidité de la recherche.

On remarque immédiatement que les nombres pairs supérieurs à 2 ne sont pas premiers; on pourra donc les éliminer systématiquement.

De même, si 2 n'est pas diviseur de n, tous les nombres pairs ne seront pas non plus diviseurs de n.

Enfin, si $n = a.b$, a et b ne peuvent être tous deux supérieurs à $\sqrt{n}$; on pourra donc limiter la recherche aux entiers inférieurs à $\sqrt{n}$.

b. Programme

Le programme recherche et affiche tous les nombres premiers compris entre deux bornes qu'il faut lui préciser.

```
3000 CLS
3010  PRINT  TAB( 5);"NOMBRES PREMIERS": PRINT : PRINT

3020  INPUT "LIMITE INFERIEURE:";MN
3030  INPUT "LIMITE SUPERIEURE:";MX
3040  PRINT
3050  IF MN >  = MX THEN 3020
3060 MN =  INT ( ABS (MN))
3070 MX =  INT ( ABS (MX))
3080  IF MN < 3 THEN MN = 2: PRINT 2,
3090  IF MN = 2 *  INT (MN / 2) THEN MN = MN + 1
3100  IF MN = 3 THEN  PRINT 3,
```

```
3110  FOR NB = MN TO MX STEP 2
3120 RC =  INT ( SQR (NB)) + 1
3130 P = 1
3140  FOR D = 3 TO RC STEP 2
3150 Q = NB / D
3160  IF Q =  INT (Q) THEN P = 0:D = RC
3170  NEXT D
3180  IF P = 1 THEN  PRINT NB,
3190  NEXT NB
3200  PRINT : PRINT
3210  INPUT "VOULEZ-VOUS REUTILISER LE PROGRAMME?:";Z$
3220  IF Z$ = "O" THEN 3000
3230  END
```

c. Exemples commentés

- ***Exemple 1*** :

```
    NOMBRES PREMIERS


LIMITE INFERIEURE:2
LIMITE SUPERIEURE:200

2                3                5
7                11               13
17               19               23
29               31               37
41               43               47
53               59               61
67               71               73
79               83               89
97               101              103
107              109              113
127              131              137
139              149              151
157              163              167
173              179              181
191              193              197
199

VOULEZ-VOUS REUTILISER LE PROGRAMME?:O
```

Le programme trouve les nombres premiers inférieurs à 2ØØ en une dizaine de secondes environ.

● ***Exemple 2*** :

```
    NOMBRES PREMIERS


LIMITE INFERIEURE:1000
LIMITE SUPERIEURE:1200

1009            1013            1019
1021            1031            1033
1039            1049            1051
1061            1063            1069
1087            1091            1093
1097            1103            1109
1117            1123            1129
1151            1153            1163
1171            1181            1187
1193

VOULEZ-VOUS REUTILISER LE PROGRAMME?:O
```

Il met vingt secondes pour les nombres premiers compris entre 1 ØØØ et 1 2ØØ : la vitesse encore élevée s'explique par le fait qu'à ce niveau, la recherche nécessite au plus 17 divisions par nombre, puisque $\sqrt{1\,2ØØ} \simeq 34{,}5$.

5. Décomposition en facteurs premiers

a. Principe

La décomposition en facteurs premiers consiste à écrire un entier naturel sous forme d'un produit de nombres premiers.

Par exemple : $60 = 2^2 \times 3 \times 5$.

La recherche s'effectue comme pour les nombres premiers : le nombre n à décomposer est divisé successivement par 2 puis par tous les nombres impairs inférieurs à sa racine carrée, et ceci autant de fois que le facteur figure dans la décomposition.

b. Programme

Il suffit d'introduire le nombre à décomposer; le programme affiche alors les facteurs et leur ordre de multiplicité.

```
4000 CLS
4010  PRINT  TAB( 5);"DECOMPOSITION EN FACTEURS PREMIER
    S": PRINT : PRINT
4020  INPUT "VOTRE NOMBRE:";NB
4030  PRINT
4040 NB =  INT ( ABS (NB))
4050  IF NB = 0 THEN 4260
4060  IF NB = 1 THEN 4020
4070 D = 2:P = 0
4080 Q = NB / D
4090  IF Q =  INT (Q) THEN P = P + 1:NB = Q: GOTO 4080
4100  IF P <  > 0 THEN  PRINT D;
4110  IF P = 1 THEN  PRINT
4120  IF P > 1 THEN  PRINT " PUISSANCE ";P
4130 D = 1
4140 P = 0
4150 D = D + 2
4160  IF D >  SQR (NB) THEN 4230
4170 Q = NB / D
4180  IF Q =  INT (Q) THEN P = P + 1:NB = Q: GOTO 4170
4190  IF P <  > 0 THEN  PRINT D;
4200  IF P = 1 THEN  PRINT
4210  IF P > 1 THEN  PRINT " PUISSANCE ";P
4220  GOTO 4140
4230  IF NB <  > 1 THEN  PRINT NB
4240  PRINT
4250  GOTO 4020
4260  INPUT "VOULEZ-VOUS REUTILISER LE PROGRAMME?:";Z$
4270  IF Z$ = "O" THEN 4000
4280  END
```

c. Exemple commenté

● ***Exemple***:

```
    DECOMPOSITION EN FACTEURS PREMIERS


VOTRE NOMBRE:234432
```

```
2 PUISSANCE 6
3 PUISSANCE 2
11
37

VOTRE NOMBRE:1441

11
131

VOTRE NOMBRE:65536

2 PUISSANCE 16

VOTRE NOMBRE:0

VOULEZ-VOUS REUTILISER LE PROGRAMME?:N
```

Les trois décompositions sont obtenues en quelques secondes.

On peut également utiliser ce programme dans le but de vérifier si un nombre est premier ou composé.

6. *PGCD et PPCM de plusieurs nombres*

a. Définitions

Soient deux entiers a et b, désignons par D l'ensemble de leurs diviseurs communs et par M l'ensemble de leur multiples communs.

On appelle alors PGCD (plus grand commun diviseur) de a et b le plus grand élément de D, et PPCM (plus petit commun multiple) de a et b le plus petit élément de M.

Le PGCD de a et b sera noté δ et leur PPCM μ ; on peut démontrer la relation :

$$ab = \mu\delta$$

Si $\delta = 1$, on dit que les nombres a et b sont premiers entre eux.

Plus généralement, si l'on considère N nombres $n_1, n_2, \ldots, n_N$, le PGCD de ces nombres sera le plus grand élément de l'ensemble de leurs diviseurs communs, leur PPCM le plus petit élément de l'ensemble de leurs multiples communs.

Si le PGCD des N nombres vaut 1, on dit que les N nombres sont premiers entre eux dans leur ensemble. Ils sont premiers entre eux deux à deux si les PGCD des nombres pris deux à deux sont tous égaux à 1.

b. Algorithme d'Euclide

L'algorithme d'Euclide est une méthode de calcul permettant d'obtenir le PGCD de deux nombres.

On procède par divisions euclidiennes successives :

$$
\begin{aligned}
a &= b\, q_1 + r_1 \\
b &= r_1\, q_2 + r_2 \\
&\cdots\cdots\cdots\cdots \\
r_{n-2} &= r_{n-1} \cdot q_n + r_n \\
r_{n-1} &= r_n \cdot q_{n+1}
\end{aligned}
$$

jusqu'à obtenir un reste nul (ici r_{n+1}) ; le PGCD est alors le dernier reste non nul, ici r_n.

Le PPCM s'obtient ensuite grâce à la relation :

$$ab = \mu\delta$$

D'autre part, on calcule le PGCD de p nombres en substituant successivement deux nombres par leur PGCD :

$$
\begin{aligned}
\delta_1 &= \text{PGCD}\,(n_1, n_2) \\
\delta_2 &= \text{PGCD}\,(\delta_1, n_3) = \text{PGCD}\,(n_1, n_2, n_3) \\
\delta_3 &= \text{PGCD}\,(\delta_2, n_4) = \text{PGCD}\,(n_1, n_2, n_3, n_4) \\
&\cdots\cdots\cdots\cdots \\
\delta_{p-1} &= \text{PGCD}\,(\delta_{p-2}, n_p) = \text{PGCD}\,(n_1, \ldots, n_p)
\end{aligned}
$$

Les PPCM sont également calculés de proche en proche :

$$
\begin{aligned}
\mu_1 &= \frac{n_1\, n_2}{\delta_1} = \text{PPCM}\,(n_1, n_2) \\
\mu_2 &= \frac{\delta_1\, n_3}{\delta_2} = \text{PPCM}\,(n_1, n_2, n_3) \\
&\cdots\cdots\cdots\cdots \\
\mu_{p-1} &= \frac{\delta_{p-2} \cdot n_p}{\delta_{p-1}} = \text{PPCM}\,(n_1, \ldots, n_p)
\end{aligned}
$$

c. Programme

Il suffit d'introduire successivement les entiers en ayant préalablement précisé leur nombre: le programme calcule et affiche alors les valeurs du PPCM et du PGCD.

```
5000 CLS
5010  PRINT  TAB( 5);"PGCD ET PPCM DE N NOMBRES": PRINT
    : PRINT
5020  INPUT "COMBIEN DE NOMBRES Y A-T-IL?:";N
5030 N =  INT ( ABS (N))
5040  IF N < 2 THEN 5020
5050  PRINT
5060  INPUT "PREMIER NOMBRE:";PG
5070 PG =  INT ( ABS (PG))
5080 PP = PG
5090  FOR I = 2 TO N
5100  INPUT "NOMBRE SUIVANT:";NB
5110 NB =  INT ( ABS (NB))
5120  IF NB > PG THEN A = NB:B = PG: GOTO 5140
5130 A = PG:B = NB
5140  GOSUB 5300
5150 PG = A
5160  IF NB > PP THEN A = NB:B = PP: GOTO 5180
5170 A = PP:B = NB
5180  GOSUB 5300
5190 PP = PP * NB / A
5200  NEXT I
5210  PRINT : PRINT "PGCD=";PG
5220  PRINT : PRINT "PPCM=";PP
5230  PRINT : INPUT "VOULEZ-VOUS REUTILISER LE PROGRAMM
    E?:";Z$
5240  IF Z$ = "O" THEN 5000
5250  END
5300 R = A - B *  INT (A / B)
5310 A = B:B = R
5320  IF R <  > 0 THEN 5300
5330  RETURN
```

d. Exemples commentés

● ***Exemple 1*** :

```
    PGCD ET PPCM DE N NOMBRES


COMBIEN DE NOMBRES Y A-T-IL?:4

PREMIER NOMBRE:13068
NOMBRE SUIVANT:90882
NOMBRE SUIVANT:31944
NOMBRE SUIVANT:29376

PGCD=6

PPCM=351895104

VOULEZ-VOUS REUTILISER LE PROGRAMME?:O
```

● ***Exemple 2*** :

```
    PGCD ET PPCM DE N NOMBRES


COMBIEN DE NOMBRES Y A-T-IL?:3

PREMIER NOMBRE:9936
NOMBRE SUIVANT:414
NOMBRE SUIVANT:3174

PGCD=138

PPCM=228528

VOULEZ-VOUS REUTILISER LE PROGRAMME?:O
```

● ***Exemple 3*** :

Nous allons préciser avec cet exemple la notion de nombres premiers entre eux dans leur ensemble et premiers entre eux deux à deux.

```
PGCD ET PPCM DE N NOMBRES
```

```
COMBIEN DE NOMBRES Y A-T-IL?:3

PREMIER NOMBRE:105
NOMBRE SUIVANT:715
NOMBRE SUIVANT:561

PGCD=1

PPCM=255255

VOULEZ-VOUS REUTILISER LE PROGRAMME?:0
```

Le PGCD de 105, 715 et 561 est 1 ; les trois nombres sont donc premiers entre eux dans leur ensemble.

```
    PGCD ET PPCM DE N NOMBRES


COMBIEN DE NOMBRES Y A-T-IL?:2

PREMIER NOMBRE:105
NOMBRE SUIVANT:715

PGCD=5

PPCM=15015

VOULEZ-VOUS REUTILISER LE PROGRAMME?:0


    PGCD ET PPCM DE N NOMBRES


COMBIEN DE NOMBRES Y A-T-IL?:2

PREMIER NOMBRE:105
NOMBRE SUIVANT:561

PGCD=3

PPCM=19635

VOULEZ-VOUS REUTILISER LE PROGRAMME?:0
```

```
    PGCD ET PPCM DE N NOMBRES


COMBIEN DE NOMBRES Y A-T-IL?:2

PREMIER NOMBRE:715
NOMBRE SUIVANT:561

PGCD=11

PPCM=36465

VOULEZ-VOUS REUTILISER LE PROGRAMME?:N

    PGCD ET PPCM DE N NOMBRES
```

Ils ne sont cependant pas premiers entre eux deux à deux ; il aurait fallu pour cela que les trois PGCD soient égaux à 1.

7. Résolution dans $\mathbb{Z}$ de l'équation ax + by = c

a. Rappels théoriques

Résoudre dans **Z** l'équation $ax + by = c$, c'est déterminer tous les couples (x,y) d'entiers relatifs vérifiant la relation.

L'équation ne peut être résolue que si le PGCD de a et b divise c ; il y a alors une infinité de solutions.

Posons : $\delta = \text{PGCD}\,(a,b)$ et $a' = \dfrac{a}{\delta}$, $b' = \dfrac{b}{\delta}$, $c' = \dfrac{c}{\delta}$

L'équation devient alors :

$a'x + b'y = c'$ avec a′ et b′ premiers entre eux.

Supposons que l'on connaisse une solution particulière (x_0,y_0) ; on a donc :

$a'x_0 + b'y_0 = c'$

On a toujours :

$a'x + b'y = c'$

Effectuons alors la différence de ces deux expressions ; il vient :

$a'(x-x_0) + b'(y-y_0) = 0$

a' et b' étant premiers entre eux, b' divise $(x-x_0)$ et a' divise $(y-y_0)$:

$\exists k \in \mathbb{Z}$ tel que $\quad x - x_0 = kb'$

soit $\quad x = x_0 + kb'$

En reportant dans l'équation, on obtient:

$y = y_0 - a'k$

Il est donc possible d'obtenir toutes les solutions à partir d'une solution particulière (x_0,y_0) grâce aux relations:

$$\begin{cases} x = x_0 + b'k \\ y = y_0 - a'k \end{cases}$$

b. Programme

Il suffit d'introduire les valeurs des coefficients a, b et c; dans le cas où l'équation admet des solutions, le programme les affiche.

La solution particulière est déterminée par essais successifs. A partir de 0, on fixe alternativement l'une des valeurs x ou y et on calcule l'autre avec l'équation; si le résultat obtenu est un nombre entier, le couple est solution de l'équation.

```
6000 CLS
6010  PRINT  TAB( 5);"RESOLUTION DANS Z DE A*X+B*Y=C": PRINT
   : PRINT
6020  INPUT "COEFFICIENT A:";A
6030  INPUT "COEFFICIENT B:";B
6040  INPUT "COEFFICIENT C:";C
6050 A =  INT (A):B =  INT (B):C =  INT (C)
6060  IF A > B THEN A1 = A:B1 = B
6070  IF A <  = B THEN A1 = B:B1 = A
6080 R = A1 - B1 *  INT (A1 / B1)
6090 A1 = B1:B1 = R
6100  IF R <  > 0 THEN 6080
6110 PG = A1
6120 A = A / PG
6130 B = B / PG
6140 C = C / PG
6150  IF C <  >  INT (C) THEN 6380
6160 X1 = 0:Y2 = 0
6170 Y1 = (C - X1 * A) / B
6180  IF  INT (Y1) = Y1 THEN 6270
6190 X1 =  - X1
6200  IF X1 >  = 0 THEN X1 = X1 + 1
```

```
6210 X2 = (C - Y2 * B) / A
6220  IF  INT (X2) = X2 THEN 6260
6230 Y2 =  - Y2
6240  IF Y2 >  = 0 THEN Y2 = Y2 + 1
6250  GOTO 6170
6260 X1 = X2:Y1 = Y2
6270  PRINT : PRINT
6280  PRINT "LES SOLUTIONS SONT:": PRINT : PRINT
6290  PRINT "  X=";X1;"+"; ABS (B);"*K"
6300  PRINT "  Y=";Y1;
6310  IF  SGN (A * B) = 1 THEN  PRINT "-";
6320  IF  SGN (A * B) =  - 1 THEN  PRINT "+";
6330  PRINT  ABS (A);"*K"
6340  PRINT
6350  INPUT "VOULEZ-VOUS REUTILISER LE PROGRAMME?:";Z$
6360  IF Z$ = "O" THEN 6000
6370  END
6380  PRINT : PRINT "PAS DE SOLUTION"
6390  GOTO 6340
```

c. Exemples commentés

- ***Exemple 1*** :

```
    RESOLUTION DANS Z DE A*X+B*Y=C


COEFFICIENT A:2
COEFFICIENT B:4
COEFFICIENT C:5

PAS DE SOLUTION

VOULEZ-VOUS REUTILISER LE PROGRAMME?:N
```

Le PGCD de 2 et 4 vaut 2, et 2 ne divise pas 5; l'équation n'a donc pas de solutions.

● ***Exemple 2*** :

```
    RESOLUTION DANS Z DE A*X+B*Y=C


COEFFICIENT A:3
COEFFICIENT B:-2
COEFFICIENT C:7


LES SOLUTIONS SONT:


  X=1+2*K
  Y=-2+3*K

VOULEZ-VOUS REUTILISER LE PROGRAMME?:O
```

Cette fois, le PGCD de 3 et 2 vaut 1 ; l'équation admet les solutions :

$$\begin{cases} x = 1 + 2k \\ y = -2 + 3k \end{cases} \qquad k \in \mathbb{Z}$$

● ***Exemple 3*** :

```
    RESOLUTION DANS Z DE A*X+B*Y=C


COEFFICIENT A:1
COEFFICIENT B:1
COEFFICIENT C:1


LES SOLUTIONS SONT:


  X=0+1*K
  Y=1-1*K

VOULEZ-VOUS REUTILISER LE PROGRAMME?:O
```

8. *Analyse combinatoire*

a. Rappels théoriques

Depuis toujours, les entiers naturels ont été utilisés pour dénombrer les éléments de certains ensembles.

On définit dans ce but des applications, des injections, des bijections de parties de $\mathbb{N}$ vers l'ensemble considéré.

• *Nombre d'applications d'un ensemble fini dans un autre* :

Soit l'intervalle de $\mathbb{N}$ $[1,p]$ et un ensemble E de n éléments ; le nombre d'applications de $[1,p]$ vers E est alors le nombre de p-uplets que l'on peut former à partir des éléments de E.

On démontre que le nombre d'applications de $[1,p]$ vers E est égal à p^n.

• *Nombre d'injections d'un ensemble fini dans un autre* :

Dans un cas où l'on considère des injections à la place d'applications, on obtient le nombre de p-uplets que l'on peut former à partir des n éléments de E, mais sans répéter un quelconque élément.

Remarquons que l'existence d'injections de $[1,p]$ vers E suppose $p \leqslant n$.

On démontre que le nombre d'injections vaut :

$$A_n^p = n(n-1)(n-2)\ldots(n-p+1) = \frac{n!}{(n-p)!}$$

• *Nombre de bijections d'un ensemble fini dans un autre* :

Dans le cas où $n = p$, les injections sont alors des bijections et le nombre de ces bijections vaut $n!$ (factorielle n) :

$$n! = n(n-1)(n-2)\ldots 3.2.1$$

avec la convention $\emptyset ! = 1! = 1$.

• *Nombre de combinaisons de p éléments de E* :

On appelle combinaison de p éléments de E (de cardinal n) toute partie de E de cardinal p.

Le nombre de ces combinaisons est noté C_n^p et vaut :

$$C_n^p = \frac{A_n^p}{p!} = \frac{n!}{p!(n-p)!} \quad p \leqslant n$$

b. Programme

Le programme effectue le calcul de n !, recherche la primitive d'une factorielle, calcule le nombre d'arrangements et de combinaisons.

Le calcul des grandes factorielles est effectué grâce à la formule de Stirling:

$n! \# \sqrt{2\pi n}.n^n.e^{-n}$ pour n grand.

```
10 PI = 3.1415926536
20 MX = 30
7000 CLS
7010  PRINT  TAB( 5);"ANALYSE COMBINATOIRE": PRINT
7020  PRINT : PRINT "1- FACTORIELLE"
7030  PRINT : PRINT "2- FACTORIELLE INVERSE"
7040  PRINT : PRINT "3- ARRANGEMENT"
7050  PRINT : PRINT "4- COMBINAISON"
7060  PRINT : INPUT "VOTRE CHOIX ";E
7070  ON E GOSUB 7200,7400,7600,7800
7080  PRINT : INPUT "VOULEZ-VOUS REUTILISER LE PROGRAMME
     ";Z$
7090  IF Z$ = "O" GOTO 7000
7100  END
7200  PRINT : PRINT "CALCUL DE N!"
7210  PRINT : INPUT "VALEUR DE N ";N
7220 N =  INT ( ABS (N))
7230 FA = 1
7240  IF N < 2 GOTO 7290
7250  IF N > MX GOTO 7310
7260  FOR I = 1 TO N
7270 FA = FA * I
7280  NEXT I
7290  PRINT "N! =";FA
7300  GOTO 7350
7310 NL = ( LOG (2 * PI * N) / 2 + N *  LOG (N) - N) /  LOG
    (10)
7320 EX =  INT (NL)
7330 MA = 10 ^ (NL - EX)
7340  PRINT "N! =";MA;" 10^";EX
7350  RETURN
7400  PRINT : PRINT "FACTORIELLE INVERSE"
7410  PRINT : INPUT "VOTRE NOMBRE ";N
7420 N =  INT ( ABS (N))
7430  IF N < 1 GOTO 7410
7440 I = 1:F = N
7450 F = F / I
```

```
7460  IF F > 1 THEN I = I + 1: GOTO 7450
7470  IF F <  > 1 THEN  PRINT "LE NOMBRE N'EST PAS UNE F
   ACTORIELLE": GOTO 7490
7480  PRINT N;"=";I;"!"
7490  RETURN
7600  PRINT : PRINT "CALCUL DE A(N,P)"
7610  PRINT : INPUT "VALEUR DE N ";N
7620  PRINT : INPUT "VALEUR DE P ";P
7630 N =  INT ( ABS (N)):P =  INT ( ABS (P))
7640  IF P > N GOTO 7610
7650 A = 1
7660  FOR I = N - P + 1 TO N
7670 A = A * I
7680  NEXT I
7690  PRINT : PRINT "A (";N;",";P;") =";A
7700  RETURN
7800  PRINT : PRINT "CALCUL DE C(N,P)"
7810  PRINT : INPUT "VALEUR DE N ";N
7820  PRINT : INPUT "VALEUR DE P ";P
7830 N =  INT ( ABS (N)):U =  INT ( ABS (P))
7840  IF P > N GOTO 7810
7850 C = 1
7860  IF P > N - P THEN U = N - P
7870  FOR I = 0 TO U - 1
7880 C = C * (N - I) / (U - I)
7890  NEXT I
7900  PRINT : PRINT "C (";N;",";P;") =";C
7910  RETURN
```

c. Exemples commentés

● ***Exemple 1*** :

Quel est le nombre de mots de cinq lettres distinctes que l'on peut former à partir du mot TABLE ?

C'est le nombre de bijections de $[1,5]$ vers l'ensemble $\{T,A,B,L,E\}$, soit 5 !

```
CALCUL DE N!

VALEUR DE N 5
N! =120
```

● ***Exemple 2*** :

Pour les nombres élevés, la formule de Stirling, utilisée en logarithme, permet de dépasser la capacité de $1Ø^{39}$ des ordinateurs usuels.

```
CALCUL DE N!

VALEUR DE N 100
N! =9.32484765 10^157

CALCUL DE N!

VALEUR DE N 1000
N! =4.02353313 10^2567
```

● ***Exemple 3*** :

Nous allons chercher si les nombres 36288Ø, 72Ø et 6228 sont des factorielles :

```
FACTORIELLE INVERSE

VOTRE NOMBRE 362880
362880=9!

FACTORIELLE INVERSE

VOTRE NOMBRE 720
720=6!

FACTORIELLE INVERSE

VOTRE NOMBRE 6228
LE NOMBRE N'EST PAS UNE FACTORIELLE
```

Le programme répond affirmativement pour les deux premiers nombres et fournit leur primitive.

● ***Exemple 4*** :

Soient dix jetons numérotés de 1 à 1Ø, et placés dans une urne. On tire successivement six jetons, sans remise.

Nous sommes donc dans le cas des arrangements, puisque l'on tient compte de l'ordre.

Le nombre de tirages possibles est donc A_{10}^{6}.

```
CALCUL DE A(N,P)

VALEUR DE N 10

VALEUR DE P 6

A (10,6) =151200
```

● ***Exemple 5*** :

Soient sept boules de couleurs différentes ; on tire simultanément trois boules. Combien peut-on obtenir de triplets distincts ?

Nous sommes ici dans le cas où l'ordre du tirage n'intervient pas ; il nous suffit donc de connaître le nombre de sous-ensembles de trois éléments que l'on peut constituer à partir des sept boules de départ.

Il vaut donc C_7^3 :

```
CALCUL DE C(N,P)

VALEUR DE N 7

VALEUR DE P 3

C (7,3) =35
```

2 Géométrie

1. Introduction

Après avoir été écartée des programmes scolaires pendant quelques années au profit de théories plus abstraites (espaces vectoriels notamment), la géométrie, un des tout premiers domaines étudiés dans l'histoire des mathématiques, revient actuellement en force dans les classes secondaires, car les raisonnements qu'elle implique développent l'intuition et la vision dans l'espace.

Nous nous intéresserons ici tout d'abord aux calculs de surfaces et d'arcs de cercles, puis à la recherche et à la réduction de coniques, enfin à la résolution de triangles.

Pour utiliser les programmes simultanément, il faut remplacer les instructions END par GOTO 1ØØ, et ajouter le menu :

```
10 PI = 3.1415926536
20  DIM XX(100),YY(100),AA(5,6),MM$(9),ID(3),VL(3)
30  FOR I = 1 TO 3: READ MM$(I): NEXT I
40  DATA  " DEGRES "," RADIANS "," GRADES "
```

```
80 VL(1) = 180:VL(2) = PI:VL(3) = 200
90 TP = 1:MO = VL(TP)
100 CLS
110  PRINT  TAB( 10);"GEOMETRIE": PRINT
120  PRINT "LES ANGLES SONT EXPRIMES EN";MM$(TP): PRINT
   : PRINT
130  PRINT "1- CHANGEMENT DE MODE - CONVERSIONS": PRINT

140  PRINT "2- CALCULS DE SURFACES": PRINT
150  PRINT "3- CALCULS D'ARCS DE CERCLES": PRINT
160  PRINT "4- REDUCTIONS DE CONIQUES": PRINT
170  PRINT "5- RECHERCHES DE CONIQUES": PRINT
180  PRINT "6- RESOLUTION D'UN TRIANGLE QUELCONQUE": PRINT

190  PRINT "7- FIN": PRINT
200  INPUT "VOTRE CHOIX:";E
210  ON E GOTO 1000,2000,3000,4000,5000,6000,7000
220  GOTO 200
7000  END
```

L'ensemble nécessite environ 11 Koctets de mémoire.

2. Conversions d'angles et changement de mode

Le programme suivant (qui doit nécessairement être utilisé avec le menu précédent) réalise deux fonctions :

● *Conversions d'angles* : toutes les conversions sont possibles entre degrés, grades et radians. Il faut tout d'abord choisir le type de conversion, le programme l'effectue ensuite pour tout nombre introduit. Pour revenir au menu, il faut entrer Ø comme nombre à convertir.

● *Changement de mode* : ceci permet de modifier l'unité des angles utilisés dans les programmes de calcul d'arcs de cercles et de résolution de triangles. Ainsi, si l'on choisit l'option "Passages en mode grades", tous les angles devront être exprimés en grades, et les résultats fournis par le programme seront également donnés en grades. Le mode de fonctionnement par défaut est le mode "degrés".

```
1000 CLS
1010  PRINT " CHANGEMENT DE MODE - CONVERSIONS": PRINT :
    PRINT
1020  PRINT "1- PASSAGE EN MODE DEGRES": PRINT
1030  PRINT "2- PASSAGE EN MODE RADIANS": PRINT
```

```
1040  PRINT "3- PASSAGE EN MODE GRADES": PRINT
1050  PRINT "4- CONVERSION DEGRES->RADIANS"
1060  PRINT "5- CONVERSION RADIANS->DEGRES": PRINT
1070  PRINT "6- CONVERSION DEGRES->GRADES"
1080  PRINT "7- CONVERSION GRADES->DEGRES": PRINT
1090  PRINT "8- CONVERSION GRADES->RADIANS"
1100  PRINT "9- CONVERSION RADIANS->GRADES": PRINT
1110  PRINT "10- RETOUR AU MENU PRINCIPAL": PRINT
1120  INPUT "VOTRE CHOIX ";E
1130  IF E < 1 GOTO 1120
1140  IF E >  = 10 GOTO 100
1150  IF E < 4 THEN TP = E:MO = VL(E): GOTO 1000
1160  IF E = 4 THEN T1 = 1:T2 = 2
1170  IF E = 5 THEN T1 = 2:T2 = 1
1180  IF E = 6 THEN T1 = 1:T2 = 3
1190  IF E = 7 THEN T1 = 3:T2 = 1
1200  IF E = 8 THEN T1 = 3:T2 = 2
1210  IF E = 9 THEN T1 = 2:T2 = 3
1220  PRINT : PRINT "NOMBRE EN";MM$(T1);
1230  INPUT N
1240  IF N = 0 GOTO 1000
1250  PRINT N;MM$(T1);"= ";N * VL(T2) / VL(T1);MM$(T2)
1260  GOTO 1220
```

Cherchons par exemple la valeur en radians de 50 grades ($\pi/4$) et la valeur en degrés de 1 radian :

```
NOMBRE EN GRADES ?50
50 GRADES = .785398163 RADIANS

NOMBRE EN GRADES ?0

NOMBRE EN RADIANS ?1
1 RADIANS = 57.2957795 DEGRES

NOMBRE EN RADIANS ?0
```

3. *Calculs de surfaces*

a. Principe

Considérons un contour fermé quelconque; nous allons chercher une valeur approchée de sa surface intérieure.



Pour cela, nous allons approximer le contour ($\mathscr{C}$) par une série de points appartenant à ($\mathscr{C}$), reliés entre eux par des segments de droite:



Le polynôme défini par les points M_1, M_2, ..., $M_{12} = M_1$ constitue une approximation du contour ($\mathscr{C}$).

L'aire de ce polygône peut être calculée de manière très simple par le programme; nous obtiendrons alors une valeur approchée de la surface initiale S.

Bien sûr, plus le nombre de points du contour sera grand, meilleure sera la précision du résultat.

b. Programme

Il suffit de fournir au programme les coordonnées des points définissant le contour.

L'origine du repère orthonormé choisi pour exprimer les coordonnées n'a pas d'incidence sur le résultat, et on peut par exemple prendre un des points du contour pour origine.

Par contre, les points doivent impérativement être introduits dans l'ordre où ils se présentent sur le contour.

Le dernier point doit être confondu avec le premier ; une fois entré, le programme commence automatiquement le calcul, et affiche la valeur de la surface.

Si le contour admet des points doubles (c'est-à-dire s'il se recoupe en un ou plusieurs points), il faut choisir une série de points définissant une seule surface intérieure :



Dans le cas de figure ci-dessus, le résultat fourni par le programme sera faux, car le contour choisi ne définit pas une surface intérieure. Il faut choisir les points de la manière suivante :



On peut donc retenir qu'il ne faut pas faire apparaître d'intersections lors de la sélection des points délimitant le contour.

Pour l'exemple ci-dessus, il est bien sûr possible de calculer séparément l'aire des deux lobes, et d'en faire la somme ensuite pour obtenir l'aire totale.

Le nombre maximal de points est fixé à 100 à la ligne 2Ø du programme. On peut éventuellement l'augmenter, sachant que cent points supplémentaires nécessitent environ 1 Koctet de mémoire.

```
20  DIM XX(100),YY(100),AA(5,6),MM$(9),ID(3),VL(3)
2000 CLS
2010  PRINT  TAB( 5);"CALCULS DE SURFACES"
2020 N = 0: PRINT : PRINT
2030 N = N + 1
2040  PRINT : PRINT "X";N;: INPUT XX(N)
2050  PRINT "Y";N;: INPUT YY(N)
2060  IF N = 1 OR XX(N) <  > XX(1) OR YY(N) <  > YY(1) GOTO
    2030
2070 S = 0
2080  FOR I = 1 TO N - 1
2090 S = S + (XX(I) + XX(I + 1)) * (YY(I) - YY(I + 1))
2100  NEXT I
2110  PRINT : PRINT "L'AIRE VAUT:"; ABS (S) / 2
2120  PRINT : INPUT "VOULEZ-VOUS REUTILISER LE PROGRAMME
    ?:";Z$
2130  IF Z$ = "O" GOTO 2000
2140  END
```

c. Exemples commentés

- ***Exemple 1*** :

Considérons le polygône suivant :



```
    CALCULS DE SURFACES


X1?3
Y1?3

X2?12
Y2?3

X3?12
Y3?6

X4?11
Y4?6

X5?11
Y5?7

X6?15
Y6?7

X7?15
Y7?9

X8?8
Y8?9

X9?8
Y9?6

X10?4
Y10?6

X11?4
Y11?9

X12?3
Y12?9

X13?3
Y13?3

L'AIRE VAUT:47

VOULEZ-VOUS REUTILISER LE PROGRAMME?:N
```

Le résultat fourni par le programme, 47, correspond bien au nombre de carreaux unité contenus dans le contour.

• ***Exemple 2*** :

Nous allons effectuer un calcul approximatif de la valeur de π.

Considérons pour cela le cercle de centre O de rayon 1, représentons le graphiquement et effectuons vingt mesures réparties sur la circonférence :




```
CALCULS DE SURFACES


X1?1
Y1?0

X2?.95
Y2?.31

X3?.81
Y3?.59

X4?.59
Y4?.81

X5?.31
Y5?.95

X6?0
Y6?1

X7?-.31
Y7?.95
```

```
X8?-.59
Y8?.81

X9?-.81
Y9?.59

X10?-.95
Y10?.31

X11?-1
Y11?0

X12?-.95
Y12?-.31

X13?-.81
Y13?-.59

X14?-.59
Y14?-.81

X15?-.31
Y15?-.95

X16?0
Y16?-1

X17?.31
Y17?-.95

X18?.59
Y18?-.81

X19?.81
Y19?-.59

X20?.95
Y20?-.31

X21?1
Y21?0

L'AIRE VAUT:3.0936

VOULEZ-VOUS REUTILISER LE PROGRAMME?:N
```

Le programme donne S = 3,0936, ce qui correspond à une valeur approchée de π à $5 \cdot 10^{-2}$ près.

On aurait naturellement obtenu un meilleur résultat en mesurant plus précisément les points, ou en en prenant un nombre plus grand.

4. *Calculs d'arcs de cercle*

a. Principe

Un arc de cercle peut être caractérisé par quatre paramètres :



le rayon du cercle	R
l'angle d'ouverture	θ
l'abscisse curviligne	s
la corde	c

Ces quatre paramètres ne sont pas indépendants ; deux suffisent en effet à caractériser un arc de cercle donné.

Les formules de passage sont les suivantes :

$$\begin{cases} s = R \cdot \theta \\ c = 2\,R \sin\left(\frac{\theta}{2}\right) \end{cases}$$

Déduire les paramètres manquants pose cependant une difficulté lorsque les données sont s et c.

En effet, il n'existe alors pas de formule algébrique donnant R, celui-ci étant solution de l'équation :

$$2\,R \sin\left(\frac{s}{2R}\right) - c = 0$$

Nous allons la résoudre par la méthode de Newton. (Voir Tome 1, résolution d'équations).

Pour cela, posons $x = 2R$ et partons de $x_0 = c$ (valeur minimale possible pour x).

Calculons alors :

$$x_{n+1} = x_n \left[\frac{c - s\cos\left(\frac{s}{x_n}\right)}{x_n \sin\left(\frac{s}{x_n}\right) - s\cos\left(\frac{s}{x_n}\right)} \right]$$

jusqu'à ce que la précision maximale (10^{-8}) soit atteinte.

Nous sommes ici dans un cas où la méthode de Newton permet d'obtenir très rapidement un résultat précis.

b. Programme

Le programme demande successivement les quatre valeurs de R, S, C, T (T représente l'angle θ).

Les deux paramètres connus sont introduits normalement, et on répond par un point d'interrogation pour les paramètres inconnus.

Le programme affiche alors les quatre paramètres, ainsi que la surface du secteur :



$$\Sigma = \frac{s.R}{2}$$

et la surface de la portion comprise entre la corde et le cercle.



$$\Sigma' = \Sigma - \frac{cR}{2}\cos\left(\frac{\theta}{2}\right)$$

REMARQUE : dans le cas où R et c sont connus, le calcul de T fait appel à la fonction Arcsin. Celle-ci étant absente sur la plupart des interpréteurs BASIC, on l'exprime alors en fonction de Arctg :

$$\text{Arcsin } x = \text{Arctg } \left(\frac{x}{\sqrt{1 - x^2}}\right)$$

```
10 PI = 3.1415926536
20  DIM XX(100),YY(100),AA(5,6),MM$(9),ID(3),VL(3)
80 VL(1) = 180:VL(2) = PI:VL(3) = 200
90 TP = 1:MO = VL(TP)
3000 .CLS
3010  PRINT  TAB( 5);"CALCULS D'ARCS DE CERCLES"
3020 NB = 0: PRINT : PRINT
3030  INPUT "T ";Z$: PRINT :T = 0
3040  IF Z$ = "?" THEN NB = NB + 1: GOTO 3080
3050 T =  VAL (Z$)
3060  IF T <  = 0 OR T > MO GOTO 3030
3070 T = T * PI / MO
3080  INPUT "R ";Z$: PRINT :R = 0
3090  IF Z$ = "?" THEN NB = NB + 1: GOTO 3120
3100 R =  VAL (Z$)
3110  IF R <  = 0 GOTO 3080
3120  INPUT "S ";Z$: PRINT :S = 0
3130  IF Z$ = "?" THEN NB = NB + 1: GOTO 3160
3140 S =  VAL (Z$)
3150  IF S <  = 0 GOTO 3120
3160  INPUT "C ";Z$: PRINT :C = 0
3170  IF Z$ = "?" THEN NB = NB + 1: GOTO 3200
3180 C =  VAL (Z$)
3190  IF C <  = 0 GOTO 3160
3200  PRINT
3210  IF NB <  > 2 THEN  PRINT "VOUS DEVEZ ENTRER 2 PARA
   METRES...": GOTO 3020
3220  IF R > 0 AND C > R + R THEN  PRINT "C DOIT ETRE IN
   FERIEUR A 2R": GOTO 3020
3230  IF S > 0 AND C >  = S THEN  PRINT "C DOIT ETRE INF
   ERIEUR A S": GOTO 3020
3240  IF C > 0 AND 2 * S > PI * C THEN  PRINT "S DOIT ET
   RE INFERIEUR A PI*C/2": GOTO 3020
3250  IF R > 0 AND S > PI * R THEN  PRINT "S DOIT ETRE I
   NFERIEUR A R*PI"
3260  IF T = 0 GOTO 3300
3270  IF R > 0 THEN S = R * T: GOTO 3410
3280  IF S > 0 THEN R = S / T: GOTO 3410
```

```
3290  IF C > 0 THEN R = C / 2 /  SIN (T / 2):S = R * T: GOTO
    3420
3300  IF C = 0 THEN T = S / R: GOTO 3410
3310  IF R = 0 GOTO 3340
3320 T = PI: IF C < 2 * R THEN T = 2 *  ATN (C /  SQR (4
     * R * R - C * C))
3330 S = R * T: GOTO 3420
3340 X = C
3350 U = S / X:A = S *  COS (U):B = X *  SIN (U) - A
3360  IF (C - A) * B <  = 0 THEN  PRINT "C ET S SONT TEL
    S QUE R>";Y / 2: GOTO 3500
3370 Y = X * (C - A) / B
3380  IF  ABS (Y - X) > 1E - 8 THEN X = Y: GOTO 3350
3390 R = Y / 2:T = S / R
3400  GOTO 3420
3410 C = 2 * R *  SIN (T / 2)
3420  HOME : PRINT "T=";MO * T / PI: PRINT
3430  PRINT "R=";R: PRINT
3440  PRINT "S=";S: PRINT
3450  PRINT "C=";C: PRINT
3460 A = S * R / 2: PRINT
3470  PRINT "AIRE DU SECTEUR: ";A: PRINT
3480  PRINT "AIRE COMPRISE ENTRE CORDE ET CERCLE : "
3490  PRINT  TAB( 10);A - C * R *  COS (T / 2) / 2: PRINT

3500  PRINT : INPUT "VOULEZ-VOUS REUTILISER LE PROGRAMME
     ";Z$
3510  IF Z$ = "O" GOTO 3000
3520  END
```

c. Exemples commentés

• ***Exemple 1*** :

Considérons l'arc de cercle défini par:

$$\begin{cases} \theta = 1 \text{ rd} \\ c = 2 \end{cases}$$

```
    CALCULS D'ARCS DE CERCLES

T 1

R ?
```

```
S ?

C 2


T=1

R=2.08582964

S=2.08582964

C=2


AIRE DU SECTEUR: 2.17534265

AIRE COMPRISE ENTRE CORDE ET CERCLE :
         .344854929


VOULEZ-VOUS REUTILISER LE PROGRAMME O
```

Le programme donne alors:

$$\begin{cases} R = 2{,}085\,829\,64 \\ s\ = 2{,}085\,829\,64 \end{cases}$$

ce qui est normal puisque $s = R\theta$ et $\theta = 1$ rd.

- ***Exemple 2*** :

Soient les paramètres:

$$\begin{cases} s = 2{,}5 \\ c = 2 \end{cases}$$

Nous sommes ici dans le cas où le rayon R doit être calculé par la méthode de Newton.

```
    CALCULS D'ARCS DE CERCLES


T ?

R ?
```

```
S 2.5

C 2


T=2.26220517

R=1.10511638

S=2.5

C=2


AIRE DU SECTEUR: 1.38139548

AIRE COMPRISE ENTRE CORDE ET CERCLE :
         .910989036


VOULEZ-VOUS REUTILISER LE PROGRAMME N
```

5. Réduction de coniques

a. Rappels théoriques

On appelle conique une courbe dont l'équation cartésienne se présente sous la forme suivante :

$$ax^2 + 2bxy + cy^2 + 2dx + 2ey + f = 0$$

On distingue trois principales familles de coniques :

- les hyperboles (comprenant en particulier les hyperboles équilatères),
- les paraboles,
- les ellipses (comprenant en particulier les cercles).

Il existe également des coniques dégénérées : droite, réunion de deux droites, ou point.

Le but du chapitre est la détermination, à partir de l'équation cartésienne, du type de la conique et de ses éléments de réduction : centre, foyers, axes et excentricité.

La discussion consiste en une série de tests et de calculs sur les coefficients de l'équation, ce qui se prête particulièrement bien à un traitement par ordinateur.

Voici le principe de cette discussion.

Calculons tout d'abord le discriminant:

$$\Delta = b^2 - ac$$

Si $\Delta = 0$, la conique est une parabole.
Si $\Delta < 0$, la conique est une ellipse.
Si $\Delta > 0$, la conique est une hyperbole.

Premier cas: $\Delta \neq 0$

Le centre Ω de la conique, si elle existe, a pour coordonnées:

$$\Omega \begin{cases} x_1 = \dfrac{dc - be}{\Delta} \\ \\ Y_1 = \dfrac{ae - bd}{\Delta} \end{cases}$$

et l'équation de la conique dans le nouveau repère $(\Omega, \vec{i}, \vec{j})$ est alors:

$$aX^2 + 2bXY + cY^2 + f' = 0$$

avec:

$$X = x - x_1 \qquad Y = y - y_1$$

et

$$f' = ax_1^2 + 2bx_1y_1 + cy_1^2 + 2dx_1 + 2ey_1 + f$$

● *Si $f' = 0$*:

Si $\Delta < 0$, la conique se réduit au point Ω.
Si $\Delta > 0$, la conique est la réunion de deux droites.

● *Si $f' \neq 0$*:

La matrice de la forme quadratique associée à la conique est:

$$\begin{pmatrix} a & b \\ b & c \end{pmatrix}$$

dont les deux valeurs propres sont:

$$\lambda = \frac{a + c - \sqrt{(a - c)^2 + 4b^2}}{2}$$

$$\mu = \frac{a + c + \sqrt{(a - c)^2 + 4b^2}}{2}$$

* Si $a = c$ et $b = 0$:
Si $f'.a > 0$, pas de solution.
Si $f'.a < 0$, la conique est le cercle de centre Ω et de rayon $R = \sqrt{\frac{-f'}{a}}$.

* Si $a \neq c$ ou $b \neq 0$:
On considère les deux nouveaux vecteurs :

$$\vec{I}\begin{pmatrix} \dfrac{b}{\sqrt{b^2 + (\lambda - a)^2}} \\ \\ \dfrac{\lambda - a}{\sqrt{b^2 + (\lambda - a)^2}} \end{pmatrix} \qquad \vec{J}\begin{pmatrix} \dfrac{-(\lambda - a)}{\sqrt{b^2 + (\lambda - a)^2}} \\ \\ \dfrac{b}{\sqrt{b^2 + (\lambda - a)^2}} \end{pmatrix}$$

La base $(\vec{I},\vec{J})$ est orthonormée, et l'équation de la conique dans le repère $(\Omega,\vec{I},\vec{J})$ est :

$$\lambda X^2 + \mu Y^2 + f' = 0$$

→ Si $\Delta < 0$:
Si $\operatorname{sgn}(\lambda) = \operatorname{sgn}(f')$, pas de solution
Si $\operatorname{sgn}(\lambda) \neq \operatorname{sgn}(f')$, la conique est une ellipse de centre Ω et d'axes $(\Omega X,\Omega Y)$.

Ses foyers sont sur le grand axe, à la distance :

$$d(\Omega,F) = \sqrt{a'^2 + b'^2}$$

où :

$$a' = \sqrt{\frac{-f'}{\lambda}} \text{ et } b' = \sqrt{\frac{-f'}{\mu}}$$

sont les demi-longueurs des axes.

L'excentricité de l'ellipse vaut alors :

$$e = \frac{\sqrt{a'^2 + b'^2}}{a'}$$

→ Si $\Delta > 0$:
La conique est une hyperbole de centre Ω et d'axes ΩX, ΩY.

Elle est équilatère si $a + c = 0$.

Si $f' > 0$:

$$a' = \sqrt{\frac{-|f'|}{\lambda}} \qquad b' = \sqrt{\frac{-|f'|}{\mu}}$$

l'axe transverse est ΩX.

Si $f' < 0$:

$$a' = \sqrt{\frac{-|f'|}{\mu}} \qquad b' = \sqrt{\frac{-|f'|}{\lambda}}$$

l'axe transverse est ΩY.

L'excentricité vaut : $\dfrac{\sqrt{a'^2 + b'^2}}{a'}$.

Les foyers sont sur l'axe transverse, à la distance $d(\Omega,F) = \sqrt{a'^2 + b'^2}$.

Les sommets sont sur l'axe transverse, à la distance $d(\Omega,S) = a'$.

Deuxième cas : $\Delta = 0$

- Si a, b et c sont simultanément nuls, la conique est une droite, ou n'existe pas.
- Si a, b et c ne sont pas tous nuls :

* Si $a = 0$ (donc $b = 0$ et $c \neq 0$) :

→ Si $d = 0$

La conique est la réunion de deux droites si $e^2 - cf \geqslant 0$, et n'existe pas si $e^2 - cf < 0$

→ Si $d \neq 0$:

Les coordonnées du centre sont alors :

$$\Omega \begin{cases} x_1 = \dfrac{e^2 - cf}{2cd} \\[2ex] y_1 = -\dfrac{e}{c} \end{cases}$$

L'équation de la conique dans $(\Omega, \vec{i}, \vec{j})$ est alors :

$$Y^2 = -2\frac{d}{c}X$$

C'est donc une parabole, d'axe ΩX, de paramètre $(-\frac{d}{c})$.

Le foyer est sur l'axe de la parabole, à la distance $(-\dfrac{d}{2c})$ du centre.

* Si $a \neq 0$:

On commence par rendre a positif, en multipliant chaque coefficient par sgn (a), puis on pose :

$$\begin{cases} \alpha = \sqrt{a} \\ \beta = \sqrt{c}\,\mathrm{sgn}\,(b) \end{cases}$$

et

$$\vec{I} \begin{pmatrix} \dfrac{\beta}{\sqrt{a+c}} \\ \\ \dfrac{-\alpha}{\sqrt{a+c}} \end{pmatrix} \qquad \vec{J} \begin{pmatrix} \dfrac{\alpha}{\sqrt{a+c}} \\ \\ \dfrac{\beta}{\sqrt{a+c}} \end{pmatrix}$$

L'équation de la conique dans le repère $(O,\vec{I},\vec{J})$ est alors :

$$(\alpha^2 + \beta^2)^{3/2}\, Y^2 + (2d\beta - 2e\alpha) X + (2d\alpha + 2e\beta) Y + f\sqrt{\alpha^2 + \beta^2} = 0$$

Cette équation est du type :

$$c'\, Y^2 + 2d'\, X + 2e'\, Y + f' = 0$$

et on se ramène ainsi au cas $a = 0$ avec les nouveaux paramètres c', d', e' et f'.

Si $d' \neq 0$, le centre de la parabole est :

$$\Omega \begin{cases} x_1 = \dfrac{e'^2 - c'f'}{2c'd'} \\ \\ y_1 = -\dfrac{e'}{c'} \end{cases}$$

et l'équation la parabole dans le repère $(\Omega,\vec{I},\vec{J})$ est :

$$Y = -\frac{2d'}{c'}\, X$$

b. Programme

Il suffit de donner au programme les six coefficients de l'équation cartésienne pour qu'il donne le type de la conique ainsi que tous les éléments de réduction : centre, sommets, foyers et vecteurs directeurs des axes.

Le programme détecte également toutes les formes dégénérées de coniques : ensemble vide, point, droite unique, réunion de deux droites.

REMARQUE : l'équation de la conique s'écrit :

$$ax^2 + 2bxy + cy^2 + 2dx + 2ey + f = 0$$

Cependant, les coefficients à fournir au programme ne sont pas a, b, c, d, e et f mais a, (2b), c, (2d), (2e) et f.

```
4000 CLS
4010  PRINT  TAB( 5);"REDUCTIONS DE CONIQUES": PRINT
4020  INPUT "COEFFICIENT DE X2?";A
4030  INPUT "COEFFICIENT DE XY?";B:B = B / 2
4040  INPUT "COEFFICIENT DE Y2?";C
4050  INPUT "COEFFICIENT DE X ?";D:D = D / 2
4060  INPUT "COEFFICIENT DE Y ?";E:E = E / 2
4070  INPUT "TERME  CONSTANT  ?";F: PRINT
4080 G = A * C - B * B
4090  IF G = 0 GOTO 4680
4100 X1 = (B * E - D * C) / G:Y1 = (B * D - A * E) / G
4110 F = A * X1 * X1 + 2 * B * X1 * Y1 + C * Y1 * Y1 + 2
     * D * X1 + 2 * E * Y1 + F
4120  IF F <  > 0 GOTO 4150
4130  IF G > 0 THEN  PRINT "LA CONIQUE EST REDUITE AU PO
    INT:": PRINT : PRINT "X=";X1: PRINT "Y=";Y1: GOTO 49
    30
4140  PRINT "LA CONIQUE EST LA REUNION DE 2 DROITES": GOTO
    4930
4150 U =  SQR ((A - C) * (A - C) + 4 * B * B)
4160 L = (A + C - U) / 2
4170 M = (A + C + U) / 2
4180  IF A <  > C OR B <  > 0 GOTO 4260
4190  IF F * A < 0 GOTO 4210
4200  PRINT "CETTE CONIQUE N'EXISTE PAS": GOTO 4930
4210  PRINT "CETTE CONIQUE EST LE CERCLE"
4220  PRINT : PRINT  TAB( 5);"DE CENTRE: X=";X1
4230  PRINT  TAB( 16);"Y=";Y1
4240  PRINT : PRINT  TAB( 5);"DE RAYON:  R="; SQR ( - F /
    A)
4250  GOTO 4930
4260  IF B = 0 AND A < C THEN X2 = 1:X = 1:Y = 0:Y2 = 0:
     GOTO 4290
4270 X = B:Y = L - A:U =  SQR (X * X + Y * Y)
4280 X2 = X / U:Y2 = Y / U
4290  IF G < 0 GOTO 4480
4300  IF L * F > 0 GOTO 4200
4310 A =  SQR ( - F / L):B =  SQR ( - F / M)
4320  IF L < 0 THEN U = A:A = B:B = U:U = X2:X2 =  - Y2:
    Y2 = U:U = X:X =  - Y:Y = U
4330  PRINT "CETTE CONIQUE EST UNE ELLIPSE": PRINT
4340  PRINT "CENTRE:  X=";X1;"  Y=";Y1: PRINT
```

```
4350  PRINT "VECTEUR DIRECTEUR DU GRAND AXE:"
4360  PRINT  TAB( 5);"X=";X;"  Y=";Y
4370  PRINT : PRINT "VECTEUR DIRECTEUR DU PETIT AXE:"
4380  PRINT  TAB( 5);"X="; - Y;"  Y=";X
4390  PRINT : PRINT "DEMIE-LONGUEUR DU GRAND AXE:";A
4400  PRINT "DEMIE-LONGUEUR DU PETIT AXE:";B
4410 U =  SQR (A * A - B * B)
4420 X = X1 + X2 * U:Y = Y1 + Y2 * U
4430  PRINT : PRINT "FOYERS 1:  X=";X;"  Y=";Y
4440 X = X1 - X2 * U:Y = Y1 - Y2 * U
4450  PRINT  TAB( 8);"2:  X=";X;"  Y=";Y
4460  PRINT : PRINT "EXCENTRICITE:";U / A
4470  GOTO 4930
4480  PRINT "CETTE CONIQUE EST UNE HYPERBOLE"
4490  IF A + C = 0 THEN  PRINT  TAB( 12);"EQUILATERE"
4500  PRINT : PRINT "CENTRE:  X=";X1;"  Y=";Y1: PRINT
4510 A =  SQR ( -  ABS (F) / L):B =  SQR ( ABS (F) / M)
4520  IF F < 0 THEN U = A:A = B:B = U:U = X2:X2 =  - Y2:
    Y2 = U:U = X:X =  - Y:Y = U
4530  PRINT "VECTEUR DIRECTEUR DE L'AXE TRANSVERSAL:"
4540  PRINT  TAB( 5);"X=";X;"  Y=";Y: PRINT
4550  PRINT "VECTEUR DIRECTEUR DE L'AXE CONJUGUE:"
4560  PRINT  TAB( 5);"X="; - Y;"  Y=";X
4570 X = X1 + A * X2:Y = Y1 + A * Y2
4580  PRINT : PRINT "SOMMETS 1:  X=";X;"  Y=";Y
4590 X = X1 - A * X2:Y = Y1 - A * Y2
4600  PRINT  TAB( 9);"2:  X=";X;"  Y=";Y
4610 U =  SQR (A * A + B * B)
4620 X = X1 + X2 * U:Y = Y1 + Y2 * U
4630  PRINT : PRINT "FOYERS 1:  X=";X;"  Y=";Y
4640 X = X1 - X2 * U:Y = Y1 - Y2 * U
4650  PRINT  TAB( 8);"2:  X=";X;"  Y=";Y
4660  PRINT : PRINT "EXCENTRICITE:";U / A
4670  GOTO 4930
4680  IF (C <  > 0) OR (A <  > 0) GOTO 4710
4690  IF D = 0 AND E = 0 GOTO 4200
4700  PRINT "CETTE CONIQUE EST UNE DROITE": GOTO 4930
4710  IF A = 0 THEN X2 = 1:Y2 = 0:L = 0:M = 1: GOTO 4780

4720 F = F *  SGN (A):E = E *  SGN (A):D = D *  SGN (A)
4730 C = C *  SGN (A):B = B *  SGN (A):A =  ABS (A)
4740 L =  SQR (A):M =  SQR (C) *  SGN (B)
4750 U =  SQR (A + C):X2 = M / U:Y2 =  - L / U
4760 F = F * U:C = (A + C) * U
4770 U = D * M - E * L:E = D * L + E * M:D = U
```

```
4780  IF D <  > 0 GOTO 4810
4790  IF E * E < C * F GOTO 4200
4800  GOTO 4140
4810 X = (E * E - C * F) / 2 / C / D:Y =  - E / C
4820 X1 = X * X2 - Y * Y2:Y1 = Y * X2 + X * Y2
4830  PRINT "CETTE CONIQUE EST UNE PARABOLE": PRINT
4840  PRINT "COORDONNEES DU CENTRE:": PRINT  TAB( 5);"X=
    ";X1
4850  PRINT  TAB( 5);"Y=";Y1: PRINT
4860  PRINT "VECTEUR DIRECTEUR DE L'AXE DE SYMETRIE:"
4870  PRINT  TAB( 5);"X=";M: PRINT  TAB( 5);"Y="; - L
4880  PRINT : PRINT "COORDONNES DU  FOYER:"
4890 P =  - D / C
4900  PRINT  TAB( 5);"X=";X1 + P * X2 / 2
4910  PRINT  TAB( 5);"Y=";Y1 + P * Y2 / 2
4920  PRINT : PRINT "PARAMETRE P=";P
4930  PRINT : INPUT "VOULEZ-VOUS REUTILISER LE PROGRAMME
    ?:";Z$
4940  IF Z$ = "O" GOTO 4000
4950  END
```

c. Exemples commentés

● ***Exemple 1*** :

Considérons l'équation suivante :

$$x^2 + y^2 + 2xy - 13x - 11y + 32 = 0$$

```
    REDUCTIONS DE CONIQUES

COEFFICIENT DE X2?1
COEFFICIENT DE XY?2
COEFFICIENT DE Y2?1
COEFFICIENT DE X ?-13
COEFFICIENT DE Y ?-11
TERME  CONSTANT  ?32

CETTE CONIQUE EST UNE PARABOLE

COORDONNEES DU CENTRE:
    X=1
    Y=5
```

```
VECTEUR DIRECTEUR DE L'AXE DE SYMETRIE:
    X=1
    Y=-1

COORDONNES DU  FOYER:
    X=1.125
    Y=4.875

PARAMETRE P=.353553391

VOULEZ-VOUS REUTILISER LE PROGRAMME?:N
```

L'équation proposée est donc celle d'une parabole de centre Ω (1,5), et dont l'axe de symétrie admet $(\vec{i} - \vec{j})$ pour vecteur directeur.

Les coordonnées du foyer sont:

$$\begin{cases} x_F = \dfrac{9}{8} \\ y_F = \dfrac{39}{8} \end{cases}$$



- ***Exemple 2*** :

Soit l'équation:

$$9x^2 + 4y^2 - 18x + 16y - 11 = 0$$

```
    REDUCTIONS DE CONIQUES

COEFFICIENT DE X2?9
COEFFICIENT DE XY?0
```

```
COEFFICIENT DE Y2?4
COEFFICIENT DE X ?-18
COEFFICIENT DE Y ?16
TERME  CONSTANT  ?-11

CETTE CONIQUE EST UNE ELLIPSE

CENTRE:  X=1  Y=-2

VECTEUR DIRECTEUR DU GRAND AXE:
    X=0  Y=-5

VECTEUR DIRECTEUR DU PETIT AXE:
    X=5  Y=0

DEMIE-LONGUEUR DU GRAND AXE:3
DEMIE-LONGUEUR DU PETIT AXE:2

FOYERS 1:  X=1  Y=-4.23606798
       2:  X=1  Y=.23606798

EXCENTRICITE:.745355993

VOULEZ-VOUS REUTILISER LE PROGRAMME?:N
```

L'équation proposée est celle d'une ellipse de centre Ω (1,—2), dont le grand axe admet $\vec{j}$ comme vecteur directeur. Les demi-longueurs des axes sont 2 et 3, ce qui nous permet de tracer facilement l'ellipse :



● ***Exemple 3*** :

Cherchons enfin la nature de la conique d'équation :

$$xy + 5x - y - 7 = 0$$

```
    REDUCTIONS DE CONIQUES

COEFFICIENT DE X2?0
COEFFICIENT DE XY?1
COEFFICIENT DE Y2?0
COEFFICIENT DE X ?5
COEFFICIENT DE Y ?-1
TERME  CONSTANT  ?-7

CETTE CONIQUE EST UNE HYPERBOLE
           EQUILATERE

CENTRE:  X=1  Y=-5

VECTEUR DIRECTEUR DE L'AXE TRANSVERSAL:
    X=.5  Y=.5

VECTEUR DIRECTEUR DE L'AXE CONJUGUE:
    X=-.5  Y=.5

SOMMETS 1:  X=2.41421356  Y=-3.58578644
        2:  X=-.414213562  Y=-6.41421356

FOYERS 1:  X=3  Y=-3
       2:  X=-1  Y=-7

EXCENTRICITE:1.41421356

VOULEZ-VOUS REUTILISER LE PROGRAMME?:N
```

L'équation proposée est donc celle d'une hyperbole équilatère de centre $\Omega\,(1,-5)$ et dont l'axe transversal est dirigé par $(\vec{i} + \vec{j})$.

Les foyers sont $F_1\,(3,-3)$ et $F_2\,(-1,-7)$, l'excentricité vaut $\sqrt{2}$, ce qui donne :



6. *Recherche de coniques*

Nous avons vu, dans le paragraphe précédent, comment obtenir, à partir de l'équation cartésienne, le type et les paramètres de la conique associée.

Nous allons maintenant aborder le problème sous un angle différent, et chercher à reconstituer l'équation cartésienne de la conique passant par cinq points donnés.

Remarquons que pour certaines combinaisons de points, on ne trouvera pas de conique non dégénérée ; par exemple, si les points sont alignés, on trouvera une droite.

a. Principe de la recherche

Nous cherchons à obtenir l'équation cartésienne :

$$ax^2 + bxy + cy^2 + dx + ey + f = 0$$

Les inconnues sont donc a, b, c, d, e et f. Comme l'équation peut être multipliée par un nombre quelconque, le nombre effectif d'inconnues est donc cinq.

C'est pourquoi la donnée de cinq points suffit à déterminer parfaitement la conique ; ceux-ci conduisent à un système de cinq équations homogènes à six inconnues.

La résolution se fera par suppositions successives.

Supposons tout d'abord $f \neq 0$; on peut alors fixer $f = 1$, et chercher les autres paramètres en résolvant le système linéaire de cinq équations à cinq inconnues suivant :

$$\begin{cases} x_1^2\,a + x_1y_1\,b + y_1^2\,c + x_1\,d + y_1\,e = -1 \\ x_2^2\,a + x_2y_2\,b + y_2^2\,c + x_2\,d + y_2\,e = -1 \\ \dots\dots\dots\dots\dots\dots\dots\dots\dots \\ x_5^2\,a + x_5y_5\,b + y_5^2\,c + x_5\,d + y_5\,e = -1 \end{cases}$$

La résolution de ce système est faite par la méthode de Gauss-Jordan, exposée dans le chapitre "Matrices et vecteurs".

Si le système admet une solution, le problème est résolu ; sinon, cela signifie que l'hypothèse $f = 0$ était mauvaise.

On suppose alors $e \neq 0$, et on tente de résoudre le système :

$$\begin{cases} x_1^2\,a + x_1y_1\,b + y_1^2\,c + x_1\,d + f = -y_1 \\ x_2^2\,a + x_2y_2\,b + y_2^2\,c + x_2\,d + f = -y_2 \\ \dots\dots\dots\dots\dots\dots\dots\dots\dots \\ x_5^2\,a + x_5y_5 b + y_5^2 c + x_5 d + f = -y_5 \end{cases}$$

S'il admet une solution, le problème est résolu. Sinon, on poursuit les hypothèses ($d \neq 0$, puis $c \neq 0$,...).

Dans le cas le plus défavorable, c'est-à-dire s'il n'existe pas de conique non dégénérée qui réponde au problème, le programme devra résoudre six systèmes d'ordre 5.

Dans les cas les plus courants ($f \neq 0$ ou $e \neq 0$), il n'en résoud qu'un ou deux.

b. Programme

Il faut introduire les cinq couples de coordonnées.

Le programme vérifie qu'un même point n'est pas introduit plusieurs fois, ce qui empêcherait la résolution.

Au bout de quelques secondes, il affiche les six coefficients de l'équation de la conique ainsi que les approximations rationnelles de ces coefficients.

```
20  DIM XX(100),YY(100),AA(5,6),MM$(9),ID(3),VL(3)
50  FOR I = 4 TO 8: READ Z$:MM$(I) = "COEFF. DE " + Z$ +
    " :": NEXT I
60  DATA  "X2","XY","Y2","X ","Y "
70 MM$(9) = "CONSTANTE     :"
100  GOTO 5000
500  IF X = 0 THEN NN = 0:DD = 1: GOTO 590
510 AX =  ABS (X):AB = AX:AC =  INT (AB):NN = AC
520 DD = 1:AD = 1:AF = 0: GOTO 570
530 AB = 1 / (AB - AC):AC =  INT (AB)
540 AY = AC * NN + AD:AD = NN:NN = AY
550 AY = AC * DD + AF:AF = DD:DD = AY
560  IF DD > 1E5 OR DD * NN > 1E9 THEN NN = AD:DD = AF: GOTO
    580
570  IF 1E7 *  ABS (AX * DD - NN) > AX * DD GOTO 530
580 NN = NN *  SGN (X)
590  RETURN
5000 CLS
5010  PRINT  TAB( 5);"RECHERCHE DE LA CONIQUE PASSANT"
5020  PRINT  TAB( 10);"PAR 5 POINTS DISTINCTS"
5030  FOR I = 1 TO 5
5040  PRINT : PRINT "X";I;: INPUT X
5050  PRINT "Y";I;: INPUT Y
5060 XX(10 + I) = X:YY(10 + I) = Y
5070  IF I = 1 GOTO 5130
5080 ER = 0
5090  FOR J = 11 TO I + 9
5100  IF X = XX(J) AND Y = YY(J) THEN ER = 1:J = I + 9
5110  NEXT J
5120  IF ER = 1 GOTO 5040
5130  NEXT I
5140 N = 5:NN = 6
5150  FOR I = 1 TO 5
5160 X = XX(I + 10):Y = YY(I + 10)
5170 AA(I,1) = X * X:AA(I,2) = X * Y:AA(I,3) = Y * Y
5180 AA(I,4) = X:AA(I,5) = Y:AA(I,6) = 1
5190 U =  - AA(I,NN)
5200  IF NN < 6 THEN  FOR J = NN TO 5:AA(I,J) = AA(I,J +
    1): NEXT J
5210 AA(I,6) = U
5220  NEXT I
5230 CR = 0
5240  FOR K = 1 TO N
5250 I = K
5260  IF I > N THEN CR = 2:K = N: GOTO 5330
```

```
5270  IF  ABS (AA(I,K)) < 1E - 7 THEN I = I + 1: GOTO 52
    60
5280  IF I > K THEN  FOR J = 1 TO N + 1:M = AA(I,J):AA(I
    ,J) = AA(K,J):AA(K,J) = M: NEXT J
5290  FOR I = K + 1 TO N + 1:AA(K,I) = AA(K,I) / AA(K,K)
    : NEXT I:AA(K,K) = 1
5300  FOR I = 1 TO N
5310  IF I <  > K THEN  FOR J = K + 1 TO N + 1:AA(I,J) =
    AA(I,J) - AA(I,K) * AA(K,J): NEXT J:AA(I,K) = 0
5320  NEXT I
5330  NEXT K
5340  FOR I = 1 TO N:XX(I) = AA(I,N + 1): NEXT I
5350  IF CR = 0 GOTO 5400
5360  IF NN > 1 THEN NN = NN - 1: GOTO 5150
5370  PRINT "IL EXISTE PLUSIEURS CONIQUES DEGENEREES"
5380  PRINT  TAB( 8);"PASSANT PAR CES 5 POINTS": PRINT
5390  GOTO 5460
5400  FOR J = 5 TO NN STEP  - 1:XX(J + 1) = XX(J): NEXT
    J
5410 XX(NN) = 1
5420  FOR I = 1 TO 6
5430  PRINT : PRINT MM$(3 + I);XX(I);
5440 X = XX(I): GOSUB 500: IF DD <  > 1 THEN  PRINT  TAB(
    28);NN;"/";DD;
5450  NEXT I
5460  PRINT : PRINT : INPUT "VOULEZ-VOUS REUTILISER LE P
    ROGRAMME ";Z$
5470  IF Z$ = "O" GOTO 5000
5480  END
```

c. Exemples commentés

● ***Exemple 1*** :

Considérons les cinq points suivants :

M_1 (0,2), M_2 (1,5), M_3 (5,77), M_4 (−4,50), M_5 (2,14)



Le programme cherche alors l'équation de la conique passant par ces cinq points :

```
    RECHERCHE DE LA CONIQUE PASSANT
         PAR 5 POINTS DISTINCTS

X1?0
Y1?2

X2?1
Y2?5

X3?5
Y3?77

X4?-4
Y4?50

X5?2
Y5?14

COEFF. DE X2 :1.5           3/2
COEFF. DE XY :0
COEFF. DE Y2 :0
COEFF. DE X  :0
COEFF. DE Y  :-.5           -1/2
CONSTANTE    :1

VOULEZ-VOUS REUTILISER LE PROGRAMME N
```

L'équation est donc :

$$\frac{3}{2}x^2 - \frac{1}{2}y + 1 = 0$$

ce qui peut encore s'écrire :

$$y = 3x^2 + 2$$

ce qui est bien l'équation d'une parabole.

• ***Exemple 2*** :

Soient les cinq points suivants :

$M_1\,(0,1)$, $M_2\,(0,-1)$, $M_3\,(1,0)$, $M_4\,(-1,0)$, $M_5\,(\frac{\sqrt{2}}{2},\frac{\sqrt{2}}{2})$.



La conique passant par ces points est le cercle de centre O et de rayon 1.

```
     RECHERCHE DE LA CONIQUE PASSANT
          PAR 5 POINTS DISTINCTS

X1?0
Y1?1

X2?0
Y2?-1

X3?1
Y3?0
```

```
X4?-1
Y4?0

X5?.707106781
Y5?.707106781

COEFF. DE X2 :-1
COEFF. DE XY :-3.7252903E-09
COEFF. DE Y2 :-1
COEFF. DE X  :0
COEFF. DE Y  :0
CONSTANTE    :1

VOULEZ-VOUS REUTILISER LE PROGRAMME O
```

L'équation trouvée par le programme est :

$$x^2 + y^2 = 1$$

ce qui est exact.

REMARQUE : le coefficient de xy n'est pas nul, à cause des erreurs numériques ; le nombre décimal entré à la place de $\frac{\sqrt{2}}{2}$ étant forcément inexact.

● ***Exemple 3*** :

Reprenons l'exemple précédent, et examinons l'influence d'une incertitude sur un des points, en l'occurrence M_5.

Prenons M_5 (0.7, 0.7).

```
    RECHERCHE DE LA CONIQUE PASSANT
         PAR 5 POINTS DISTINCTS

X1?0
Y1?1

X2?0
Y2?-1

X3?1
Y3?0

X4?-1
Y4?0
```

```
X5?.7
Y5?.7

COEFF. DE X2 :-1
COEFF. DE XY :-.0408163266 -2/49
COEFF. DE Y2 :-1
COEFF. DE X  :0
COEFF. DE Y  :0
CONSTANTE    :1

VOULEZ-VOUS REUTILISER LE PROGRAMME N
```

L'équation obtenue est :

$$x^2 + y^2 + 0.0408\ xy = 1$$

La conique obtenue n'est plus le cercle, mais une ellipse dont l'excentricité est fonction de l'incertitude sur les coordonnées de M_5.

7. Résolution d'un triangle quelconque

a. Principe

On peut caractériser un triangle quelconque par six paramètres : les longueurs de ses côtés et les valeurs des angles de ses sommets.



Nous noterons a, b et c les longueurs des côtés, et t_a, t_b, t_c les valeurs des angles (t_a est la mesure de l'angle opposé au côté a,...).

La connaissance de trois paramètres suffit à caractériser le triangle, donc permet de calculer les trois autres, sauf dans le cas où les trois paramètres connus sont les angles : les côtés ne peuvent pas être déterminés puisqu'aucune unité de longueur n'est donnée.

Nous allons donc examiner comment, à partir de trois paramètres du triangle, déterminer les trois autres.

Les formules utilisées pour cela sont les suivantes :

$$t_a + t_b + t_c = 180^\circ$$
$$c^2 = a^2 + b^2 - 2ab\cos(t_c)$$
$$\frac{a}{\sin(t_a)} = \frac{b}{\sin(t_b)} = \frac{c}{\sin(t_c)}$$

Il se présente donc quatre cas de figure, selon la nature des trois paramètres donnés :

- les trois côtés ;
- deux côtés et l'angle opposé au troisième côté ;
- deux côtés et un angle opposé à un de ces côtés ;
- un côté et deux angles.

* *Cas où les trois côtés a, b et c sont données*:

Les angles sont alors obtenus par :

$$\begin{cases} t_c = \operatorname{Arc\,cos}\left[\dfrac{a^2 + b^2 - c^2}{2ab}\right] \\ t_b = \operatorname{Arc\,cos}\left[\dfrac{a^2 + c^2 - b^2}{2ac}\right] \\ t_a = \operatorname{Arc\,cos}\left[\dfrac{b^2 + c^2 - a^2}{2bc}\right] \end{cases}$$

* *Cas où deux côtés et l'angle opposé au troisième côté sont donnés:*

Par exemple, supposons a, b et t_c connus.

On calcule alors :

$$c = \sqrt{a^2 + b^2 - 2ab\cos t_c}$$

Les trois côtés étant alors connus, on détermine les deux autres angles comme précédemment.

* *Cas où deux côtés et un angle opposé à un de ces côtés sont donnés:*

Par exemple, supposons a, b et t_a connus.

Ce cas est le plus compliqué ; selon les paramètres, on peut ne pas avoir de triangle solution, en avoir un ou en avoir deux.

→ Si $t_a < 90°$
- si $a \geqslant b$, une solution $c = b \cos(t_a) + \sqrt{a^2 - b^2 \sin^2(t_a)}$
- si $a < b$, deux solutions $c = b \cos(t_a) \pm \sqrt{a^2 - b^2 \sin^2(t_a)}$

→ Si $t_a \geqslant 90°$
- si $a \leqslant b$, pas de solution.
- si $a > b$, une solution $c = b \cos(t_a) + \sqrt{a^2 - b^2 \sin^2(t_a)}$.

Après avoir déterminé c, on connaît les trois côtés, et on obtient les angles inconnus comme précédemment.

* Cas où un côté et deux angles sont donnés :

On obtient le troisième angle par :

$$t_c = 180° - t_a - t_b$$

Les deux côtés inconnus sont alors calculés par :

$$b = a\,\frac{\sin(t_b)}{\sin(t_a)} \quad \text{et } c = a\,\frac{\sin(t_c)}{\sin(t_a)}$$

Dans tous les cas, on peut calculer la surface du triangle par la formule :

$$S = \sqrt{h\,(h-a)\,(h-b)\,(h-c)}$$

avec

$$h = \frac{a+b+c}{2}$$

b. Programme

Le programme demande les six paramètres du triangle : comme avec le programme de calcul d'arcs de cercles, il convient d'entrer un point d'interrogation lorsque les paramètres demandés sont les inconnues.

Il calcule alors et affiche tous les paramètres du triangle, ainsi que sa surface.

Les angles doivent appartenir à l'intervalle ouvert]0, 180°[, et les mesures des côtés doivent être strictement positives.

REMARQUE : la fonction Arc cos étant absente de la plupart des interpréteurs BASIC, le programme utilise la formule :

$$\text{Arc cos } x = \frac{\pi}{2} - \text{Arctg}\left(\frac{x}{\sqrt{1-x^2}}\right)$$

```
10 PI = 3.1415926536
20  DIM XX(100),YY(100),AA(5,6),MM$(9),ID(3),VL(3)
80 VL(1) = 180:VL(2) = PI:VL(3) = 200
90 TP = 1:MO = VL(TP)
6000 CLS: PRINT "RESOLUTION D'UN TRIANGLE QUELCONQUE"
6010  PRINT : PRINT :NB = 0
6020  PRINT : INPUT "A ";Z$:AA(1,1) = 0
6030  IF Z$ = "?" THEN NB = NB + 1: GOTO 6060
6040 AA(1,1) =  VAL (Z$)
6050  IF AA(1,1) <  = 0 GOTO 6020
6060  PRINT : INPUT "B ";Z$:AA(1,2) = 0
6070  IF Z$ = "?" THEN NB = NB + 1: GOTO 6100
6080 AA(1,2) =  VAL (Z$)
6090  IF AA(1,2) <  = 0 GOTO 6060
6100  PRINT : INPUT "C ";Z$:AA(1,3) = 0
6110  IF Z$ = "?" THEN NB = NB + 1: GOTO 6140
6120 AA(1,3) =  VAL (Z$)
6130  IF AA(1,3) <  = 0 GOTO 6100
6140 N1 = 3 - NB: PRINT
6150  IF N1 = 0 THEN  PRINT "VOUS DEVEZ ENTRER AU MOINS
   UN COTE": GOTO 6020
6160  INPUT "TA (ANGLE OPPOSE AU COTE A) ";Z$
6170  PRINT :AA(2,1) = 0
6180  IF Z$ = "?" THEN NB = NB + 1: GOTO 6210
6190 AA(2,1) =  VAL (Z$)
6200  IF AA(2,1) <  = 0 OR AA(2,1) >  = MO GOTO 6160
6210  INPUT "TB (ANGLE OPPOSE AU COTE B) ";Z$
6220  PRINT :AA(2,2) = 0
6230  IF Z$ = "?" THEN NB = NB + 1: GOTO 6260
6240 AA(2,2) =  VAL (Z$)
6250  IF AA(2,2) <  = 0 OR AA(2,2) >  = MO GOTO 6210
6260  INPUT "TC (ANGLE OPPOSE AU COTE C) ";Z$
6270  PRINT :AA(2,3) = 0
6280  IF Z$ = "?" THEN NB = NB + 1: GOTO 6310
6290 AA(2,3) =  VAL (Z$)
6300  IF AA(2,3) <  = 0 OR AA(2,3) >  = MO GOTO 6260
6310  IF NB <  > 3 THEN  PRINT "VOUS DEVEZ ENTRER 3 PARA
   METRES": GOTO 6010
```

```
6320  FOR I = 1 TO 3:AA(2,I) = AA(2,I) * PI / MO:ID(I) =
    I: NEXT I
6330  ON N1 GOTO 6630,6340,6410
6340  IF AA(1,3) = 0 GOTO 6380
6350 I = 2:J = 3
6360  IF AA(1,1) = 0 THEN I = 1
6370  GOSUB 6820
6380  IF AA(2,3) <  > 0 GOTO 6590
6390  IF AA(2,1) = 0 THEN I = 1:J = 2: GOSUB 6820
6400  GOTO 6450
6410 A = AA(1,1):B = AA(1,2):C = AA(1,3)
6420  IF C <  =  ABS (A - B) OR C >  = A + B GOTO 6720
6430  FOR I = 1 TO 3: GOSUB 6730: NEXT I
6440  GOTO 6780
6450 A = AA(1,1):B = AA(1,2):TA = AA(2,1)
6460 D = B *  SIN (TA)
6470  IF A < D GOTO 6720
6480 F =  SQR (A * A - D * D)
6490  IF TA < PI / 2 GOTO 6520
6500  IF A <  = B GOTO 6720
6510 AA(1,3) = AA(1,2) *  COS (TA) + F: GOTO 6430
6520  IF A >  = B GOTO 6510
6530  PRINT  TAB( 5);"PREMIER TRIANGLE:"
6540 AA(1,3) = B *  COS (TA) - F
6550  FOR I = 2 TO 3: GOSUB 6730: NEXT I
6560  GOSUB 6860: PRINT
6570  PRINT  TAB( 5);"DEUXIEME TRIANGLE:"
6580  GOTO 6510
6590 A = AA(1,1):B = AA(1,2)
6600 AA(1,3) =  SQR (A * A + B * B - 2 * A * B *  COS (A
    A(2,3)))
6610  FOR I = 1 TO 2: GOSUB 6730: NEXT I
6620  GOTO 6780
6630  FOR I = 1 TO 3
6640  IF AA(2,I) = 0 THEN AA(2,I) = PI - AA(2,1) - AA(2,
    2) - AA(2,3)
6650  NEXT I
6660  IF AA(1,2) <  > 0 THEN I = 1:J = 2: GOSUB 6820
6670  IF AA(1,3) <  > 0 THEN I = 1:J = 3: GOSUB 6820
6680 A = AA(1,1):TA = AA(2,1)
6690 AA(1,2) = A *  SIN (AA(2,2)) /  SIN (TA)
6700 AA(1,3) = A *  SIN (AA(2,3)) /  SIN (TA)
6710  GOTO 6780
6720  PRINT : PRINT "IL N'EXISTE PAS DE TRIANGLE SOLUTIO
    N": GOTO 6790
```

```
6730 J = I + 1: IF J > 3 THEN J = J - 3
6740 K = I + 2: IF K > 3 THEN K = K - 3
6750 X = (AA(1,J) * AA(1,J) + AA(1,K) * AA(1,K) - AA(1,I
    ) * AA(1,I)) / 2 / AA(1,J) / AA(1,K)
6760 AA(2,I) = PI / 2 -  ATN (X /  SQR (1 - X * X))
6770  RETURN
6780  GOSUB 6860
6790  PRINT : INPUT "VOULEZ-VOUS REUTILISER LE PROGRAMME
    ";Z$: TEXT
6800  IF Z$ = "O" GOTO 6000
6810  END
6820 K = ID(I):ID(I) = ID(J):ID(J) = K
6830 K = AA(1,I):AA(1,I) = AA(1,J):AA(1,J) = K
6840 K = AA(2,I):AA(2,I) = AA(2,J):AA(2,J) = K
6850  RETURN
6860 A = AA(1,ID(1)):B = AA(1,ID(2)):C = AA(1,ID(3))
6870 TA = AA(2,ID(1)):TB = AA(2,ID(2)):TC = AA(2,ID(3))
6880  PRINT "A=";A
6890  PRINT "B=";B
6900  PRINT "C=";C: PRINT
6910  PRINT "TA=";MO * TA / PI
6920  PRINT "TB=";MO * TB / PI
6930  PRINT "TC=";MO * TC / PI: PRINT
6940 H = (A + B + C) / 2
6950 S =  SQR (H * (H - A) * (H - B) * (H - C))
6960  PRINT "SURFACE DU TRIANGLE:";S
6970  RETURN
```

c. Exemples commentés

● ***Exemple 1*** :

Considérons le triangle suivant, défini par ses trois côtés:



$$\begin{cases} a = 25 \\ b = 40 \\ c = 58 \end{cases}$$

```
RESOLUTION D'UN TRIANGLE QUELCONQUE


A 25

B 40

C 58

TA (ANGLE OPPOSE AU COTE A) ?

TB (ANGLE OPPOSE AU COTE B) ?

TC (ANGLE OPPOSE AU COTE C) ?

A=25
B=40
C=58

TA=20.750954
TB=34.5336794
TC=124.715367

SURFACE DU TRIANGLE:410.995666

VOULEZ-VOUS REUTILISER LE PROGRAMME N
```

Le programme fournit alors les trois angles :

$$\begin{cases} t_a = 20{,}75° \\ t_b = 34{,}53° \\ t_c = 124{,}7\ ° \end{cases}$$

et la surface :

$S = 410{,}996$

- ***Exemple 2*** :

Soit le triangle suivant, défini par un côté et deux angles :



$$\begin{cases} a = 18 \\ t_b = 110° \\ t_c = 52{,}2° \end{cases}$$

```
RESOLUTION D'UN TRIANGLE QUELCONQUE


A 18

B ?

C ?

TA (ANGLE OPPOSE AU COTE A) ?

TB (ANGLE OPPOSE AU COTE B) 110

TC (ANGLE OPPOSE AU COTE C) 52.2

A=18
B=55.3311316
C=46.5260342

TA=17.8
TB=110
TC=52.2

SURFACE DU TRIANGLE:393.48154

VOULEZ-VOUS REUTILISER LE PROGRAMME N
```

Le programme fournit alors les paramètres complémentaires :

$$\begin{cases} b = 55{,}33 \\ c = 46{,}53 \\ t_a = 17{,}8° \end{cases}$$

et la surface :

$S = 393{,}48$

- ***Exemple 3*** :

Cherchons enfin un triangle défini par deux côtés et un angle :

$$\begin{cases} a = 2\sqrt{3} \\ b = 5{,}6 \\ t_a = 15{,}5° \end{cases}$$

```
RESOLUTION D'UN TRIANGLE QUELCONQUE



A 3.4641016

B 5.6

C ?

TA (ANGLE OPPOSE AU COTE A) 15.5

TB (ANGLE OPPOSE AU COTE B) ?

TC (ANGLE OPPOSE AU COTE C) ?

    PREMIER TRIANGLE:
A=3.4641016
B=5.6
C=2.27216934

TA=15.5
TB=154.404659
TC=10.095341
```

```
SURFACE DU TRIANGLE:1.70019037

    DEUXIEME TRIANGLE:
A=3.4641016
B=5.6
C=8.52049173

TA=15.5
TB=25.5953412
TC=138.904659

SURFACE DU TRIANGLE:6.37560669

VOULEZ-VOUS REUTILISER LE PROGRAMME N
```

Dans le cas de figure, il existe deux triangles solutions:



$$\left\{\begin{array}{l} c = 2{,}27 \\ t_b = 154{,}4° \\ t_c = 10{,}10° \end{array}\right.$$



$$\left\{\begin{array}{l} c = 8{,}5 \\ t_b = 25{,}6° \\ t_c = 139° \end{array}\right.$$

3 Nombres complexes

1. Introduction

Les nombres complexes ont été introduits pour la première fois par des mathématiciens italiens du XVI^e siècle, dans le but de résoudre les équations du troisième degré.

Ils permettent en réalité bien plus, et entre autres la résolution de toute équation du type :

$$a_n x^n + a_{n-1} x^{n-1} + \dots + a_1 x + a_0 = 0$$

où les coefficients a_i sont des réels.

L'emploi des nombres complexes est très répandu chez les physiciens et les techniciens, car ils facilitent les calculs, en les présentant sous une forme beaucoup plus concise ; citons par exemple la transformée de Laplace qui permet entre autres l'étude des régimes permanents sinusoïdaux des systèmes électriques et électroniques, et qui fait appel à une représentation complexe des tensions, intensités et fonctions de transfert.

Le programme que nous décrivons dans ce chapitre se comportera essentiellement comme une calculatrice (utilisant la notation polonaise inversée) possédant toutes les fonctions usuelles (opérations élémentaires, fonction puissance, logarithme, fonctions circulaires et hyperboliques) et permettant le stockage des résultats dans dix mémoires.

Nous pourrons également résoudre des systèmes linéaires de N équations à N inconnues, les équations du second degré, et afficher en haute résolution une représentation du plan complexe.

Il est fortement recommandé d'introduire l'ensemble des programmes du chapitre accompagné du menu suivant, ceci afin d'exploiter leur interactivité ; les résultats des programmes de résolution de systèmes et d'équations du second degré peuvent en effet être réintroduits dans la calculatrice, et être affichés dans le plan complexe.

```
10 MS = 10
20  DIM P(3,1,1),M(9,1),D(2 * MS,2 * MS + 1),EP(1),EN(1)
    ,IN(1,1)
30 PI = 3.1415926536
40 PR = 1E - 8
50 L = 159:C = 279
60  DATA   "T(ALG):","T(POL):","Z(ALG):","Z(POL):", "Y(A
    LG):", "Y(POL):",  "X(ALG):","X(POL):"
70  FOR I = 0 TO 3: FOR J = 0 TO 1: READ PL$(I,J): NEXT
    J,I
100  TEXT :CLS
110  PRINT  TAB( 5);"NOMBRES COMPLEXES": PRINT : PRINT
120  PRINT "1- CALCULS SUR LES NOMBRES COMPLEXES": PRINT

130  PRINT "2- SYSTEME DE N EQUATIONS": PRINT
140  PRINT "3- EQUATION DU SECOND DEGRE": PRINT
150  PRINT "4- REPRESENTATION DANS LE PLAN COMPLEXE": PRINT

160  PRINT "5- FIN": PRINT
170  INPUT "VOTRE CHOIX ";E
180  ON E GOTO 1000,5000,6000,7000,8000
190  GOTO 170
8000  END
```

L'ensemble nécessite environ 12 Koctets de mémoire, dont 2 Koctets pour le stockage des variables.

2. *Calculs sur les nombres complexes*

a. Représentations des nombres complexes

Les nombres complexes sont usuellement représentés de deux manières différentes : la notation rectangulaire ou algébrique, et la notation polaire.

Soit un nombre complexe z ; sa représentation rectangulaire est alors :

$$z = a + ib$$

dans laquelle a et b sont des réels, et i le nombre complexe tel que :

$$i^2 = -1$$

On appelle a partie réelle de z, et ib partie imaginaire de z.

Si $b = 0$, le nombre z est réel ; si $a = 0$, on dit que z est imaginaire pur.

La représentation polaire du nombre z est :

$$z = r(\cos\theta + i\sin\theta)$$

que l'on peut encore écrire sous la forme :

$$z = r\,e^{i\theta}$$

Le réel r est appelé module du nombre z ; il est usuellement noté :

$$r = |z|$$

Le réel θ est appelé argument de z et est noté :

$$\theta = \operatorname{Arg}(z)$$

Il existe bien sûr des relations de passage des coordonnées rectangulaires aux coordonnées polaires ; on a en effet :

$$\begin{cases} r = \sqrt{a^2 + b^2} & \\ \theta = \operatorname{Arctg} \dfrac{b}{a} & \text{si } a \neq 0 \\ \theta = \dfrac{\pi}{2} & \text{si } a = 0 \text{ et } b > 0 \\ \theta = -\dfrac{\pi}{2} & \text{si } a = 0 \text{ et } b < 0 \end{cases}$$

et la conversion inverse est donnée par :

$$\begin{cases} a = r\cos\theta \\ b = r\sin\theta \end{cases}$$

Nous utiliserons l'une ou l'autre des représentations selon le type de l'opération à exécuter ; par exemple, les additions seront effectuées à partir des coordonnées rectangulaires :

si :

$$\begin{cases} z_1 = a_1 + ib_1 \\ z_2 = a_2 + ib_2 \end{cases}$$

alors : $z = z_1 + z_2 = (a_1 + a_2) + i\,(b_1 + b_2)$

et les multiplications à partir des coordonnées polaires :

si :

$$z_1 = r_1\, e^{i\theta_1}$$

$$z_2 = r_2\, e^{i\theta_2}$$

alors :

$$z = z_1 . z_2 = r_1\, r_2\, e^{i(\theta_1 + \theta_2)}$$

alors que l'utilisation des autres coordonnées conduirait à des calculs beaucoup plus lourds.

b. La notation polonaise inversée

Cette méthode de calcul est calquée sur l'architecture des ordinateurs ; les opérandes sont introduits en premiers, et ensuite le code opération est précisé.

Par exemple, si l'on désire effectuer la multiplication 2 * 5, on introduira :

```
2       (RETURN)
5       (RETURN)
*       (RETURN)
```

et l'ordinateur exécutera alors l'opération.

Le principal intérêt de cette notation est de faciliter les calculs chaînés, en supprimant l'emploi des parenthèses utilisées habituellement pour définir les opérations prioritaires.

Les données sont introduites et mémorisées dans une structure appelée pile ; c'est un ensemble de quatre registres X, Y, Z et T, organisés de la manière suivante :



Le premier opérande introduit est placé dans le registre X ; si un second opérande est entré, le registre X est recopié dans le registre Y, et le nouveau nombre est stocké dans X.

De manière plus générale, à chaque fois qu'un nombre est introduit, la pile se décale d'un cran vers le haut :

$$N \rightarrow X \rightarrow Y \rightarrow Z \rightarrow T$$

Remarquons que si la pile est pleine au moment où N est introduit, la valeur que contient T est perdue.

Si alors un code opération est introduit, par exemple *, l'ordinateur calcule alors X*Y, décale la pile vers le bas et place le résultat dans X.

Considérons par exemple le calcul suivant :

$$\sqrt{1 + 3 * \cos(0)}$$

La suite des opérations à effectuer et les états successifs de la pile sont alors les suivants :

1	(RETURN)	T	
		Z	
		Y	
		X	1
3	(RETURN)	T	
		Z	
		Y	1
		X	3
0	(RETURN)	T	
		Z	1
		Y	3
		X	0

COS	(RETURN)	T	
		Z	1
		Y	3
		X	1
*	(RETURN)	T	
		Z	
		Y	1
		X	3
+	(RETURN)	T	
		Z	
		Y	
		X	4
SQR	(RETURN)	T	
		Z	
		Y	
		X	2

REMARQUE : lorsque la pile est décalée vers le bas, la valeur du registre T n'est pas modifiée.

c. Programme

Le programme affiche sur l'écran les quatre registres de la pile, sous les deux représentations rectangulaire et polaire, puis la question "Opération ?".

Pour introduire un nombre dans la pile, il faut répondre par "X" ; on peut alors entrer l'opérande soit sous forme rectangulaire, soit sous forme polaire ; dans ce dernier cas, l'argument devra être exprimé en radians.

Les calculs s'effectuent alors selon les règles précisées dans le paragraphe précédent.

Il existe trois fonctions dont le but est de faciliter l'exploitation de la pile :

"CLX" efface le registre X (et effectue un décalage de la pile vers le bas), ceci pour permettre la correction d'une erreur d'introduction.

"CLS" efface l'ensemble des quatre registres de la pile.

"↑" effectue une rotation de la pile, à savoir :

$$X \leftarrow Y \leftarrow Z \leftarrow T \leftarrow$$

Nous avons prévu la possibilité de conserver certains résultats, en les rangeant dans les dix mémoires prévues à cet effet.

Quatre commandes permettent leur exploitation :

"STOK", où K est un entier compris entre 0 et 9, provoque la recopie du registre X dans la mémoire K.

"RCLK" effectue la recopie de la mémoire K dans le registre X, après avoir décalé la pile vers le haut.

"M" permet de visualiser à chaque instant le contenu des mémoires.

"CLM" efface le contenu de toutes les mémoires.

Toutes les autres opérations que connaît la calculatrice sont des opérations mathématiques, et dont voici la liste :

"+"	: addition $Y + X$
"−"	: soustraction $Y - X$
"*"	: multiplication $Y * X$
"/"	: division Y/X
"EXP"	: exponentielle e^X
"LN"	: logarithme népérien $\ln X$
"X↑2"	: élévation au carré X^2
"SQR"	: racine carrée $\sqrt{X}$
"1/X"	: inverse $1/X$
"Y↑X"	: élévation à la puissance Y^X
"SIN"	: sinus sin (X)
"COS"	: cosinus cos (X)
"TAN"	: tangente tg (X)
"ASIN"	: arcsinus Arcsin (X)
"ACOS"	: arccosinus Arccos (X)

"ATAN" : arctangente Arctg (X)
"CH" : cosinus hyperbolique ch (X)
"SH" : sinus hyperbolique sh (X)
"TH" : tangente hyperbolique th (X)
"ACH" : argument cosinus hyperbolique Arg ch (X)
"ASH" : argument sinus hyperbolique Arg sh (X)
"ATH" : argument tangente hyperbolique Arg th (X)

Nous donnons en annexe les formules donnant les fonctions trigonométriques complexes.

Pour quitter le mode calculatrice, il suffit d'entrer la commande "Q" et le programme retourne au menu principal; toutefois, les registres de la pile et le contenu des mémoires sont conservés.

```
1000 CLS
1010  GOSUB 1500: PRINT : PRINT
1020 CR = 0
1030  INPUT "OPERATION ";OP$
1040  IF OP$ = "X" THEN CR = 1
1050  IF OP$ = "M" THEN CR = 2
1060  IF OP$ = "+" THEN CR = 3
1070  IF OP$ = "-" THEN CR = 4
1080  IF OP$ = "*" THEN CR = 5
1090  IF OP$ = "/" THEN CR = 6
1100  IF OP$ = "EXP" THEN CR = 7
1110  IF OP$ = "LN" THEN CR = 8
1120  IF OP$ = "X^2" THEN CR = 9
1130  IF OP$ = "SQR" THEN CR = 10
1140  IF OP$ = "1/X" THEN CR = 11
1150  IF OP$ = "Y^X" THEN CR = 12
1160  IF OP$ = "SIN" THEN CR = 13
1170  IF OP$ = "COS" THEN CR = 14
1180  IF OP$ = "TAN" THEN CR = 15
1190  IF OP$ = "ACOS" THEN CR = 16
1200  IF OP$ = "ASIN" THEN CR = 17
1210  IF OP$ = "ATAN" THEN CR = 18
1220  IF OP$ = "CH" THEN CR = 19
1230  IF OP$ = "SH" THEN CR = 20
1240  IF OP$ = "TH" THEN CR = 21
1250  IF OP$ = "ACH" THEN CR = 22
1260  IF OP$ = "ASH" THEN CR = 23
1270  IF OP$ = "ATH" THEN CR = 24
```

```
1280  IF  LEN (OP$) < 4 GOTO 1310
1290  IF  LEFT$ (OP$,3) = "STO" THEN CR = 25
1300  IF  LEFT$ (OP$,3) = "RCL" THEN CR = 26
1310  IF OP$ = "CLX" THEN CR = 27
1320  IF OP$ = "CLS" THEN CR = 28
1330  IF OP$ = "CLM" THEN CR = 29
1340  IF OP$ = "^" THEN CR = 30
1350  IF OP$ = "Q" GOTO 100
1360  ON  INT ((CR + 9) / 10) GOSUB 1380,1400,1420
1370  GOTO 1000
1380  ON CR GOSUB 2000,2150,2200,2300,2400,2500,2600,270
    0,2800,2900
1390  RETURN
1400  ON CR - 10 GOSUB 3000,3100,3200,3300,3400,3500,360
    0,3700,3800,3900
1410  RETURN
1420  ON CR - 20 GOSUB 4000,4100,4200,4300,4400,4500,470
    0,4800,4900,4950
1430  RETURN
1500  FOR I = 0 TO 3
1510  FOR J = 0 TO 1
1520  PRINT PL$(I,J); TAB( 8);P(I,J,0);
1530  IF J = 1 THEN  PRINT  TAB( 19);"EXP";
1540  IF P(I,J,1) < 0 THEN  PRINT  TAB( 23);"- I";: GOTO
    1560
1550  PRINT  TAB( 23);"+ I";
1560  PRINT  TAB( 27); ABS (P(I,J,1))
1570  NEXT J
1580  PRINT
1590  NEXT I
1600  RETURN
1700  FOR I = 0 TO 9
1710  PRINT "M";I;":  ";M(I,0); TAB( 21);
1720  IF M(I,1) < 0 THEN  PRINT "- I";: GOTO 1740
1730  PRINT "+ I";
1740  PRINT  TAB( 25); ABS (M(I,1))
1750  NEXT I
1760  RETURN
1800 P(3,1,0) =  SQR (P(3,0,0) * P(3,0,0) + P(3,0,1) * P
    (3,0,1))
1810  IF P(3,0,0) = 0 AND P(3,0,1) > 0 THEN P(3,1,1) = P
    I / 2
1820  IF P(3,0,0) = 0 AND P(3,0,1) < 0 THEN P(3,1,1) =  -
    PI / 2
1830  IF P(3,0,0) = 0 AND P(3,0,1) = 0 THEN P(3,1,1) = 0
```

```
1840  IF P(3,0,0) <  > 0 THEN P(3,1,1) =  ATN (P(3,0,1) /
    P(3,0,0))
1850  IF P(3,0,0) < 0 AND P(3,0,1) >  = 0 THEN P(3,1,1) =
    P(3,1,1) + PI
1860  IF P(3,0,0) < 0 AND P(3,0,1) < 0 THEN P(3,1,1) = P
    (3,1,1) - PI
1870  RETURN
1900 P(3,0,0) = P(3,1,0) *  COS (P(3,1,1))
1910 P(3,0,1) = P(3,1,0) *  SIN (P(3,1,1))
1920  RETURN
2000  FOR I = 1 TO 3: FOR J = 0 TO 1: FOR K = 0 TO 1:P(I
     - 1,J,K) = P(I,J,K): NEXT K,J,I
2010  INPUT "FORME ALGEBRIQUE (A) OU POLAIRE (P) ";F$
2020  IF F$ <  > "A" AND F$ <  > "P" GOTO 2010
2030  IF F$ = "P" GOTO 2080
2040  INPUT "PARTIE REELLE ";P(3,0,0)
2050  INPUT "PARTIE IMAGINAIRE ";P(3,0,1)
2060  GOSUB 1800
2070  RETURN
2080  INPUT "MODULE ";P(3,1,0)
2090  INPUT "ARGUMENT ";P(3,1,1)
2100  GOSUB 1900
2110  RETURN
2150  HOME
2160  GOSUB 1700
2170  PRINT : INPUT "RETOUR A LA PILE ";Z$
2180  RETURN
2200 P(3,0,0) = P(2,0,0) + P(3,0,0)
2210 P(3,0,1) = P(2,0,1) + P(3,0,1)
2220  GOSUB 1800
2230  FOR I = 0 TO 1: FOR J = 0 TO 1: FOR K = 0 TO 1:P(2
     - I,J,K) = P(1 - I,J,K): NEXT K,J,I
2240  RETURN
2300 P(3,0,0) = P(2,0,0) - P(3,0,0)
2310 P(3,0,1) = P(2,0,1) - P(3,0,1)
2320  GOSUB 1800
2330  FOR I = 0 TO 1: FOR J = 0 TO 1: FOR K = 0 TO 1:P(2
     - I,J,K) = P(1 - I,J,K): NEXT K,J,I
2340  RETURN
2400 P(3,1,0) = P(2,1,0) * P(3,1,0)
2410 P(3,1,1) = P(2,1,1) + P(3,1,1)
2420  IF P(3,1,1) > PI THEN P(3,1,1) = P(3,1,1) - 2 * PI

2430  IF P(3,1,1) <  - PI THEN P(3,1,1) = P(3,1,1) + 2 *
    PI
```

```
2440  GOSUB 1900
2450  FOR I = 0 TO 1: FOR J = 0 TO 1: FOR K = 0 TO 1:P(2
    - I,J,K) = P(1 - I,J,K): NEXT K,J,I
2460  RETURN
2500  IF P(3,1,0) = 0 THEN  PRINT "IMPOSSIBLE": GOTO 255
   0
2510 P(3,1,0) = P(2,1,0) / P(3,1,0)
2520 P(3,1,1) = P(2,1,1) - P(3,1,1)
2530  IF P(3,1,1) > PI THEN P(3,1,1) = P(3,1,1) - 2 * PI

2540  IF P(3,1,1) <  - PI THEN P(3,1,1) = P(3,1,1) + 2 *
   PI
2550  GOSUB 1900
2560  FOR I = 0 TO 1: FOR J = 0 TO 1: FOR K = 0 TO 1:P(2
    - I,J,K) = P(1 - I,J,K): NEXT K,J,I
2570  RETURN
2600 P(3,1,0) =  EXP (P(3,0,0)) *  COS (P(3,0,1))
2610 P(3,1,1) =  EXP (P(3,0,0)) *  SIN (P(3,0,1))
2620 P(3,0,0) = P(3,1,0):P(3,0,1) = P(3,1,1)
2630  GOSUB 1800
2640  RETURN
2700  IF P(3,1,0) = 0 THEN  PRINT "IMPOSSIBLE": GOTO 274
   0
2710 P(3,0,0) =  LOG (P(3,1,0))
2720 P(3,0,1) = P(3,1,1)
2730  GOSUB 1800
2740  RETURN
2800 P(3,0,0) = P(3,1,0) * P(3,1,0) *  COS (P(3,1,1) * 2
   )
2810 P(3,0,1) = P(3,1,0) * P(3,1,0) *  SIN (P(3,1,1) * 2
   )
2820  GOSUB 1800
2830  RETURN
2900 P(3,0,0) =  SQR (P(3,1,0)) *  COS (P(3,1,1) / 2)
2910 P(3,0,1) =  SQR (P(3,1,0)) *  SIN (P(3,1,1) / 2)
2920  GOSUB 1800
2930  RETURN
3000  IF P(3,1,0) = 0 THEN  PRINT "IMPOSSIBLE": GOTO 304
   0
3010 P(3,1,0) = 1 / P(3,1,0)
3020 P(3,1,1) =  - P(3,1,1)
3030  GOSUB 1900
3040  RETURN
3100  IF P(2,1,0) = 0 GOTO 3180
3110  FOR I = 0 TO 1: FOR J = 0 TO 1
```

```
3120 IN(I,J) = P(3,I,J):P(3,I,J) = P(2,I,J):P(2,I,J) = I
    N(I,J)
3130  NEXT J,I
3140  GOSUB 2700
3150  GOSUB 2400
3160  GOSUB 2600
3170  FOR I = 0 TO 1: FOR J = 0 TO 1: FOR K = 0 TO 1:P(2
     - I,J,K) = P(1 - I,J,K): NEXT K,J,I
3180  RETURN
3200 U = P(3,0,0):P(3,0,0) =  - P(3,0,1):P(3,0,1) = U
3210  GOSUB 3900
3220 U = P(3,0,1):P(3,0,1) =  - P(3,0,0):P(3,0,0) = U
3230  GOSUB 1800
3240  RETURN
3300 U = P(3,0,0):P(3,0,0) =  - P(3,0,1):P(3,0,1) = U
3310  GOSUB 3800
3320  RETURN
3400 IN(0,0) = P(3,0,0):IN(0,1) = P(3,0,1): GOSUB 3200
3410 IN(1,0) = P(3,1,0):IN(1,1) = P(3,1,1)
3420 P(3,0,0) = IN(0,0):P(3,0,1) = IN(0,1)
3430  GOSUB 3300
3440  IF P(3,1,0) = 0 GOTO 3490
3450 P(3,1,0) = IN(1,0) / P(3,1,0):P(3,1,1) = IN(1,1) -
    P(3,1,1)
3460  IF P(3,1,1) <  - PI THEN P(3,1,1) = P(3,1,1) + 2 *
    PI
3470  IF P(3,1,1) > PI THEN P(3,1,1) = P(3,1,1) - 2 * PI

3480  GOSUB 1900
3490  RETURN
3500 IN(0,0) = P(3,0,0):IN(0,1) = P(3,0,1)
3505  GOSUB 2800
3510 P(3,0,0) = 1 - P(3,0,0):P(3,0,1) =  - P(3,0,1)
3515  GOSUB 1800
3520  GOSUB 2900
3525 IN(1,0) = P(3,0,0):IN(1,1) = P(3,0,1)
3530 P(3,0,0) = IN(0,0) - IN(1,1):P(3,0,1) = IN(0,1) + I
    N(1,0)
3535  GOSUB 1800
3540  GOSUB 2700
3545 IN(0,0) = P(3,0,1):P(3,0,1) =  - P(3,0,0):P(3,0,0) =
    IN(0,0)
3550  GOSUB 1800
3555  RETURN
3600 IN(0,0) = P(3,0,0):IN(0,1) = P(3,0,1)
```

```
3610  GOSUB 2800
3620 P(3,0,0) = 1 - P(3,0,0):P(3,0,1) =  - P(3,0,1)
3630  GOSUB 1800: GOSUB 2900
3640 P(3,0,0) = P(3,0,0) - IN(0,1):P(3,0,1) = P(3,0,1) +
    IN(0,0)
3650  GOSUB 1800: GOSUB 2700
3660 IN(0,0) = P(3,0,1):P(3,0,1) =  - P(3,0,0):P(3,0,0) =
    IN(0,0)
3670  GOSUB 1800
3680  RETURN
3700  IF (P(3,0,0) = 0 AND P(3,0,1) = 1) OR (P(3,0,0) =
    0 AND P(3,0,1) =  - 1) GOTO 3790
3710 IN(0,0) = P(3,0,0):IN(0,1) = P(3,0,1)
3720 P(3,0,0) =  - P(3,0,0):P(3,0,1) = 1 - P(3,0,1): GOSUB
    1800
3730 IN(1,0) = P(3,1,0):IN(1,1) = P(3,1,1)
3740 P(3,0,0) = IN(0,0):P(3,0,1) = 1 + IN(0,1): GOSUB 18
    00
3750 P(3,1,0) = IN(1,0) / P(3,1,0):P(3,1,1) = IN(1,1) -
    P(3,1,1)
3760  GOSUB 2700
3770 U = P(3,0,1) / 2:P(3,0,1) =  - P(3,0,0) / 2:P(3,0,0
    ) = U
3780  GOSUB 1800
3790  RETURN
3800  GOSUB 4600
3810 P(3,0,0) = (EP(0) + EN(0)) / 2
3820 P(3,0,1) = (EP(1) + EN(1)) / 2
3830  GOSUB 1800
3840  RETURN
3900  GOSUB 4600
3910 P(3,0,0) = (EP(0) - EN(0)) / 2
3920 P(3,0,1) = (EP(1) - EN(1)) / 2
3930  GOSUB 1800
3940  RETURN
4000 IN(0,0) = P(3,0,0):IN(0,1) = P(3,0,1): GOSUB 3900
4010 IN(1,0) = P(3,1,0):IN(1,1) = P(3,1,1)
4020 P(3,0,0) = IN(0,0):P(3,0,1) = IN(0,1): GOSUB 3800
4030  IF P(3,1,0) = 0 GOTO 4080
4040 P(3,1,0) = IN(1,0) / P(3,1,0):P(3,1,1) = IN(1,1) -
    P(3,1,1)
4050  IF P(3,1,1) <  - PI THEN P(3,1,1) = P(3,1,1) + 2 *
    PI
4060  IF P(3,1,1) > PI THEN P(3,1,1) = P(3,1,1) - 2 * PI
```

```
4070  GOSUB 1900
4080  RETURN
4100 IN(0,0) = P(3,0,0):IN(0,1) = P(3,0,1)
4110  GOSUB 2800
4120 P(3,0,0) = P(3,0,0) - 1
4130  GOSUB 1800
4140  GOSUB 2900
4150 P(3,0,0) = P(3,0,0) + IN(0,0):P(3,0,1) = P(3,0,1) +
    IN(0,1)
4160  GOSUB 1800
4170  GOSUB 2700
4180  RETURN
4200 IN(0,0) = P(3,0,0):IN(0,1) = P(3,0,1)
4210  GOSUB 2800
4220 P(3,0,0) = P(3,0,0) + 1
4230  GOSUB 1800
4240  GOSUB 2900
4250 P(3,0,0) = P(3,0,0) + IN(0,0):P(3,0,1) = P(3,0,1) +
    IN(0,1)
4260  GOSUB 1800
4270  GOSUB 2700
4280  RETURN
4300  IF (P(3,0,0) = 1 AND P(3,0,1) = 0) OR (P(3,0,0) =
     - 1 AND P(3,0,1) = 0) GOTO 4380
4310 IN(0,0) = P(3,0,0):IN(0,1) = P(3,0,1)
4320 P(3,0,0) = P(3,0,0) + 1: GOSUB 1800
4330 IN(1,0) = P(3,1,0):IN(1,1) = P(3,1,1)
4340 P(3,0,0) = 1 - IN(0,0):P(3,0,1) =  - IN(0,1): GOSUB
    1800
4350 P(3,1,0) = IN(1,0) / P(3,1,0):P(3,1,1) = IN(1,1) -
    P(3,1,1)
4360  GOSUB 2700
4370 P(3,0,0) = P(3,0,0) / 2:P(3,0,1) = P(3,0,1) / 2: GOSUB
    1800
4380  RETURN
4400 ME$ =  MID$ (OP$,4,1)
4410  IF ME$ < "0" OR ME$ > "9" GOTO 4440
4420 MM =  VAL (ME$)
4430  FOR I = 0 TO 1:M(MM,I) = P(3,0,I): NEXT I
4440  RETURN
4500 ME$ =  MID$ (OP$,4,1)
4510  IF ME$ < "0" OR ME$ > "9" GOTO 4550
4520 MM =  VAL (ME$)
4530  FOR I = 1 TO 3: FOR J = 0 TO 1: FOR K = 0 TO 1:P(I
     - 1,J,K) = P(I,J,K): NEXT K,J,I
```

```
4540  FOR I = 0 TO 1:P(3,0,I) = M(MM,I): NEXT I
4550  GOSUB 1800
4560  RETURN
4600 EP(0) =  EXP (P(3,0,0)) *  COS (P(3,0,1))
4610 EP(1) =  EXP (P(3,0,0)) *  SIN (P(3,0,1))
4620 EN(0) =  EXP ( - P(3,0,0)) *  COS (P(3,0,1))
4630 EN(1) =  -  EXP ( - P(3,0,0)) *  SIN (P(3,0,1))
4640  RETURN
4700  FOR I = 0 TO 2: FOR J = 0 TO 1: FOR K = 0 TO 1:P(3
     - I,J,K) = P(2 - I,J,K): NEXT K,J,I
4710  RETURN
4800  FOR I = 0 TO 3: FOR J = 0 TO 1: FOR K = 0 TO 1:P(I
    ,J,K) = 0: NEXT K,J,I
4810  RETURN
4900  FOR I = 0 TO 9: FOR J = 0 TO 1:M(I,J) = 0: NEXT J,
    I
4910  RETURN
4950  FOR J = 0 TO 1: FOR K = 0 TO 1
4960 IN(J,K) = P(0,J,K): FOR I = 0 TO 2:P(I,J,K) = P(I +
    1,J,K): NEXT I:P(3,J,K) = IN(J,K)
4970  NEXT K,J
4980  RETURN
```

d. Exemples commentés

● ***Exemple 1*** :

Soit à effectuer le calcul suivant :

$$z = \sqrt{(5 - i) . \ln (3 + 2i)}$$

que nous écrivons avec la notation informatique :

z = SQR ((5 − i) * LN (3 + 2i))

Introduisons dans l'ordre les deux nombres (5 − i) et (3 + 2i) :

```
T(ALG):0                 + I 0
T(POL):0             EXP + I 0

Z(ALG):0                 + I 0
Z(POL):0             EXP + I 0
```

```
Y(ALG):5                - I 1
Y(POL):5.09901951 EXP - I .19739556

X(ALG):3                + I 2
X(POL):3.60555128 EXP + I .588002604
```

(5 − i) est donc placé dans Y et (3 + 2i) dans X; entrons alors LN pour calculer le logarithme de (3 + 2i):

```
T(ALG):0                + I 0
T(POL):0          EXP + I 0

Z(ALG):0                + I 0
Z(POL):0          EXP + I 0

Y(ALG):5                - I 1
Y(POL):5.09901951 EXP - I .19739556

X(ALG):1.28247468       + I .588002604
X(POL):1.41084668 EXP + I .429892248
```

Effectuons alors le produit (5 − i) * LN (3 + 2i) en entrant "*":

```
T(ALG):0                + I 0
T(POL):0          EXP + I 0

Z(ALG):0                + I 0
Z(POL):0          EXP + I 0

Y(ALG):0                + I 0
Y(POL):0          EXP + I 0

X(ALG):7.000376         + I 1.65753834
X(POL):7.19393478 EXP + I .232496688
```

Enfin, prenons en la racine carrée en entrant"SQR":

```
T(ALG):0                + I 0
T(POL):0          EXP + I 0

Z(ALG):0                + I 0
Z(POL):0          EXP + I 0
```

```
Y(ALG):0                  + I 0
Y(POL):0              EXP + I 0

X(ALG):2.66404868         + I .311093853
X(POL):2.68215115 EXP + I .116248344
```

Nous avons sauvé dans les mémoires les résultats intermédiaires :

```
M0:  5               - I 1
M1:  3               + I 2
M2:  1.28247468      + I .588002604
M3:  7.000376        + I 1.65753834
M4:  2.66404868      + I .311093853
M5:  0               + I 0
M6:  0               + I 0
M7:  0               + I 0
M8:  0               + I 0
M9:  0               + I 0
```

Nous allons pouvoir les représenter graphiquement dans le plan complexe grâce au programme que nous présenterons dans le cinquième paragraphe :



- ***Exemple 2*** :

Effectuons le calcul de :

$z = \text{SIN}\ (2 + 3i)$

Introduisons (2 + 3i):

```
T(ALG):0              + I 0
T(POL):0          EXP + I 0

Z(ALG):0              + I 0
Z(POL):0          EXP + I 0

Y(ALG):0              + I 0
Y(POL):0          EXP + I 0

X(ALG):2              + I 3
X(POL):3.60555128 EXP + I .982793723
```

Prenons en le sinus:

```
T(ALG):0              + I 0
T(POL):0          EXP + I 0

Z(ALG):0              + I 0
Z(POL):0          EXP + I 0

Y(ALG):0              + I 0
Y(POL):0          EXP + I 0

X(ALG):9.15449916     - I 4.16890696
X(POL):10.0590576 EXP - I .427330784
```

Effectuons maintenant la fonction inverse, c'est-à-dire ASIN (Arcsinus):

```
T(ALG):0              + I 0
T(POL):0          EXP + I 0

Z(ALG):0              + I 0
Z(POL):0          EXP + I 0

Y(ALG):0              + I 0
Y(POL):0          EXP + I 0

X(ALG):1.14159265     - I 3
X(POL):3.20986507 EXP - I 1.2071855
```

Le résultat obtenu ne correspond pas à la première valeur (2 + 3i) ; ceci provient du fait que plusieurs nombres complexes possèdent le même sinus, et que la fonction Arcsinus ne renvoie qu'une seule détermination.

On pourra vérifier que :

SIN (1.14159265 − 3i) = 9.15449916 − 4.16890696i

Représentons ces valeurs dans le plan complexe :

```
M0:  2              + I 3
M1:  9.15449916     - I 4.16890696
M2:  1.14159265     - I 3
M3:  0              + I 0
M4:  0              + I 0
M5:  0              + I 0
M6:  0              + I 0
M7:  0              + I 0
M8:  0              + I 0
M9:  0              + I 0
```



● ***Exemple 3*** :

Effectuons enfin le calcul :

$$z = ((2 + 3i).(1 - i) - \frac{3e^{5i}}{\sqrt{2 - i}})^{(1/2 - 2i)}$$

Entrons tout d'abord les deux premiers nombres et effectuons leur produit :

```
T(ALG):0                + I 0
T(POL):0            EXP + I 0

Z(ALG):0                + I 0
Z(POL):0            EXP + I 0

Y(ALG):2                + I 3
Y(POL):3.60555128 EXP + I .982793723

X(ALG):1                - I 1
X(POL):1.41421356 EXP - I .785398163
```

```
T(ALG):0                + I 0
T(POL):0            EXP + I 0

Z(ALG):0                + I 0
Z(POL):0            EXP + I 0

Y(ALG):0                + I 0
Y(POL):0            EXP + I 0

X(ALG):5                + I .999999999
X(POL):5.09901951 EXP + I .19739556
```

Introduisons maintenant $3e^{5i}$ et $(2 - i)$:

```
T(ALG):0                + I 0
T(POL):0            EXP + I 0

Z(ALG):5                + I .999999999
Z(POL):5.09901951 EXP + I .19739556

Y(ALG):.850986557       - I 2.87677283
Y(POL):3            EXP + I 5

X(ALG):2                - I 1
X(POL):2.23606798 EXP - I .463647609
```

Calculons $\sqrt{2 - i}$:

```
T(ALG):0                + I 0
T(POL):0            EXP + I 0
```

```
Z(ALG):5                + I .999999999
Z(POL):5.09901951 EXP + I .19739556

Y(ALG):.850986557       - I 2.87677283
Y(POL):3          EXP + I 5

X(ALG):1.45534669       - I .34356075
X(POL):1.49534878 EXP - I .231823805
```

Divisons $3e^{5i}$ par $\sqrt{2-i}$:

```
T(ALG):0                + I 0
T(POL):0          EXP + I 0

Z(ALG):0                + I 0
Z(POL):0          EXP + I 0

Y(ALG):5                + I .999999999
Y(POL):5.09901951 EXP + I .19739556

X(ALG):.995867172       - I 1.74160011
X(POL):2.00622092 EXP - I 1.0513615
```

Retranchons ce nombre à $(2+3i)(1-i)$:

```
T(ALG):0                + I 0
T(POL):0          EXP + I 0

Z(ALG):0                + I 0
Z(POL):0          EXP + I 0

Y(ALG):0                + I 0
Y(POL):0          EXP + I 0

X(ALG):4.00413283       + I 2.74160011
X(POL):4.85277764 EXP + I .600378494
```

Introduisons l'exposant:

```
T(ALG):0                + I 0
T(POL):0          EXP + I 0

Z(ALG):0                + I 0
Z(POL):0          EXP + I 0
```

```
Y(ALG):4.00413283     + I 2.74160011
Y(POL):4.85277764 EXP + I .600378494

X(ALG):.5             - I 2
X(POL):2.06155281 EXP - I 1.32581766
```

Et élevons à la puissance:

```
T(ALG):0              + I 0
T(POL):0          EXP + I 0

Z(ALG):0              + I 0
Z(POL):0          EXP + I 0

Y(ALG):0              + I 0
Y(POL):0          EXP + I 0

X(ALG):-7.02893407    - I 2.04160675
X(POL):7.31943115 EXP - I 2.85891326
```

Nous avons sauvé les résultats intermédiaires dans les mémoires:

```
M0:  2              + I 3
M1:  1              - I 1
M2:  5              + I .999999999
M3:  .850986557     - I 2.87677283
M4:  2              - I 1
M5:  1.45534669     - I .34356075
M6:  .995867172     - I 1.74160011
M7:  4.00413283     + I 2.74160011
M8:  .5             - I 2
M9:  -7.02893407    - I 2.04160675
```

Ce qui nous permet de les représenter dans le plan complexe:



3. *Système linéaire de N équations à N inconnues*

a. Principe

On appelle système linéaire de N équations à N inconnues un ensemble de N équations de la forme :

$$\begin{cases} a_{11} z_1 + a_{12} z_2 + \dots + a_{1N} z_N = b_1 \\ a_{21} z_1 + a_{22} z_2 + \dots + a_{2N} z_N = b_2 \\ \dots\dots\dots\dots\dots\dots\dots\dots\dots\dots \\ a_{N1} z_1 + a_{N2} z_2 + \dots + a_{NN} z_N = b_N \end{cases}$$

où $z_1, \dots z_N$ sont les N nombres complexes inconnus, et a_{ij} ($i \in [1,N]$, $j \in [1,N]$), b_i ($i \in [1,N]$) des nombres complexes donnés.

Il est possible de représenter ce système sous une forme matricielle :

$$\begin{pmatrix} a_{11} & a_{12} & \dots & a_{1N} \\ a_{21} & a_{22} & \dots & a_{2N} \\ \vdots & \vdots & & \vdots \\ a_{N1} & a_{N2} & \dots & a_{NN} \end{pmatrix} \begin{pmatrix} z_1 \\ z_2 \\ \vdots \\ z_N \end{pmatrix} = \begin{pmatrix} b_1 \\ b_2 \\ \vdots \\ b_N \end{pmatrix}$$

Posons alors :

$$z_k = x_k + iy_k$$

$$a_{kl} = c_{kl} + id_{kl}$$

$$b_k = e_k + if_k$$

où x_k, y_k, c_{kl}, d_{kl}, e_k et f_k sont des réels.

Le système complexe est alors équivalent au système réel suivant, de dimension 2N :

$$\begin{pmatrix} c_{11} & -d_{11} & c_{12} & -d_{12} & \dots & c_{1N} & -d_{1N} \\ d_{11} & c_{11} & d_{12} & c_{12} & \dots & d_{1N} & c_{1N} \\ c_{21} & -d_{21} & c_{22} & -d_{22} & \dots & c_{2N} & -d_{2N} \\ d_{21} & c_{21} & d_{22} & c_{22} & \dots & d_{2N} & c_{2N} \\ \vdots & \vdots & \vdots & \vdots & & \vdots & \vdots \\ c_{N1} & -d_{N1} & c_{N2} & -d_{N2} & \dots & c_{NN} & -d_{NN} \\ d_{N1} & c_{N1} & d_{N2} & c_{N2} & \dots & d_{NN} & c_{NN} \end{pmatrix} \begin{pmatrix} x_1 \\ y_1 \\ x_2 \\ y_2 \\ \vdots \\ x_N \\ y_N \end{pmatrix} = \begin{pmatrix} e_1 \\ f_1 \\ e_2 \\ f_2 \\ \vdots \\ e_N \\ f_N \end{pmatrix}$$

ce système à coefficients réels est alors résolu par la méthode de Gauss-Jordan, que nous expliquerons dans le chapitre "Matrices et vecteurs".

b. Programme

Après avoir précisé la dimension N du système (qui doit être inférieure ou égale à 10), il faut introduire les N^2 cóefficient a_{kl}, puis les N coordonnées b_k du second membre.

Si le déterminant du système est nul, celui-ci est dégénéré et n'admet pas de solution unique.

Si le déterminant n'est pas nul, le programme affiche alors l'unique N uplet solution, et place ces N valeurs dans les N premières mémoires, ce qui permet leur rappel ou leur utilisation ultérieure par le programme précédent.

```
5000 CLS
5010  PRINT  TAB( 5);"SYSTEME DE N EQUATIONS": PRINT : PRINT

5020  INPUT "DIMENSION DU SYSTEME ";MM:MM =  INT ( ABS (
    MM))
5030  IF MM > MS THEN  PRINT "N DOIT ETRE EGAL AU PLUS A
     ";MS: GOTO 5020
5040  PRINT "INTRODUISEZ LES COEFFICIENTS A :"
5050  FOR I = 1 TO MM
5060  FOR J = 1 TO MM
5070  PRINT "A (";I;",";J;")"
5080  INPUT " PARTIE REELLE ";D(2 * I - 1,2 * J - 1)
5090  INPUT " PARTIE IMAGINAIRE ";D(2 * I,2 * J - 1)
5100 D(2 * I,2 * J) = D(2 * I - 1,2 * J - 1)
5110 D(2 * I - 1,2 * J) =  - D(2 * I,2 * J - 1)
5120  NEXT J
5130  NEXT I
5140  PRINT "INTRODUISEZ LES COEFFICIENTS B :"
5150  FOR I = 1 TO MM
5160  PRINT "B (";I;;")"
5170  INPUT " PARTIE REELLE ";D(2 * I - 1,2 * MM + 1)
5180  INPUT " PARTIE IMAGINAIRE ";D(2 * I,2 * MM + 1)
5190  NEXT I
5200 N = 2 * MM:PP = 1
5210 DR = 1:RG = N
5220  FOR K = 1 TO N
5230 I = K
5240  IF I > N THEN DR = 0:RG = RG - 1: GOTO 5320
5250  IF  ABS (D(I,K)) < PR THEN I = I + 1:DR =  - DR: GOTO
    5240
5260  IF I > K THEN  FOR J = 1 TO N + PP:U = D(I,J):D(I,
    J) = D(K,J):D(K,J) = U: NEXT J
```

```
5270  FOR I = K + 1 TO N + PP:D(K,I) = D(K,I) / D(K,K): NEXT
   I
5280 DR = DR * D(K,K):D(K,K) = 1
5290  FOR I = 1 TO N
5300  IF I <  > K THEN  FOR J = K + 1 TO N + PP:D(I,J) =
    D(I,J) - D(I,K) * D(K,J): NEXT J:D(I,K) = 0
5310  NEXT I
5320  NEXT K
5330  IF DR = 0 THEN  PRINT  TAB( 8);"LE SYSTEME N'ADMET
     PAS": PRINT  TAB( 10);"DE SOLUTION UNIQUE": GOTO 54
    40
5340  FOR I = 0 TO MM - 1
5350 M(I,0) = D(2 * I + 1,N + 1):M(I,1) = D(2 * I + 2,N +
    1)
5360  NEXT I
5370  PRINT : PRINT "LES SOLUTIONS SONT :"
5380  FOR I = 0 TO MM - 1
5390  PRINT "Z";I + 1;"= ";M(I,0);
5400  IF M(I,1) < 0 THEN  PRINT "  - ";: GOTO 5420
5410  PRINT "  + ";
5420  PRINT  ABS (M(I,1));" I"
5430  NEXT I
5440  PRINT : INPUT "VOULEZ-VOUS REUTILISER LE PROGRAMME
     ";Z$
5450  IF Z$ = "O" GOTO 5000
5460  GOTO 100

§PR£0
```

c. Exemples commentés

● ***Exemple 1*** :

Considérons le système d'ordre 3 suivant :

$$\begin{cases} z_1 + z_2 + iz_3 = 5 \\ z_1 + iz_2 + z_3 = -2i \\ iz_1 + z_2 + z_3 = 1 \end{cases}$$

SYSTEME DE N EQUATIONS

```
DIMENSION DU SYSTEME 3
INTRODUISEZ LES COEFFICIENTS A :
A (1,1)
 PARTIE REELLE 1
 PARTIE IMAGINAIRE 0
A (1,2)
 PARTIE REELLE 1
 PARTIE IMAGINAIRE 0
A (1,3)
 PARTIE REELLE 0
 PARTIE IMAGINAIRE 1
A (2,1)
 PARTIE REELLE 1
 PARTIE IMAGINAIRE 0
A (2,2)
 PARTIE REELLE 0
 PARTIE IMAGINAIRE 1
A (2,3)
 PARTIE REELLE 1
 PARTIE IMAGINAIRE 0
A (3,1)
 PARTIE REELLE 0
 PARTIE IMAGINAIRE 1
A (3,2)
 PARTIE REELLE 1
 PARTIE IMAGINAIRE 0
A (3,3)
 PARTIE REELLE 1
 PARTIE IMAGINAIRE 0
INTRODUISEZ LES COEFFICIENTS B :
B (1)
 PARTIE REELLE 5
 PARTIE IMAGINAIRE 0
B (2)
 PARTIE REELLE 0
 PARTIE IMAGINAIRE -2
B (3)
 PARTIE REELLE 1
 PARTIE IMAGINAIRE 0

LES SOLUTIONS SONT :
Z1= 1.5  - .5 I
```

```
Z2= 1   + 1 I
Z3= -.5   - 2.5 I

VOULEZ-VOUS REUTILISER LE PROGRAMME N
```

Le programme fournit alors le triplet solution :

$$\begin{cases} z_1 = \dfrac{3}{2} - \dfrac{1}{2}i \\ z_2 = 1 + i \\ z_3 = -\dfrac{1}{2} - \dfrac{5}{2}i \end{cases}$$

rangé dans les mémoires M0, M1 et M2, ce qui permet l'utilisation du programme de représentation dans le plan complexe :



● ***Exemple 2*** :

Soit le système à deux inconnues :

$$(1 - 5i)\,z_1 + (8 + 2i)\,z_2 = 3i$$
$$-3\,z_1 + (6 + 2i)\,z_2 = -7$$

Le programme fournit alors la solution :

```
    SYSTEME DE N EQUATIONS


DIMENSION DU SYSTEME 2
INTRODUISEZ LES COEFFICIENTS A :
A (1,1)
 PARTIE REELLE 1
 PARTIE IMAGINAIRE -5
```

```
A (1,2)
 PARTIE REELLE 8
 PARTIE IMAGINAIRE 2
A (2,1)
 PARTIE REELLE -3
 PARTIE IMAGINAIRE 0
A (2,2)
 PARTIE REELLE 6
 PARTIE IMAGINAIRE 2
INTRODUISEZ LES COEFFICIENTS B :
B (1)
 PARTIE REELLE 0
 PARTIE IMAGINAIRE 3
B (2)
 PARTIE REELLE -7
 PARTIE IMAGINAIRE 0

LES SOLUTIONS SONT :
Z1= .621880997  + 1.14203455 I
Z2= -.598848369  + .770633397 I

VOULEZ-VOUS REUTILISER LE PROGRAMME N
```



4. *Equations du second degré*

a. Principe

Une équation du second degré à coefficients complexes est une équation du type :

$$az^2 + bz + c = 0 \qquad \text{avec } a \neq 0$$

où les nombres a, b et c sont des complexes.

Contrairement au cas réel, cette équation admet toujours deux solutions (distinctes ou non), données par :

$$z' = \frac{-b - \sqrt{b^2 - 4ac}}{2a}$$

$$z'' = \frac{-b + \sqrt{b^2 - 4ac}}{2a}$$

b. Programme

Il faut tout d'abord introduire les valeurs des coefficients a, b et c, en observant bien la condition $a \neq 0$.

Le programme calcule et affiche les deux racines, et place les résultats dans les mémoires M8 et M9, de manière à permettre leur représentation dans le plan complexe et leur utilisation dans le programme calculatrice.

```
6000 CLS
6010  PRINT  TAB( 5);"EQUATION DU SECOND DEGRE": PRINT :
     PRINT
6020  PRINT "INTRODUISEZ LES COEFFICIENTS :"
6030  PRINT " COEFFICIENT A :"
6040  INPUT "  PARTIE REELLE ";A1
6050  INPUT "  PARTIE IMAGINAIRE ";A2
6060  IF A1 = 0 AND A2 = 0 THEN  PRINT "A DOIT ETRE NON
    NUL": GOTO 6030
6070  PRINT " COEFFICIENT B :"
6080  INPUT "  PARTIE REELLE ";B1
6090  INPUT "  PARTIE IMAGINAIRE ";B2
6100  PRINT " COEFFICIENT C :"
```

```
6110  INPUT "  PARTIE REELLE ";C1
6120  INPUT "  PARTIE IMAGINAIRE ";C2
6130 D1 = B1 * B1 - B2 * B2 - 4 * A1 * C1 + 4 * A2 * C2
6140 D2 = 2 * B1 * B2 - 4 * A1 * C2 - 4 * A2 * C1
6150 H =  SQR (D1 * D1 + D2 * D2)
6160  IF D1 = 0 THEN H =  ABS (D2)
6170  IF D2 = 0 THEN H =  ABS (D1)
6180 E1 =  SQR ((H + D1) / 2)
6190 E2 =  SQR ((H - D1) / 2): IF D2 < 0 THEN E2 =  - E2

6200 F1 =  - (B1 + E1) / 2
6210 F2 =  - (B2 + E2) / 2
6220 G1 = (E1 - B1) / 2
6230 G2 = (E2 - B2) / 2
6240 H = A1 * A1 + A2 * A2
6250 H1 = (F1 * A1 + F2 * A2) / H:M(8,0) = H1
6260 H2 = (F2 * A1 - F1 * A2) / H:M(8,1) = H2
6270 I1 = (G1 * A1 + G2 * A2) / H:M(9,0) = I1
6280 I2 = (G2 * A1 - G1 * A2) / H:M(9,1) = I2
6290  PRINT : PRINT "LES SOLUTIONS SONT :"
6300  PRINT "Z' = ";: IF H1 <  > 0 OR H2 = 0 THEN  PRINT
    H1;
6310  IF H2 = 0 THEN  PRINT : GOTO 6340
6320  IF H2 > 0 THEN  PRINT " + "; ABS (H2);" I"
6330  IF H2 < 0 THEN  PRINT " - "; ABS (H2);" I"
6340  PRINT "Z''= ";: IF I1 <  > 0 OR I2 = 0 THEN  PRINT
    I1;
6350  IF I2 = 0 THEN  PRINT : GOTO 6380
6360  IF I2 > 0 THEN  PRINT " + "; ABS (I2);" I"
6370  IF I2 < 0 THEN  PRINT " - "; ABS (I2);" I"
6380  PRINT : INPUT "VOULEZ-VOUS REUTILISER LE PROGRAMME
     ";Z$
6390  IF Z$ = "O" GOTO 6000
6400  GOTO 100
```

c. Exemples commentés

- ***Exemple 1*** :

Soit l'équation :

$$(-4 + i)\,z^2 + (2 - 3i)\,z + (5 - i) = 0$$

Le programme fournit les solutions suivantes:

```
    EQUATION DU SECOND DEGRE


INTRODUISEZ LES COEFFICIENTS :
 COEFFICIENT A :
  PARTIE REELLE -4
  PARTIE IMAGINAIRE 1
 COEFFICIENT B :
  PARTIE REELLE 2
  PARTIE IMAGINAIRE -3
 COEFFICIENT C :
  PARTIE REELLE 5
  PARTIE IMAGINAIRE -1

LES SOLUTIONS SONT :
Z' = 1.4446445 - .352759213 I
Z''= -.797585677 - .235476081 I

VOULEZ-VOUS REUTILISER LE PROGRAMME N
```



● ***Exemple 2*** :

Considérons l'équation suivante, dont les solutions sont les trois racines cubiques de l'unité:

$z^3 = 1$

et qui peut être factorisée sous la forme :

$$(z-1)(z^2+z+1)=0$$

La valeur 1 est bien entendu solution, et les deux autres racines sont solutions de l'équation :

$$z^2+z+1=0$$

C'est une équation à coefficients réels, donc ses solutions sont conjuguées, c'est-à-dire que si $z'=a+ib$ est solution, alors $z''=z'=a-ib$ est également solution.

Les racines exactes sont :

$$z'=-\frac{1}{2}-\frac{\sqrt{3}}{2}i$$

$$z''=-\frac{1}{2}+\frac{\sqrt{3}}{2}i$$

Le programme les obtient avec la précision maximale :

```
     EQUATION DU SECOND DEGRE


INTRODUISEZ LES COEFFICIENTS :
 COEFFICIENT A :
  PARTIE REELLE 1
  PARTIE IMAGINAIRE 0
 COEFFICIENT B :
  PARTIE REELLE 1
  PARTIE IMAGINAIRE 0
 COEFFICIENT C :
  PARTIE REELLE 1
  PARTIE IMAGINAIRE 0

LES SOLUTIONS SONT :
Z' = -.5 - .866025404 I
Z''= -.5 + .866025404 I

VOULEZ-VOUS REUTILISER LE PROGRAMME N
```

On remarque sur la représentation graphique que les trois solutions forment un triangle équilatéral, ce qui est conforme à la théorie, qui prévoit que les N racines

Nième de l'unité forment un polygône régulier circonscrit au cercle de rayon unité.



5. *Représentation dans le plan complexe*

a. Principe

Le corps des complexes constituant un espace vectoriel de dimension 2, il est possible d'établir une correspondance entre C et le plan affine ; un point M de coordonnées (x,y) du plan est alors associé au nombre complexe $z = x + iy$ que l'on appelle affixe de M.



L'axe OX est appelé axe des réels, l'axe OY axe des imaginaires purs; la partie réelle de z correspond donc à l'abscisse du point M, la partie imaginaire à l'ordonnée.

Le module de z est la distance entre le point M et l'origine O, l'argument de z, l'angle défini par la demi-droite OX et le vecteur $\overrightarrow{OM}$.

b. Programme

Le programme affiche en haute résolution les dix points dont les affixes sont dans les mémoires M0 à M9.

Il permet donc entre autres de représenter les solutions d'un système linéaire ou d'une équation du second degré, puisque les programmes correspondants placent leurs solutions dans les mémoires.

```
7000 CLS
7010 XN = 0:XM = 0:YN = 0:YM = 0
7020  FOR I = 0 TO 9
7030  IF M(I,0) < XN THEN XN = M(I,0)
7040  IF M(I,0) > XM THEN XM = M(I,0)
7050  IF M(I,1) < YN THEN YN = M(I,1)
7060  IF M(I,1) > YM THEN YM = M(I,1)
7070  NEXT I
7080 XX = XM: IF  - XN > XX THEN XX =  - XN
7090 YY = YM: IF  - YN > YY THEN YY =  - YN
7100  IF YY = 0 THEN YY = 1
7110 HIRES
7120 LG = YY * C / L
7130  IF XX > LG THEN LG = XX
7140 PX = 10 ^ ( INT ( LOG (LG / 2) /  LOG (10)))
7150 LG =  INT (LG / PX + 1.5) * PX
7160 HX =  INT (PX / 2 / LG * (C - 4))
7170 X1 =  INT (C / 2):Y1 =  INT (L / 2)
7180 LINE(0,Y1) - (C,Y1):LINE(X1,0) - (X1,L)
7190 X = X1:K = 0
7200 X = X + HX:K = K + 1
7210  IF X > C THEN HX =  - HX: GOTO 7190
7220  IF X < 0 GOTO 7260
7230 LINE(X,Y1 - 2) - (X,Y1 + 2)
7240  IF K = 10 *  INT (K / 10) THEN LINE(X,Y1 - 4) - (X
    ,Y1 + 4)
7250  GOTO 7200
7260 Y = Y1:K = 0
```

```
7270 Y = Y + HX:K = K + 1
7280  IF Y < 0 THEN HX =  - HX: GOTO 7260
7290  IF Y > L GOTO 7330
7300 LINE(X1 - 2,Y) - (X1 + 2,Y)
7310  IF K = 10 *  INT (K / 10) THEN LINE(X1 - 4,Y) - (X
    1 + 4,Y)
7320  GOTO 7270
7330  FOR I = 0 TO 9
7340 X = X1 +  INT (M(I,0) / PX * HX)
7350 Y = Y1 -  INT (M(I,1) / PX * HX)
7360 LINE(X1,Y1) - (X,Y)
7370 LINE(X - 1,Y + 1) - (X + 1,Y + 1):LINE(X + 1,Y + 1)
     - (X + 1,Y - 1)
7380 LINE(X + 1,Y - 1) - (X - 1,Y - 1):LINE(X - 1,Y - 1)
     - (X - 1,Y + 1)
7390  NEXT I
7400  INPUT "APPUYEZ SUR RETURN POUR REVENIR AU MENU";Z$

7410  GOTO 100
```

4 Polynômes

1. Introduction

Les fonctions polynômes constituent pour l'informaticien une classe privilégiée de fonctions numériques car le traitement de calculs sur ces fonctions est relativement facile à mettre en œuvre en programmation.

Rappelons qu'une fonction polynôme s'écrit de manière générale :

$$\begin{aligned} P(x) &= a_0 + a_1 x + a_2 x^2 + \dots + a_p x^p \\ &= \sum_{i=0}^{p} a_i x^i \end{aligned}$$

où les réels a_i ($i \in [0,p]$) sont les coefficients du polynôme. Le rang du terme de plus haut degré est appelé degré du polynôme, le rang du terme de plus petit degré est appelé valuation du polynôme.

Les opérations que l'on effectue habituellement sur les polynômes ne sont pas compliquées du point de vue théorique, mais sont très fastidieuses car les calculs sont nombreux, particulièrement lorsque le degré des polynômes considérés est grand.

Les programmes que nous étudierons dans ce chapitre nous permettront de calculer la valeur en un point d'un polynôme, de multiplier ou composer deux polynômes entre eux, de diviser un polynôme par un autre selon deux méthodes différentes (division euclidienne et division selon les puissances croissantes), de déterminer le PGCD et le PPCM de deux polynômes, et enfin de calculer les dérivées successives d'un polynôme.

Avant d'introduire les programmes réalisant ces fonctions, il est nécessaire d'entrer dans l'ordinateur le menu ainsi que les sous-programmes indispensables au fonctionnement de l'ensemble.

```
10  DIM P(50),Q(50),R(50),H(50),V(50),W(50)
100 CLS
110  PRINT  TAB( 5);"CALCULS SUR LES POLYNOMES": PRINT :
     PRINT
120  PRINT "1- ENTREES/MANIPULATIONS DE POLYNOMES": PRINT

130  PRINT "2- CALCUL DE P(X0)": PRINT
140  PRINT "3- MULTIPLICATION DE P PAR Q": PRINT
150  PRINT "4- CALCUL DU POLYNOME SUBSTITUE R=P(Q)": PRINT

160  PRINT "5- DIVISION EUCLIDIENNE DE P PAR Q": PRINT
170  PRINT "6- DIVISION SUIVANT LES PUISSANCES"
180  PRINT  TAB( 5);"CROISSANTES DE P PAR Q": PRINT
190  PRINT "7- CALCUL DU PGCD ET DU PPCM DE P ET Q": PRINT

200  PRINT "8- DERIVEES SUCCESSIVES DE P": PRINT
210  PRINT "9- FIN": PRINT
220  PRINT : INPUT "VOTRE CHOIX ";E
230  ON E GOTO 1000,2000,3000,4000,5000,6000,7000,8000,9
    000
240  GOTO 100
500  IF X = 0 THEN NN = 0:DD = 1: GOTO 580
510 AX =  ABS (X):AB = AX:AC =  INT (AB):NN = AC
520 DD = 1:AD = 1:AF = 0: GOTO 560
530 AB = 1 / (AB - AC):AC =  INT (AB)
540 AY = AC * NN + AD:AD = NN:NN = AY
550 AY = AC * DD + AF:AF = DD:DD = AY
560  IF 1E6 *  ABS (AX * DD - NN) > DD * AX GOTO 530
570 NN = NN *  SGN (X)
580  RETURN
700  IF A < B THEN Y = B:B = A:A = Y
```

```
710 T = A - B *  INT (A / B)
720 A = B:B = T
730  IF T <  > 0 GOTO 710
740  RETURN
800 ER = 0:LZ =  LEN (Z$):FD = 0
810  IF LZ = 0 THEN ER = 1: GOTO 890
820  FOR K = 1 TO LZ
830  IF  MID$ (Z$,K,1) = "/" THEN FD = K:K = LZ
840  NEXT K
850  IF FD = 0 THEN N =  VAL (Z$): GOTO 890
860 B =  VAL ( MID$ (Z$,FD + 1,LZ - FD))
870  IF B = 0 THEN ER = 1: GOTO 890
880 N =  VAL ( MID$ (Z$,1,FD - 1)) / B
890  RETURN
1000 CLS
1010  PRINT  TAB( 3);"ENTREES/MANIPULATIONS DE POLYNOMES
    ": PRINT : PRINT
1020  PRINT "1- ENTREE DE P": PRINT
1030  PRINT "2- ENTREE DE Q": PRINT
1040  PRINT "3- RAPPEL DE P": PRINT
1050  PRINT "4- RAPPEL DE Q": PRINT
1060  PRINT "5- TRANSFERT DE P DANS H"
1070  PRINT "6- TRANSFERT DE H DANS P": PRINT
1080  PRINT "7- TRANSFERT DE Q DANS H"
1090  PRINT "8- TRANSFERT DE H DANS Q": PRINT
1100  PRINT "9- TRANSFERT DE R DANS H"
1110  PRINT "10-TRANSFERT DE H DANS R": PRINT
1120  PRINT "11- RETOUR AU MENU": PRINT : PRINT
1130  INPUT "VOTRE CHOIX ";E
1140  ON E GOTO 1160,1280,1400,1510,1620,1630,1640,1650,
    1660,1670,100
1150  GOTO 1000
1160 CLS
1170  PRINT  TAB( 5);"ENTREE DE P": PRINT : PRINT
1180  INPUT "DEGRE DE P ";DP: PRINT
1190  IF DP < 0 OR DP <  >  INT (DP) GOTO 1180
1200  FOR I = 0 TO DP
1210  PRINT "COEFFICIENT DE X";I;
1220  INPUT " ";Z$: GOSUB 800
1230  IF ER = 1 GOTO 1210
1240 P(I) = N
1250  IF I = DP AND DP > 0 AND P(DP) = 0 THEN  PRINT "IM
    POSSIBLE...": GOTO 1210
1260  NEXT I
1270  GOTO 1000
```

```
1280 CLS
1290  PRINT  TAB( 5);"ENTREE DE Q": PRINT : PRINT
1300  INPUT "DEGRE DE Q ";DQ: PRINT
1310  IF DQ < 0 OR DQ <  >  INT (DQ) GOTO 1300
1320  FOR I = 0 TO DQ
1330  PRINT "COEFFICIENT DE X";I;
1340  INPUT " ";Z$: GOSUB 800
1350  IF ER = 1 GOTO 1330
1360 Q(I) = N
1370  IF I = DQ AND DQ > 0 AND Q(DQ) = 0 THEN  PRINT "IM
    POSSIBLE...": GOTO 1330
1380  NEXT I
1390  GOTO 1000
1400 CLS
1410  PRINT  TAB( 5);"RAPPEL DE P": PRINT : PRINT
1420  IF DP = 0 AND P(0) = 0 THEN  PRINT "P EST NUL": GOTO
    1500
1430  FOR I = 0 TO DP
1440  IF P(I) = 0 GOTO 1490
1450  PRINT "COEFF. DE X";I;":";P(I);
1460 X = P(I): GOSUB 500
1470  IF DD <  > 1 THEN  PRINT  TAB( 29);NN;"/";DD;
1480  PRINT
1490  NEXT I
1500  PRINT : PRINT : GOTO 1690
1510 CLS
1520  PRINT  TAB( 5);"RAPPEL DE Q": PRINT : PRINT
1530  IF DQ = 0 AND Q(0) = 0 THEN  PRINT "Q EST NUL": GOTO
    1610
1540  FOR I = 0 TO DQ
1550  IF Q(I) = 0 GOTO 1600
1560  PRINT "COEFF. DE X";I;":";Q(I);
1570 X = Q(I): GOSUB 500
1580  IF DD <  > 1 THEN  PRINT  TAB( 29);NN;"/";DD;
1590  PRINT
1600  NEXT I
1610  PRINT : PRINT : GOTO 1690
1620 DH = DP: FOR I = 0 TO DH:H(I) = P(I): NEXT I: GOTO
    1680
1630 DP = DH: FOR I = 0 TO DP:P(I) = H(I): NEXT I: GOTO
    1680
1640 DH = DQ: FOR I = 0 TO DH:H(I) = Q(I): NEXT I: GOTO
    1680
1650 DQ = DH: FOR I = 0 TO DQ:Q(I) = H(I): NEXT I: GOTO
    1680
```

```
1660 DH = DR: FOR I = 0 TO DH:H(I) = R(I): NEXT I: GOTO
    1680
1670 DR = DH: FOR I = 0 TO DR:R(I) = H(I): NEXT I
1680 CLS: PRINT "TRANSFERT EFFECTUE": PRINT
1690  INPUT "APPUYEZ SUR RETURN";Z$
1700  GOTO 1000
9000  END
```

On peut ensuite introduire chacun des modules suivants indépendamment des autres.

L'ensemble nécessite environ 10 Koctets de mémoire.

REMARQUE : le degré maximal des polynômes traités est fixé à 50 à la ligne 10, ce qui est largement suffisant pour la plupart des applications. Il est bien sûr possible de le modifier ; une augmentation de 10 de ce degré maximal nécessite cependant 300 octets de mémoire supplémentaire.

2. *Entrées et manipulations de polynômes*

a. Principe

Le programme utilise quatre polynômes nommés P, Q, H et R : les opérations se font à partir des polynômes P et Q et les résultats sont placés dans H et R.

La première option du menu principal conduit à un menu secondaire dont :

— les deux premières options permettent d'entrer les polynômes P et Q sur lesquels vont être faites les opérations ;

— les deux options suivantes permettent de rappeler ces polynômes pour vérifier leur contenu.

— les options restantes permettent le transfert de P, Q ou R dans H et les transferts inverses.

Ces possibilités autorisent le chaînage d'opérations, puisque les polynômes résultats peuvent être réintroduits comme polynômes d'entrée.

b. Programme

L'utilisation de la partie "entrée" est très simple : après avoir introduit le degré p du polynôme, le programme demande chacun des coefficients, depuis le terme constant jusqu'au terme en x^p. Pour introduire un coefficient rationnel, par exemple 2/3, il n'est pas nécessaire de taper 0,6666666666 : en effet, le programme accepte 2/3 où les nombres 2 et 3 sont séparés par une barre oblique.

```
    ENTREE DE P

DEGRE DE P 3

COEFFICIENT DE X0 -1
COEFFICIENT DE X1 0
COEFFICIENT DE X2 2/3
COEFFICIENT DE X3 3.2

    RAPPEL DE P


COEFF. DE X0:-1
COEFF. DE X2:.666666667      2/3
COEFF. DE X3:3.2             16/5
```

Remarquons que les coefficients nuls n'apparaissent plus lors du rappel.

Toujours lors du rappel, le programme affiche, à côté des nombres non entiers, une "approximation rationnelle" (voir Tome 1, chapitre 5 "Dérivées - Développements limités"). Cette approximation coïncide généralement avec le nombre initial ; cependant, dans certains cas (entrée d'un nombre réel, comme $\sqrt{2}$, ou d'un rationnel compliqué, comme $\frac{35123}{10001}$), l'approximation fournie ne correspond pas au nombre initial : elle s'en approche cependant à 10^{-7} près.

```
    ENTREE DE P

DEGRE DE P 4

COEFFICIENT DE X0 -5
COEFFICIENT DE X1 5/3
COEFFICIENT DE X2 2.2
COEFFICIENT DE X3 1.414213562
COEFFICIENT DE X4 35123/10001

    RAPPEL DE P
```

```
COEFF. DE X0:-5
COEFF. DE X1:1.66666667      5/3
COEFF. DE X2:2.2             11/5
COEFF. DE X3:1.41421356      1393/985
COEFF. DE X4:3.5119488       1029/293
```

Tous les autres programmes de ce chapitre utilisent les "approximations rationnelles", et affichent à côté de chaque nombre réel un nombre rationnel s'en approchant à 10^{-7} près.

3. *Calcul de $P(x_0)$*

a. Principe

Soit le polynôme P s'écrivant :

$$P(x) = a_0 + a_1 x + a_2 x^2 + ... + a_p x^p$$

Nous voulons calculer la valeur prise par le polynôme P en un point x_0 donné ; le calcul est très simple puisque le résultat est donné par la formule :

$$P(x_0) = a_0 + a_1 x_0 + a_2 x_0^2 + ... + a_p x_0^p$$

toutefois, pour augmenter la précision et la rapidité du calcul, on préfère utiliser la formule suivante dite "formule de Horner" :

$$P(x_0) = ((...((a_p x_0 + a_{p-1}) x_0 + a_{p-2}) x_0 + ...) x_0 + a_1) x_0 + a_0$$

Dans cette formule, les seules opérations effectuées sont des additions et des multiplications, alors que la première formule fait appel à la fonction puissance. Dans un ordinateur, les élévations à la puissance sont longues et relativement imprécises : il est donc préférable de les éviter.

b. Programme

L'utilisation de ce programme est très simple ; après avoir introduit le polynôme P grâce à l'option 1 du menu, il suffit de fournir la valeur x_0 : l'ordinateur affiche alors la valeur $P(x_0)$ ainsi que l'approximation rationnelle correspondante.

```
2000 CLS
2010  PRINT  TAB( 5);"CALCUL DE P(X0)": PRINT : PRINT
2020  INPUT "X0=";Z$: GOSUB 800
2030  IF ER = 1 GOTO 2020
2040 X = N:S = 0
2050  FOR I = 0 TO DP:S = S * X + P(DP - I): NEXT I
2060  PRINT : PRINT "P(";Z$;")=";S
2070 X = S: GOSUB 500
2080  IF DD <  > 1 THEN  PRINT  TAB( 5);"= ";NN;" / ";D
    D
2090  PRINT : INPUT "VOULEZ-VOUS REUTILISER LE PROGRAMM
    E ";Z$
2100  IF Z$ = "O" GOTO 2000
2110  GOTO 100
```

c. Exemples commentés

● ***Exemple 1*** :

```
    ENTREE DE P

DEGRE DE P 3

COEFFICIENT DE X0 4
COEFFICIENT DE X1 7
COEFFICIENT DE X2 -5
COEFFICIENT DE X3 1



    CALCUL DE P(X0)

X0=0

P(0)=4

VOULEZ-VOUS REUTILISER LE PROGRAMME O

X0=1

P(1)=7
```

```
VOULEZ-VOUS REUTILISER LE PROGRAMME O

X0=10

P(10)=574

VOULEZ-VOUS REUTILISER LE PROGRAMME O

X0=-2.5

P(-2.5)=-60.375
    = -483 / 8

VOULEZ-VOUS REUTILISER LE PROGRAMME O

X0=1/3

P(1/3)=5.81481482
    = 157 / 27

VOULEZ-VOUS REUTILISER LE PROGRAMME N
```

● ***Exemple 2*** :

```
    ENTREE DE P

DEGRE DE P 16

COEFFICIENT DE X0 0
COEFFICIENT DE X1 0
COEFFICIENT DE X2 0
COEFFICIENT DE X3 0
COEFFICIENT DE X4 0
COEFFICIENT DE X5 0
COEFFICIENT DE X6 0
COEFFICIENT DE X7 0
COEFFICIENT DE X8 0
COEFFICIENT DE X9 0
COEFFICIENT DE X10 0
COEFFICIENT DE X11 0
COEFFICIENT DE X12 0
COEFFICIENT DE X13 0
COEFFICIENT DE X14 0
COEFFICIENT DE X15 0
COEFFICIENT DE X16 1
```

```
    CALCUL DE P(X0)

X0=2

P(2)=65536

VOULEZ-VOUS REUTILISER LE PROGRAMME O

X0=3

P(3)=43046721

VOULEZ-VOUS REUTILISER LE PROGRAMME O

X0=1/2

P(1/2)=1.52587891E-05
    = 1 / 65536

VOULEZ-VOUS REUTILISER LE PROGRAMME N
```

On vérifie avec ces deux exemples que la méthode de Horner fournit rapidement des résultats très précis.

4. *Multiplication de polynômes*

a. Principe

Soient deux polynômes P et Q s'écrivant :

$$\begin{cases} P(x) = a_0 + a_1 x + a_2 x^2 + \ldots + a_p x^p \\ Q(x) = b_0 + b_1 x + b_2 x^2 + \ldots + b_q x^q \end{cases}$$

Le produit de ces deux polynômes est alors :

$$\begin{aligned} R(x) &= [P.Q](x) \\ &= P(x).Q(x) \\ &= c_0 + c_1 x + c_2 x^2 + \ldots + c_r x^r \end{aligned}$$

avec :

$$\begin{cases} r = p + q \\ c_n = \sum\limits_{(i,j)} a_i b_j \end{cases}$$

la sommation portant sur tous les termes (i,j) tels que :

$$\begin{cases} i + j = n \\ i \in [0,p] \\ j \in [0,q] \end{cases}$$

b. Programme

Le programme calcule le produit des polynômes P et Q, et place le résultat dans R, afin que celui-ci puisse être utilisé lors de calculs ultérieurs. Les coefficients réels sont suivis de leur approximation rationnelle.

```
3000 CLS
3010  PRINT  TAB( 5);"MULTIPLICATION DE P PAR Q (R=P*Q)
    ": PRINT : PRINT
3020  IF (DP = 0 AND P(0) = 0) OR (DQ = 0 AND Q(0) = 0)
     THEN  PRINT "R EST NUL": PRINT : GOTO 3160
3030 DR = DP + DQ
3040  FOR N = 0 TO DR
3050 R(N) = 0
3060  FOR I = 0 TO DP
3070 J = N - I: IF J < 0 OR J > DQ GOTO 3090
3080 R(N) = R(N) + P(I) * Q(J)
3090  NEXT I
3100  IF R(N) = 0 GOTO 3150
3110 X = R(N): GOSUB 500
3120  PRINT "COEFF. DE X";N;": ";R(N);
3130  IF DD <  > 1 THEN  PRINT  TAB( 29);NN;"/";DD;
3140  PRINT
3150  NEXT N
3160  PRINT : INPUT "APPUYEZ SUR RETURN POUR REVENIR AU
     MENU ";Z$
3170  GOTO 100
```

c. Exemples commentés

- ***Exemple 1*** :

```
    ENTREE DE P

DEGRE DE P 1

COEFFICIENT DE X0 -1
COEFFICIENT DE X1 1


    ENTREE DE Q

DEGRE DE Q 1

COEFFICIENT DE X0 3
COEFFICIENT DE X1 2


    MULTIPLICATION DE P PAR Q (R=P*Q)

COEFF. DE X0: -3
COEFF. DE X1: 1
COEFF. DE X2: 2

APPUYEZ SUR RETURN POUR REVENIR AU MENU
```

- ***Exemple 2*** :

```
    ENTREE DE P

DEGRE DE P 3

COEFFICIENT DE X0 7
COEFFICIENT DE X1 -6
COEFFICIENT DE X2 0
COEFFICIENT DE X3 1
```

```
    ENTREE DE Q

DEGRE DE Q 5

COEFFICIENT DE X0 -3
COEFFICIENT DE X1 0
COEFFICIENT DE X2 1
COEFFICIENT DE X3 0
COEFFICIENT DE X4 -3
COEFFICIENT DE X5 5


    MULTIPLICATION DE P PAR Q (R=P*Q)

COEFF. DE X0: -21
COEFF. DE X1: 18
COEFF. DE X2: 7
COEFF. DE X3: -9
COEFF. DE X4: -21
COEFF. DE X5: 54
COEFF. DE X6: -30
COEFF. DE X7: -3
COEFF. DE X8: 5

APPUYEZ SUR RETURN POUR REVENIR AU MENU
```

● ***Exemple 3*** :

```
    ENTREE DE P

DEGRE DE P 1

COEFFICIENT DE X0 2
COEFFICIENT DE X1 3/5


    ENTREE DE Q

DEGRE DE Q 2

COEFFICIENT DE X0 5/4
COEFFICIENT DE X1 0
COEFFICIENT DE X2 1
```

```
    MULTIPLICATION DE P PAR Q (R=P*Q)

COEFF. DE X0: 2.5           5/2
COEFF. DE X1: .75           3/4
COEFF. DE X2: 2
COEFF. DE X3: .6            3/5

APPUYEZ SUR RETURN POUR REVENIR AU MENU
```

Le nombre de calculs effectués par l'ordinateur étant relativement faible, la précision est excellente et l'exécution rapide.

On trouve effectivement :

$$(x-1)(2x+3) = 2x^2 + x - 3$$

$$(x^3 - 6x + 7)(5x^5 - 3x^4 + x^2 - 3) = 5x^8 - 3x^7 - 30x^6 + 54x^5 - 21x^4 - 9x^3 + 7x^2 + 18x - 21$$

$$\left(\frac{3}{5}x + 2\right)\left(x^2 + \frac{5}{4}\right) = \frac{3}{5}x^3 + 2x^2 + \frac{3}{4}x + \frac{5}{2}$$

5. *Composé de deux polynômes*

a. Principe

Soient deux polynômes P et Q s'écrivant :

$$\begin{cases} P(X) = a_0 + a_1X + a_2X^2 + \ldots + a_pX^p \\ Q(x) = b_0 + b_1x + b_2x^2 + \ldots + b_qx^q \end{cases}$$

On forme le polynôme composé $R(x) = P[Q(x)]$ en remplaçant X par $Q(x)$ dans l'expression de $P(X)$.

Par exemple, avec :

$$\begin{cases} P(X) = X^2 - 3 \\ Q(x) = x + 1 \end{cases}$$

on obtient :

$$\begin{aligned} P[Q(x)] &= (x+1)^2 - 3 \\ &= x^2 + 2x - 2 \end{aligned}$$

De manière générale, le degré du polynôme substitué R(x) est égal au produit des degrés de P et de Q :

$$r = p.q$$

b. Programme

Le programme calcule le polynôme substitué P[Q(x)] à partir des polynômes P et Q entrés en mémoire grâce à l'option 1 du menu ; il place ensuite le résultat dans R.

```
4000 CLS
4010  PRINT  TAB( 3);"CALCUL DU POLYNOME SUBSTITUE (R=P
    (Q))": PRINT : PRINT
4020 DR = DP * DQ:R(0) = P(0): FOR I = 1 TO DR:R(I) = 0
    : NEXT I
4030 DW = 0:W(0) = 1
4040  FOR I = 1 TO DP
4050 DV = DW + DQ
4060  FOR N = 0 TO DV
4070 V(N) = 0
4080  FOR J = 0 TO DW
4090 K = N - J: IF K < 0 OR K > DQ GOTO 4110
4100 V(N) = V(N) + W(J) * Q(K)
4110  NEXT J
4120  NEXT N
4130 DW = DV
4140  FOR N = 0 TO DV
4150 W(N) = V(N)
4160 R(N) = R(N) + P(I) * W(N)
4170  NEXT N
4180  NEXT I
4190  IF DR = 0 AND R(0) = 0 THEN  PRINT "R EST NUL": PRINT
    : GOTO 4270
4200  FOR I = 0 TO DR
4210  IF R(I) = 0 GOTO 4260
4220 X = R(I): GOSUB 500
4230  PRINT  TAB( 2);"COEFF. DE X";I;":";R(I);
4240  IF DD <  > 1 THEN  PRINT  TAB( 29);NN;"/";DD;
4250  PRINT
4260  NEXT I
4270  PRINT : INPUT "APPUYEZ SUR RETURN POUR REVENIR AU
     MENU ";Z$
4280  GOTO 100
```

c. Exemples commentés

● ***Exemple 1*** :

Soient :

$$\begin{cases} P(X) = -5X^2 + 3 \\ Q(x) = 2x^3 - x + 5 \end{cases}$$

```
    ENTREE DE P

DEGRE DE P 2

COEFFICIENT DE X0 3
COEFFICIENT DE X1 0
COEFFICIENT DE X2 -5


    ENTREE DE Q

DEGRE DE Q 3

COEFFICIENT DE X0 5
COEFFICIENT DE X1 -1
COEFFICIENT DE X2 0
COEFFICIENT DE X3 2


  CALCUL DU POLYNOME SUBSTITUE (R=P(Q))

 COEFF. DE X0:-122
 COEFF. DE X1:50
 COEFF. DE X2:-5
 COEFF. DE X3:-100
 COEFF. DE X4:20
 COEFF. DE X6:-20

APPUYEZ SUR RETURN POUR REVENIR AU MENU
```

Le programme fournit le résultat exact :

$$P[Q(x)] = -20x^6 + 20x^4 - 100x^3 - 5x^2 + 50x - 122$$

L'exécution est plus lente que pour une multiplication : en effet, le nombre d'opérations élémentaires nécessaires pour parvenir au résultat est ici beaucoup plus élevé.

● ***Exemple 2*** :

Le programme permet aussi d'élever un polynôme à une puissance quelconque.

Soit à calculer par exemple :

$$R(x) = (2x^2 - 3)^4$$

Posons :

$$P(X) = X^4$$
$$Q(x) = 2x^2 - 3$$

Le programme fournit alors :

$$R(x) = P[Q(x)] = (2x^2 - 3)^4$$

```
    ENTREE DE P

DEGRE DE P 4

COEFFICIENT DE X0 0
COEFFICIENT DE X1 0
COEFFICIENT DE X2 0
COEFFICIENT DE X3 0
COEFFICIENT DE X4 1

    ENTREE DE Q

DEGRE DE Q 2

COEFFICIENT DE X0 -3
COEFFICIENT DE X1 0
COEFFICIENT DE X2 2

 CALCUL DU POLYNOME SUBSTITUE (R=P(Q))

COEFF. DE X0:81
COEFF. DE X2:-216
COEFF. DE X4:216
COEFF. DE X6:-96
COEFF. DE X8:16

APPUYEZ SUR RETURN POUR REVENIR AU MENU
```

- ***Exemple 3*** :

On peut également utiliser le programme pour multiplier un polynôme par un réel λ.

Il suffit alors de prendre :

$$P(X) = \lambda . X$$

Le résultat sera :

$$R(x) = P[Q(x)] = \lambda . Q(x)$$

Prenons par exemple :

$$\begin{cases} P(X) = -\dfrac{X}{5} \\ Q(x) = 2x^2 - 5x + 10 \end{cases}$$

```
    ENTREE DE P

DEGRE DE P 1

COEFFICIENT DE X0 0
COEFFICIENT DE X1 -1/5


    ENTREE DE Q

DEGRE DE Q 2

COEFFICIENT DE X0 10
COEFFICIENT DE X1 -5
COEFFICIENT DE X2 2


  CALCUL DU POLYNOME SUBSTITUE (R=P(Q))

 COEFF. DE X0:-2
 COEFF. DE X1:1
 COEFF. DE X2:-.4             -2/5

APPUYEZ SUR RETURN POUR REVENIR AU MENU
```

On obtient effectivement :

$$R(x) = \frac{2}{5}x^2 - x + 2$$

6. *Division euclidienne*

Il existe deux types de divisions entre polynômes : la division euclidienne et la division suivant les puissances croissantes. La division euclidienne est la plus commune.

a. Principe

Étant donné deux polynômes P et Q, il existe un couple unique de polynômes (H,R) tels que :

$$P = Q.H + R \qquad \text{avec } d^o(R) < d^o(Q)$$

($d^o(R)$ et $d^o(Q)$ désignent les degrés des polynômes R et Q).

H et R sont respectivement le quotient et le reste de la division euclidienne de P par Q.

Soient par exemple :

$$\begin{cases} P(x) = 2x^5 + 2x^3 - x^2 + 2x - 1 \\ Q(x) = x^3 - 1 \end{cases}$$

La disposition manuelle du calcul est alors la suivante :

$$\begin{array}{r|l}
 & 2x^5 + 2x^3 - x^2 + 2x - 1 & x^3 - 1 \\ \cline{3-3}
2x^2.Q(x) \rightarrow & 2x^5 \qquad\quad - 2x^2 & 2x^2 \quad + \quad 2 \\ \cline{2-2}
 & 2x^3 + x^2 + 2x - 1 & (= \frac{2x^5}{x^3}) \quad (= \frac{2x^3}{x^3}) \\
2.Q(x) \rightarrow & 2x^3 \qquad\qquad\quad - 2 & \\ \cline{2-2}
 & x^2 + 2x + 1 &
\end{array}$$

Le quotient est alors :

$$H(x) = 2x^2 + 2$$

et le reste :

$$R(x) = x^2 + 2x + 1$$

On a donc la relation :

$$\underbrace{(2x^5 + 2x^3 - x^2 + 2x - 1)}_{P(x)} = \underbrace{(x^3 - 1)}_{Q(x)}\,\underbrace{(2x^2 + 2)}_{H(x)} + \underbrace{(x^2 + 2x + 1)}_{R(x)}$$

et on a bien :

$$d^o(R) < d^o(Q)$$

On appelle parfois cette division "division suivant les puissances décroissantes" parce que, lors du calcul pratique de la division, les polynômes sont ordonnés suivant les puissances décroissantes de la variable.

b. Programme

Le programme simule le calcul manuel, et procède de manière identique pour obtenir le quotient et le reste, qui sont placés dans H et R pour permettre une éventuelle utilisation ultérieure.

Les polynômes P et Q doivent être introduits grâce à l'option 1 du menu ; le choix de l'option 5 "Division euclidienne" conduit automatiquement au résultat.

```
5000 CLS
5010  PRINT  TAB( 5);"DIVISION EUCLIDIENNE DE P PAR Q"
5020  PRINT  TAB( 13);"P = Q * H + R": PRINT : PRINT
5030  IF DQ = 0 AND Q(DQ) = 0 THEN  PRINT "IMPOSSIBLE .
   ..  Q EST NUL": GOTO 5280
5040 DR = DP: FOR I = 0 TO DP:R(I) = P(I): NEXT I
5050  IF DQ > DP THEN  PRINT "QUOTIENT NUL": GOTO 5200
5060 DH = DP - DQ: PRINT "QUOTIENT H:": PRINT
5070  FOR I = DH TO 0 STEP  - 1
5080 H(I) = R(DR) / Q(DQ)
5090  IF H(I) = 0 GOTO 5140
5100 X = H(I): GOSUB 500
5110  PRINT  TAB( 2);"COEFF. DE X";I;":";H(I);
5120  IF DD <  > 1 THEN  PRINT  TAB( 29);NN;"/";DD;
5130  PRINT
5140  FOR J = I TO DR:R(J) = R(J) - H(I) * Q(J - I): NEXT
    J
```

```
5150  IF DR > 0 THEN DR = DR - 1
5160  NEXT I
5170  IF R(DR) <  > 0 GOTO 5200
5180  IF DR = 0 THEN  PRINT : PRINT "LE RESTE EST NUL":
    GOTO 5280
5190 DR = DR - 1: GOTO 5170
5200  PRINT : PRINT "RESTE R:": PRINT
5210  FOR I = DR TO 0 STEP  - 1
5220  IF R(I) = 0 GOTO 5270
5230 X = R(I): GOSUB 500
5240  PRINT  TAB( 2);"COEFF.DE X";I;":";R(I);
5250  IF DD <  > 1 THEN  PRINT  TAB( 29);NN;"/";DD;
5260  PRINT
5270  NEXT I
5280  PRINT : PRINT : INPUT "APPUYEZ SUR RETURN POUR RE
   VENIR AU MENU ";Z$
5290  GOTO 100
```

c. Exemples commentés

● ***Exemple 1*** :

Retrouvons le résultat de l'exemple précédent:

$$\begin{cases} P(x) = 2x^5 + 2x^3 - x^2 + 2x - 1 \\ Q(x) = x^3 - 1 \end{cases}$$

```
    ENTREE DE P

DEGRE DE P 5

COEFFICIENT DE X0 -1
COEFFICIENT DE X1 2
COEFFICIENT DE X2 -1
COEFFICIENT DE X3 2
COEFFICIENT DE X4 0
COEFFICIENT DE X5 2

    ENTREE DE Q

DEGRE DE Q 3

COEFFICIENT DE X0 -1
```

```
COEFFICIENT DE X1 0
COEFFICIENT DE X2 0
COEFFICIENT DE X3 1


    DIVISION EUCLIDIENNE DE P PAR Q
            P = Q * H + R

QUOTIENT H:

 COEFF. DE X2:2
 COEFF. DE X0:2

RESTE R:

 COEFF.DE X2:1
 COEFF.DE X1:2
 COEFF.DE X0:1
```

- ***Exemple 2*** :

Soient les deux polynômes :

$$\begin{cases} P(x) = x^5 - 1 \\ Q(x) = x - 1 \end{cases}$$

Le polynôme P étant divisible par le polynôme Q, le reste de la division est nul.

```
    ENTREE DE P

DEGRE DE P 5

COEFFICIENT DE X0 -1
COEFFICIENT DE X1 0
COEFFICIENT DE X2 0
COEFFICIENT DE X3 0
COEFFICIENT DE X4 0
COEFFICIENT DE X5 1

    ENTREE DE Q

DEGRE DE Q 1

COEFFICIENT DE X0 -1
COEFFICIENT DE X1 1
```

```
    DIVISION EUCLIDIENNE DE P PAR Q
            P = Q * H + R

QUOTIENT H:

 COEFF. DE X4:1
 COEFF. DE X3:1
 COEFF. DE X2:1
 COEFF. DE X1:1
 COEFF. DE X0:1

LE RESTE EST NUL

APPUYEZ SUR RETURN POUR REVENIR AU MENU
```

● ***Exemple 3*** :

Les résultats obtenus avec des coefficients rationnels sont très précis.

```
    ENTREE DE P

DEGRE DE P 3

COEFFICIENT DE X0 1
COEFFICIENT DE X1 -43/36
COEFFICIENT DE X2 29/3
COEFFICIENT DE X3 5


    ENTREE DE Q

DEGRE DE Q 2

COEFFICIENT DE X0 1/3
COEFFICIENT DE X1 -1/2
COEFFICIENT DE X2 3


    DIVISION EUCLIDIENNE DE P PAR Q
            P = Q * H + R

QUOTIENT H:

 COEFF. DE X1:1.66666667    5/3
 COEFF. DE X0:3.5           7/2
```

```
RESTE R:

 COEFF.DE X0:-.166666667     -1/6
```

7. *Division suivant les puissances croissantes*

a. Principe

Reprenons la disposition pratique de la division euclidienne, en ordonnant cette fois les polynômes suivant les puissances croissantes de la variable.

$$
\begin{array}{l|l}
-1 + 2x - x^2 + 2x^3 + 2x^5 & -1 + x^3 \\
\cline{2-2}
-1 \qquad\qquad\qquad + x^3 & -1 \\
\cline{1-1}
\qquad\quad 2x - x^2 + x^3 + 2x^5 &
\end{array}
$$

Si l'on effectue la division, on constate qu'il est possible, si P n'est pas multiple de Q, de continuer indéfiniment le calcul, ce qui n'est pas le cas avec la division euclidienne qui conduit systématiquement à un couple résultat unique.

D'autre part, on remarque qu'à chaque étape du calcul, il est possible de mettre en facteur dans le reste un monôme x^k, k augmentant à chaque étape.

La donnée des deux polynômes P et Q n'est pas suffisante pour résoudre le problème complètement; on se fixe alors un entier k, et on arrête la division lorsque x^k peut être mis en facteur dans le reste.

On a alors effectué la division suivant les puissances croissantes de P par Q à l'ordre k.

Le théorème s'énonce donc de la manière suivante:

Étant donnés deux polynômes P et Q, avec $Q(0) \neq 0$, et un entier k, il existe un couple unique (H,R) de polynômes tels que:

$$P = Q.H + x^k.R \qquad \text{avec } d°(H) < k.$$

REMARQUE: il est indispensable que le polynôme Q soit de valuation nulle, c'est-à-dire que Q(0) soit différent de 0. En effet, la première division élémentaire doit se faire avec le monôme de degré 0 du polynôme Q : si celui-ci est nul, la division est impossible.

b. Programme

Il faut toujours commencer par introduire les polynômes P et Q grâce à l'option 1 du menu.

L'ordre k de la division est alors le seul paramètre à fournir pour que le programme calcule les polynômes H et R tels que :

$$P(x) \doteq Q(x) . H(x) + x^k . R(x) \qquad \text{avec } d^\circ(H) < k$$

```
6000 CLS
6010  PRINT "   DIVISION SUIVANT LES PUISSANCES": PRINT
    "CROISSANTES DE P PAR Q  (P=Q*H + R*X^N)": PRINT : PRINT

6020  IF Q(0) = 0 THEN  PRINT "IMPOSSIBLE  CAR Q(0)=0": GOTO
    6390
6030  INPUT "ORDRE DE LA DIVISION ";N: PRINT
6040  IF N <  = 0 GOTO 6030
6050 MX = DQ + N - 1
6060  IF MX < DP THEN MX = DP
6070  IF MX > DQ THEN  FOR I = DQ + 1 TO MX:Q(I) = 0: NEXT
    I
6080  FOR I = 0 TO MX:R(I) = 0: NEXT I
6090  FOR I = 0 TO DP:R(I) = P(I): NEXT I
6100  IF N > DQ THEN  FOR I = DQ + 1 TO N:Q(I) = 0: NEXT
    I
6110  FOR I = 0 TO N - 1
6120 H(I) = R(I) / Q(0)
6130  FOR J = I TO MX:R(J) = R(J) - H(I) * Q(J - I): NEXT
    J
6140  NEXT I
6150 DH = N - 1
6160  IF DH = 0 AND H(0) = 0 THEN  PRINT : PRINT "H EST
    NUL": GOTO 6260
6170  IF H(DH) = 0 THEN DH = DH - 1: GOTO 6160
6180  PRINT "QUOTIENT H:": PRINT
6190  FOR I = 0 TO DH
6200  IF H(I) = 0 GOTO 6250
6210 X = H(I): GOSUB 500
6220  PRINT "COEFF. DE X";I;":";H(I);
6230  IF DD <  > 1 THEN  PRINT  TAB( 29);NN;"/";DD;
```

```
6240  PRINT
6250  NEXT I
6260 DR = MX - N
6270  FOR I = 0 TO DR:R(I) = R(I + N): NEXT I
6280  IF DR = 0 GOTO 6300
6290  IF R(DR) = 0 THEN DR = DR - 1: GOTO 6280
6300  IF DR = 0 AND R(0) = 0 THEN  PRINT : PRINT "R EST
    NUL": GOTO 6390
6310  PRINT : PRINT : PRINT "RESTE R:": PRINT
6320  FOR I = 0 TO DR
6330  IF R(I) = 0 GOTO 6380
6340 X = R(I): GOSUB 500
6350  PRINT "COEFF. DE X";I;":";R(I);
6360  IF DD <  > 1 THEN  PRINT  TAB( 29);NN;"/";DD;
6370  PRINT
6380  NEXT I
6390  PRINT : PRINT : INPUT "VOULEZ-VOUS REUTILISER LE P
    ROGRAMME ";Z$
6400  IF Z$ = "O" GOTO 6000
6410  GOTO 100
```

c. Exemples commentés

● ***Exemple 1*** :

Soient les deux polynômes P et Q suivants :

$P(x) = x + 2x^2$

$Q(x) = 1 - x^2 + x^3$

Effectuons alors la division à l'ordre 4 :

$$
\begin{array}{cccccc|l}
x & +\,2x^2 & & & & & 1 - x^2 + x^3 \\
x & & -\,x^3 & +\,x^4 & & & \\
\hline
& 2x^2 & +\,x^3 & -\,x^4 & & & x + 2x^2 + x^3 \\
& 2x^2 & & -\,2x^4 & +\,2x^5 & & \\
\hline
& & x^3 & +\,x^4 & -\,2x^5 & & \\
& & x^3 & & -\,x^5 & +\,x^6 & \\
\hline
& & & x^4 & -\,x^5 & -\,x^6 &
\end{array}
$$

On obtient donc :

$$\underbrace{(x + 2x^2)}_{P(x)} = \underbrace{(1 - x^2 + x^3)}_{Q(x)} \underbrace{(x + 2x^2 + x^3)}_{H(x)} + x^4 \underbrace{(1 - x - x^2)}_{R(x)}$$

La relation $d^o(H) < k$ est bien vérifiée.

Exécutons ce calcul avec le programme :

```
    ENTREE DE P

DEGRE DE P 2

COEFFICIENT DE X0 0
COEFFICIENT DE X1 1
COEFFICIENT DE X2 2

    ENTREE DE Q

DEGRE DE Q 3

COEFFICIENT DE X0 1
COEFFICIENT DE X1 0
COEFFICIENT DE X2 -1
COEFFICIENT DE X3 1

   DIVISION SUIVANT LES PUISSANCES
CROISSANTES DE P PAR Q  (P=Q*H + R*X^N)

ORDRE DE LA DIVISION 4

QUOTIENT H:

COEFF. DE X1:1
COEFF. DE X2:2
COEFF. DE X3:1

RESTE R:

COEFF. DE X0:1
COEFF. DE X1:-1
COEFF. DE X2:-1

VOULEZ-VOUS REUTILISER LE PROGRAMME O
```

Reprenons le calcul, en demandant cette fois une division à l'ordre 6:

```
   DIVISION SUIVANT LES PUISSANCES
CROISSANTES DE P PAR Q  (P=Q*H + R*X^N)

ORDRE DE LA DIVISION 6

QUOTIENT H:

COEFF. DE X1:1
COEFF. DE X2:2
COEFF. DE X3:1
COEFF. DE X4:1
COEFF. DE X5:-1


RESTE R:

COEFF. DE X1:-2
COEFF. DE X2:1


VOULEZ-VOUS REUTILISER LE PROGRAMME O
```

Le quotient a été complété par le terme $x^4 - x^5$.

On remarque qu'il est possible de mettre x en facteur dans le reste: le quotient de la division à l'ordre 7 est le même que celui de la division à l'ordre 6.

Les égalités correspondant aux deux divisions sont alors:

$$(x + 2x^2) = (1 - x^2 + x^3)(x + 2x^2 + x^3 + x^4 - x^5) + x^6(-2x + x^2) \qquad \text{ordre 6}$$

$$(x + 2x^2) = (1 - x^2 + x^3)(x + 2x^2 + x^3 + x^4 - x^5) + x^7(-2 + x) \qquad \text{ordre 7}$$

Le programme confirme ce résultat:

```
   DIVISION SUIVANT LES PUISSANCES
CROISSANTES DE P PAR Q  (P=Q*H + R*X^N)

ORDRE DE LA DIVISION 7

QUOTIENT H:

COEFF. DE X1:1
COEFF. DE X2:2
```

```
COEFF. DE X3:1
COEFF. DE X4:1
COEFF. DE X5:-1


RESTE R:

COEFF. DE X0:-2
COEFF. DE X1:1


VOULEZ-VOUS REUTILISER LE PROGRAMME O
```

REMARQUE : au départ, x peut être mis en facteur dans P ; la division à l'ordre 1 de P par Q ne nécessite aucun calcul :

$$(x + 2x^2) = 0.(1 - x^2 + x^3) + x^1 (1 + 2x)$$

donc :

$$\begin{cases} H(x) = 0 \\ R(x) = 1 + 2x \end{cases}$$

```
   DIVISION SUIVANT LES PUISSANCES
CROISSANTES DE P PAR Q  (P=Q*H + R*X^N)

ORDRE DE LA DIVISION 1


H EST NUL


RESTE R:

COEFF. DE X0:1
COEFF. DE X1:2


VOULEZ-VOUS REUTILISER LE PROGRAMME N
```

● ***Exemple 2*** :

Il est bien sûr possible d'utiliser des coefficients rationnels :

```
    ENTREE DE P


DEGRE DE P 3

COEFFICIENT DE X0 -4
COEFFICIENT DE X1 2
COEFFICIENT DE X2 0
COEFFICIENT DE X3 3/5

    ENTREE DE Q


DEGRE DE Q 2

COEFFICIENT DE X0 -5/4
COEFFICIENT DE X1 0
COEFFICIENT DE X2 1

   DIVISION SUIVANT LES PUISSANCES
CROISSANTES DE P PAR Q  (P=Q*H + R*X^N)


ORDRE DE LA DIVISION 1

QUOTIENT H:

COEFF. DE X0:3.2              16/5


RESTE R:

COEFF. DE X0:2
COEFF. DE X1:-3.2             -16/5
COEFF. DE X2:.6               3/5


VOULEZ-VOUS REUTILISER LE PROGRAMME O



   DIVISION SUIVANT LES PUISSANCES
CROISSANTES DE P PAR Q  (P=Q*H + R*X^N)


ORDRE DE LA DIVISION 5
```

```
QUOTIENT H:

COEFF. DE X0:3.2            16/5
COEFF. DE X1:-1.6           -8/5
COEFF. DE X2:2.56           64/25
COEFF. DE X3:-1.76          -44/25
COEFF. DE X4:2.048          256/125


RESTE R:

COEFF. DE X0:1.76           44/25
COEFF. DE X1:-2.048         -256/125

VOULEZ-VOUS REUTILISER LE PROGRAMME O
```

Cependant, si l'on demande un ordre trop élevé, certaines approximations rationnelles ne sont plus égales aux valeurs exactes :

```
   DIVISION SUIVANT LES PUISSANCES
CROISSANTES DE P PAR Q  (P=Q*H + R*X^N)


ORDRE DE LA DIVISION 9

QUOTIENT H:

COEFF. DE X0:3.2            16/5
COEFF. DE X1:-1.6           -8/5
COEFF. DE X2:2.56           64/25
COEFF. DE X3:-1.76          -44/25
COEFF. DE X4:2.048          256/125
COEFF. DE X5:-1.408         -176/125
COEFF. DE X6:1.6384         1024/625
COEFF. DE X7:-1.1264        -704/625
COEFF. DE X8:1.31072        966/737


RESTE R:

COEFF. DE X0:1.1264         704/625
COEFF. DE X1:-1.31072       -966/737

VOULEZ-VOUS REUTILISER LE PROGRAMME N
```

On peut vérifier par le calcul que le dernier coefficient n'est pas $\frac{966}{737}$ mais $\frac{4096}{3125}$.

8. *PGCD - PPCM de deux polynômes*

a. Principe

Considérons deux polynômes P et Q, et désignons par D_p l'ensemble des diviseurs de P, par D_q l'ensemble des diviseurs de Q. Ce sont donc deux ensembles de polynômes.

Définissons une relation d'ordre sur ces polynômes : soient P_1 et P_2 deux éléments de D_P, on dira que P_1 est supérieur à P_2 si et seulement si :

$$d^o(P_1) > d^o(P_2)$$

Cette relation est une relation d'ordre partiel puisque deux polynômes de degrés égaux ne sont pas comparables.

Considérons maintenant l'ensemble $D = D_P \cap D_Q$ des diviseurs communs à P et à Q. On ne peut pas parler *du* plus grand élément de D car, la relation d'ordre étant partielle, il y a en fait un ensemble de polynômes maximums. On démontre que tous les polynômes de cet ensemble sont proportionnels : ainsi, le PGCD de P et Q sera un polynôme défini "à un coefficient près". Si R est un des plus grands éléments de D, tous les autres seront du type λR avec $\lambda \in \mathbb{R}$.

Par exemple, le PGCD de :

$$\begin{cases} P(x) = (x-1)(x-2) \\ Q(x) = (x-1)(x-3) \end{cases}$$

vaut $\lambda(x-1)$: $(x-1)$ convient, mais $(2x-2)$ et $(-3x+3)$ conviennent également.

Le PPCM s'obtient à partir du PGCD par la relation :

$$P(x).Q(x) = \text{PGCD}(P,Q).\text{PPCM}(P,Q)$$

Il est donc également défini à un coefficient près.

b. Calcul pratique

Celui-ci repose sur l'algorithme d'Euclide.

Après avoir permuté P et Q si $d^o(P) < d^o(Q)$, on effectue la division euclidienne de P par Q :

$$P = Q.H_1 + R_1$$

Si le reste R_1 n'est pas nul, on divise Q par R_1 :

$$Q = R_1.H_2 + R_2$$

On réitère le processus jusqu'à obtenir un reste nul ; le reste précédent est alors un PGCD des polynômes P et Q.

c. Programme

Le programme opère sur les polynômes P et Q ; il place le PGCD dans R et le PPCM dans H.

Les représentants du PGCD et du PPCM choisis seront différents selon l'option de calcul.

En mode réel, les coefficients des monômes de plus haut degré sont rendus égaux à 1.

En mode rationnel, les coefficients sont multipliés par un facteur n de manière à les rendre entiers (n est égal au PPCM des dénominateurs des coefficients divisé par le PGCD de leur numérateur) ; ceci améliore la présentation.

Si par exemple deux polynômes admettent $\lambda(4x + 6)$ comme PGCD, l'option "mode réel" fournira $x + \frac{3}{2}$ tandis que l'option "mode rationnel" donnera $2x + 3$.

```
7000 CLS
7010  PRINT "CALCUL DU PGCD ET DU PPCM DE P ET Q": PRINT
    : PRINT
7020  IF DP = 0 AND P(DP) = 0 THEN  PRINT "IMPOSSIBLE .
    .. P EST NUL": PRINT : GOTO 7710
7030  IF DQ = 0 AND Q(DQ) = 0 THEN  PRINT "IMPOSSIBLE .
    .. Q EST NUL": PRINT : GOTO 7710
7040 DV = DP: FOR I = 0 TO DV:V(I) = P(I): NEXT I
7050 DW = DQ: FOR I = 0 TO DW:W(I) = Q(I): NEXT I
7060 RE = 0
7070  PRINT "CALCUL EN MODE REEL      (R)"
7080  PRINT  TAB( 13);"OU RATIONNEL (Q) ";
7090  INPUT Z$: PRINT
7100  IF Z$ <  > "R" AND Z$ <  > "Q" GOTO 7070
7110  IF Z$ = "R" THEN RE = 1
```

```
7120 DR = DP + DQ
7130  FOR N = 0 TO DR
7140 R(N) = 0
7150  FOR I = 0 TO DP
7160 J = N - I: IF J < 0 OR J > DQ GOTO 7180
7170 R(N) = R(N) + P(I) * Q(J)
7180  NEXT I
7190  NEXT N
7200  IF DQ > DP THEN  FOR I = 0 TO DQ:H(I) = P(I):P(I)
     = Q(I):Q(I) = H(I): NEXT I:DH = DP:DP = DQ:DQ = DH

7210 DH = DP - DQ
7220  FOR I = DH TO 0 STEP  - 1
7230 H(I) = P(DP) / Q(DQ)
7240  FOR J = I TO DP:P(J) = P(J) - H(I) * Q(J - I): NEXT
    J
7250 DP = DP - 1
7260  NEXT I
7270  IF DP > 0 GOTO 7310
7280 DP = 0
7290  IF  ABS (P(0)) > 1E - 6 THEN DQ = 0
7300  GOTO 7340
7310  IF  ABS (P(DP)) < 1E - 6 THEN DP = DP - 1: GOTO 7
    270
7320  FOR I = 0 TO DP - 1:P(I) = P(I) / P(DP): NEXT I:P
    (DP) = 1
7330  FOR I = 0 TO DQ:H(I) = P(I):P(I) = Q(I):Q(I) = H(
    I): NEXT I:DH = DP:DP = DQ:DQ = DH: GOTO 7210
7340  IF DQ = 0 THEN Q(0) = 1: GOTO 7450
7350 PP = 1:PG = 0
7360  IF RE = 1 GOTO 7450
7370  FOR I = 0 TO DQ
7380 X = Q(I): GOSUB 500
7390 A = PP:B =  ABS (DD): GOSUB 700:PP = PP *  ABS (DD
    ) / A
7400  IF PG = 0 THEN PG =  ABS (NN)
7410  IF NN <  > 0 THEN A = PG:B =  ABS (NN): GOSUB 700
    :PG = A
7420  NEXT I
7430  IF PG <  > 0 THEN PP = PP / PG
7440  FOR I = 0 TO DQ:X = Q(I) * PP: GOSUB 500:Q(I) = N
    N / DD: NEXT I
7450  PRINT "DEGRE DU PGCD:";DQ: PRINT
7460  FOR I = 0 TO DQ
7470  IF Q(I) <  > 0 THEN  PRINT "COEFFICIENT DE X ";I;
    ":";Q(I)
```

```
7480  NEXT I
7490 DH = DR - DQ: PRINT : PRINT "DEGRE DU PPCM:";DH: PRINT

7500  FOR I = DH TO 0 STEP  - 1
7510 H(I) = R(DR) / Q(DQ)
7520  FOR J = I TO DR:R(J) = R(J) - H(I) * Q(J - I): NEXT
    J:DR = DR - 1
7530  NEXT I
7540  IF RE = 1 THEN  FOR I = 0 TO DH - 1:H(I) = H(I) /
    H(DH): NEXT I:H(DH) = 1: GOTO 7630
7550 PV = 1:PW = 0
7560  FOR I = 0 TO DH
7570 X = H(I): GOSUB 500
7580 A = PV:B =  ABS (DD): GOSUB 700:PV = PV *  ABS (DD
    ) / A
7590  IF PW = 0 THEN PW =  ABS (NN)
7600  IF NN <  > 0 THEN A = PW:B =  ABS (NN): GOSUB 700
    :PW = A
7610  NEXT I
7620  IF PW <  > 0 THEN PV = PV / PW
7630  FOR I = 0 TO DH
7640  IF H(I) = 0 GOTO 7670
7650  IF RE = 0 THEN X = H(I) * PV: GOSUB 500:H(I) = NN
     / DD
7660  PRINT "COEFFICIENT DE X ";I;":";H(I)
7670  NEXT I
7680 DR = DQ: FOR I = 0 TO DR:R(I) = Q(I): NEXT I
7690 DP = DV: FOR I = 0 TO DP:P(I) = V(I): NEXT I
7700 DQ = DW: FOR I = 0 TO DQ:Q(I) = W(I): NEXT I
7710  PRINT : INPUT "VOULEZ-VOUS REUTILISER LE PROGRAMM
    E ";Z$
7720  IF Z$ = "O" GOTO 7000
7730  GOTO 100
```

d. Exemples commentés

● ***Exemple 1*** :

Cherchons le PGCD de:

$$\begin{cases} P(x) = x^5 + 5x^4 + 7x^3 + 5x^2 + x - 1 \\ Q(x) = x^4 + 4x^3 - 7x + 2 \end{cases}$$

```
    ENTREE DE P

DEGRE DE P 5

COEFFICIENT DE X0 -1
COEFFICIENT DE X1 1
COEFFICIENT DE X2 5
COEFFICIENT DE X3 7
COEFFICIENT DE X4 5
COEFFICIENT DE X5 1

    ENTREE DE Q

DEGRE DE Q 4

COEFFICIENT DE X0 2
COEFFICIENT DE X1 -7
COEFFICIENT DE X2 0
COEFFICIENT DE X3 4
COEFFICIENT DE X4 1

CALCUL DU PGCD ET DU PPCM DE P ET Q


CALCUL EN MODE REEL      (R)
            OU RATIONNEL (Q) ?Q

DEGRE DU PGCD:2

COEFFICIENT DE X 0:-1
COEFFICIENT DE X 1:3
COEFFICIENT DE X 2:1

DEGRE DU PPCM:7

COEFFICIENT DE X 0:2
COEFFICIENT DE X 1:-3
COEFFICIENT DE X 2:-10
COEFFICIENT DE X 3:-8
COEFFICIENT DE X 4:2
COEFFICIENT DE X 5:10
COEFFICIENT DE X 6:6
COEFFICIENT DE X 7:1

VOULEZ-VOUS REUTILISER LE PROGRAMME N
```

Le PGCD de P et Q vaut $\lambda\,(x^2 + 3x - 1)$.

● ***Exemple 2*** :

Considérons maintenant les polynômes à coefficients rationnels suivants :

$$\begin{cases} P(x) = \dfrac{4}{3}x^2 + \dfrac{35}{18}x + \dfrac{1}{2} = (\dfrac{4}{3}x + \dfrac{3}{2})\,(x + \dfrac{1}{3}) \\ Q(x) = -\dfrac{15}{8}x^2 + \dfrac{4}{3}x + \dfrac{8}{3} = (\dfrac{4}{3}x + \dfrac{3}{2})\,(2x - \dfrac{5}{4}) \end{cases}$$

Leur PGCD vaut $R(x) = \lambda\,(\dfrac{4}{3}x + \dfrac{3}{2})$.

Effectuons le calcul en mode réel :

```
    ENTREE DE P


DEGRE DE P 2

COEFFICIENT DE X0 1/2
COEFFICIENT DE X1 35/18
COEFFICIENT DE X2 4/3

    ENTREE DE Q


DEGRE DE Q 2

COEFFICIENT DE X0 -15/8
COEFFICIENT DE X1 4/3
COEFFICIENT DE X2 8/3

CALCUL DU PGCD ET DU PPCM DE P ET Q


CALCUL EN MODE REEL      (R)
            OU RATIONNEL (Q) ?R

DEGRE DU PGCD:1

COEFFICIENT DE X 0:1.125
COEFFICIENT DE X 1:1
```

```
DEGRE DU PPCM:3

COEFFICIENT DE X 0:-.234374999
COEFFICIENT DE X 1:-.536458334
COEFFICIENT DE X 2:.833333334
COEFFICIENT DE X 3:1

VOULEZ-VOUS REUTILISER LE PROGRAMME O
```

Le résultat est fourni avec 1 pour coefficient de x; on a donc $\lambda = \frac{3}{4}$ et $R(x) = x + \frac{9}{8}$.

Effectuons le même calcul en mode rationnel:

```
CALCUL DU PGCD ET DU PPCM DE P ET Q


CALCUL EN MODE REEL      (R)
            OU RATIONNEL (Q) ?Q

DEGRE DU PGCD:1

COEFFICIENT DE X 0:9
COEFFICIENT DE X 1:8

DEGRE DU PPCM:3

COEFFICIENT DE X 0:-45
COEFFICIENT DE X 1:-103
COEFFICIENT DE X 2:160
COEFFICIENT DE X 3:192

VOULEZ-VOUS REUTILISER LE PROGRAMME N
```

Le programme fournit cette fois des coefficients entiers; le PGCD est égal à $R(x) = 8x + 9$.

- ***Exemple 3*** :

Nous avons vu que l'utilisation du "mode rationnel" fournissait des résultats beaucoup plus lisibles lorsque les polynômes avaient des coefficients rationnels.

Bien entendu, si on opère sur des polynômes à coefficients réels, les résultats fournis par l'option "mode rationnel" seront sans signification; il en est de même dans le cas où les rationnels sont trop compliqués pour coïncider avec leur approximation.

Considérons l'exemple suivant :

$$\begin{cases} P(x) = -12 + (15 + 4\sqrt{2})\,x - 5\sqrt{2}\,x^2 \\ Q(x) = -3 + (\sqrt{2} - 6)\,x + 2\sqrt{2}\,x^2 \end{cases}$$

Le PGCD de ces deux polynômes vaut $\lambda(-3 + \sqrt{2}\,x)$.

Le calcul en "mode réel" fournit un résultat correct :

```
    ENTREE DE P

DEGRE DE P 2

COEFFICIENT DE X0 -12
COEFFICIENT DE X1 20.6568542
COEFFICIENT DE X2 -7.07106781

    ENTREE DE Q

DEGRE DE Q 2

COEFFICIENT DE X0 -3
COEFFICIENT DE X1 -4.58578644
COEFFICIENT DE X2 2.82842712

CALCUL DU PGCD ET DU PPCM DE P ET Q


CALCUL EN MODE REEL      (R)
          OU RATIONNEL (Q) ?R

DEGRE DU PGCD:1

COEFFICIENT DE X 0:-2.12132036
COEFFICIENT DE X 1:1

DEGRE DU PPCM:3

COEFFICIENT DE X 0:.848528138
COEFFICIENT DE X 1:.236396106
COEFFICIENT DE X 2:-2.42132033
COEFFICIENT DE X 3:1

VOULEZ-VOUS REUTILISER LE PROGRAMME O
```

alors que le calcul en "mode rationnel" fournit un résultat inexploitable:

```
CALCUL DU PGCD ET DU PPCM DE P ET Q


CALCUL EN MODE REEL      (R)
           OU RATIONNEL (Q) ?Q

DEGRE DU PGCD:1

COEFFICIENT DE X 0:-2378
COEFFICIENT DE X 1:1121

DEGRE DU PPCM:3

COEFFICIENT DE X 0:-2.74046339E+11
COEFFICIENT DE X 1:-7.63480789E+10
COEFFICIENT DE X 2:7.82005846E+11
COEFFICIENT DE X 3:-3.2296669E+11

VOULEZ-VOUS REUTILISER LE PROGRAMME N
```

REMARQUE 1 :

Dans le cas où les polynômes sont de degré élevé, il se peut que le programme ne trouve pas le PGCD exact, et réponde que les polynômes sont premiers entre eux.

L'algorithme de recherche effectue en effet des divisions successives entre polynômes, et les erreurs s'amplifient à chaque étape.

Le programme considère que tout coefficient d'un reste inférieur à 10^{-6} est nul ; cette sécurité peut cependant s'avérer insuffisante.

On veillera en particulier à entrer les nombres réels, comme $\sqrt{2}$, avec le nombre de chiffres maximal que permet l'ordinateur (8 ou 9 chiffres généralement en simple précision).

REMARQUE 2 :

Comme avec les nombres entiers, on peut étendre la définition du PGCD à plusieurs polynômes.

Ainsi, le calcul du PGCD de :

$$\begin{cases} P_1(x) = x^3 - 3x^2 - 10x + 24 \\ P_2(x) = x^3 + 5x^2 - \ \ 2x - 24 \\ P_3(x) = x^3 - \ \ x^2 - \ \ x - \ \ 2 \end{cases}$$

peut être effectué en calculant d'abord le PGCD de P_1 et P_2 (par exemple) :

$$\text{PGCD}(P_1,P_2) = x^2 + x + 6$$

puis en calculant le PGCD de $x^2 + x + 6$ avec P_3 :

$$\text{PGCD}(P_1,P_2,P_3) = x - 2$$

On peut également parler de polynômes premiers entre eux dans leur ensemble ou premiers entre eux deux à deux.

9. *Dérivées successives d'un polynôme*

a. Principe

Un polynôme s'écrit par définition sous la forme d'une somme de monômes :

$$P(x) = \sum_{k=0}^{p} a_k x^k \qquad \text{(où p est le degré de P)}$$

La dérivée de x^k étant $k\,x^{k-1}$, la dérivée d'un monôme $a_k x^k$ vaut $k\,a_k x^{k-1}$.

On obtient donc :

$$P'(x) = \sum_{k=1}^{p} k\,a_k x^{k-1}$$

que l'on peut encore écrire, en mettant en évidence le fait que P' est de degré $p - 1$:

$$P'(x) = \sum_{k=0}^{p-1} (k+1)\,a_{k+1} x^k$$

En réitérant n fois le calcul, on aboutit à la dérivée n-ième de P :

$$P^{(n)}(x) = \sum_{k=0}^{p-n} (k+1)(k+2)\ldots(k+n)\,a_{k+n} x^k$$

ou encore :

$$P^{(n)}(x) = \sum_{k=0}^{p-n} \frac{(k+n)!}{k!} a_{k+n} x^k$$

Le degré de $P^{(n)}$ est égal à $p - n$: si $n > p$, le polynôme $P^{(n)}$ est nul.

b. Programme

Le programme opère sur le polynôme P. Après avoir introduit l'ordre n de la dérivation, il calcule $P^{(n)}$ qu'il place dans R.

```
8000 CLS
8010  PRINT  TAB( 5);"R = DERIVEE N-IEME DE P": PRINT :
     PRINT
8020  INPUT "ORDRE DE LA DERIVEE ";N: PRINT
8030  IF N <  = 0 GOTO 8020
8040  IF N > DP THEN  PRINT "LA DERIVEE DE P D'ORDRE ";
    N;" EST NULLE": GOTO 8150
8050 DR = DP - N: PRINT : PRINT
8060  FOR I = 0 TO DR
8070 R(I) = P(N + I)
8080  IF R(I) = 0 GOTO 8140
8090  FOR K = 1 TO N:R(I) = R(I) * (I + K): NEXT K
8100 X = R(I): GOSUB 500
8110  PRINT "COEFF. DE X";I;": ";R(I);
8120  IF DD <  > 1 THEN  PRINT  TAB( 29);NN;"/";DD;
8130  PRINT
8140  NEXT I
8150  PRINT : PRINT : INPUT "VOULEZ-VOUS REUTILISER LE
    PROGRAMME ";Z$
8160  IF Z$ = "O" GOTO 8000
8170  GOTO 100
```

c. Exemples commentés

- ***Exemple 1*** :

Considérons le monôme $P(x) = x^8$.

```
    ENTREE DE P

DEGRE DE P 8

COEFFICIENT DE X0 0
COEFFICIENT DE X1 0
COEFFICIENT DE X2 0
COEFFICIENT DE X3 0
COEFFICIENT DE X4 0
COEFFICIENT DE X5 0
COEFFICIENT DE X6 0
COEFFICIENT DE X7 0
COEFFICIENT DE X8 1


    R = DERIVEE N-IEME DE P

ORDRE DE LA DERIVEE 1

  COEFFICIENT DE X7:8


VOULEZ-VOUS REUTILISER LE PROGRAMME O


    R = DERIVEE N-IEME DE P

ORDRE DE LA DERIVEE 3

  COEFFICIENT DE X5:336


VOULEZ-VOUS REUTILISER LE PROGRAMME O


    R = DERIVEE N-IEME DE P

ORDRE DE LA DERIVEE 8

  COEFFICIENT DE X0:40320


VOULEZ-VOUS REUTILISER LE PROGRAMME O
```

```
    R = DERIVEE N-IEME DE P

ORDRE DE LA DERIVEE 10

LA DERIVEE DE P D'ORDRE 10 EST NULLE


VOULEZ-VOUS REUTILISER LE PROGRAMME N
```

On obtient bien :

$P'(x) = 8x^7$

$P^{(3)}(x) = 336\,x^5$

$P^{(8)}(x) = 40320$

$P^{(10)}(x) = 0$

• ***Exemple 2*** :

De même, les dérivées du polynôme :

$$P(x) = 4x^5 - \frac{3}{5}x^3 + x^2 - \frac{7}{2}x + 11$$

à l'ordre 1 et 3 sont exactes :

```
    ENTREE DE P


DEGRE DE P 5

COEFFICIENT DE X0 11
COEFFICIENT DE X1 -7/2
COEFFICIENT DE X2 1
COEFFICIENT DE X3 -3/5
COEFFICIENT DE X4 0
COEFFICIENT DE X5 4

    R = DERIVEE N-IEME DE P

ORDRE DE LA DERIVEE 1

COEFF. DE X0: -3.5            -7/2
COEFF. DE X1: 2
```

```
COEFF. DE X2: -1.8          -9/5
COEFF. DE X4: 20


VOULEZ-VOUS REUTILISER LE PROGRAMME O


    R = DERIVEE N-IEME DE P

ORDRE DE LA DERIVEE 3

COEFF. DE X0: -3.6          -18/5
COEFF. DE X2: 240


VOULEZ-VOUS REUTILISER LE PROGRAMME N
```

10. Conclusion

Nous n'avons pas exploité les possibilités de chaînage entre les différentes opérations dans les exemples commentés ; leur utilisation est simple et ne pose aucun problème.

Il faut simplement retenir qu'aucun programme ne modifie les polynômes P et Q : les seuls polynômes affectés sont ceux dans lesquels sont stockés les résultats.

D'autre part, nous avons vu que le calcul de PGCD pouvait être effectué selon deux modes distincts.

Nous retiendrons que le "mode réel" fournit systématiquement un résultat exploitable alors que le "mode rationnel" ne doit être utilisé que dans les cas simples afin d'améliorer la présentation des résultats.

Nous retrouverons ces deux modes de fonctionnement dans l'ensemble du chapitre suivant, consacré aux fractions rationnelles.

5 Fractions rationnelles

1. Introduction

L'ensemble des polynômes, muni des lois addition et multiplication, possède une structure d'anneau commutatif. La principale insuffisance de cette structure réside dans le fait que les polynômes de degré supérieur ou égal à 1 ne possèdent pas de polynôme symétrique pour la loi multiplication.

On définit alors une fraction rationnelle comme le rapport de deux polynômes :

$$R(x) = \frac{P(x)}{Q(x)} \qquad \text{avec } Q(x) \neq 0$$

Le polynôme P est appelé numérateur de la fraction rationnelle, le polynôme Q, qui doit être différent du polynôme nul, dénominateur de la fraction rationnelle.

L'ensemble des polynômes est bien sûr inclus dans l'ensemble des fractions rationnelles, et celui-ci possède une structure de corps commutatif.

En effet, si R est une fraction rationnelle différente de la fraction nulle, son inverse est également une fraction rationnelle :

$$R = \frac{P}{Q} \Rightarrow \frac{1}{R} = R^{-1} = \frac{Q}{P} \qquad (R \neq 0)$$

La manipulation des fractions rationnelles est très fréquente dans les domaines du traitement du signal ou de l'analyse des systèmes linéaires, lors de l'utilisation des transformées de Laplace ou des transformées en Z, pour ne citer que ces deux exemples.

La présentation de ce chapitre est semblable à celle du chapitre concernant les polynômes.

Ainsi, le programme comprend une partie d'entrées et de manipulations sur les fractions rationnelles, et les opérations sur les fractions rationnelles seront le calcul de la valeur de la fraction en un point, multiplication, addition et substitution de deux fractions, calcul des dérivées successives, et enfin décomposition en éléments simples.

Avant d'introduire les programmes réalisant ces fonctions, il est nécessaire d'entrer dans l'ordinateur le menu ainsi que les sous-programmes indispensables au fonctionnement de l'ensemble et la partie de manipulations de fractions.

```
10  DIM P(11,50),DP(11)
20  DIM DN(10,2),PN(10),PR(10)
30 MX = 16
40  DIM F(MX,MX)
50 RE = 1
100 CLS
110  PRINT "CALCULS SUR LES FRACTIONS RATIONNELLES": PRINT
    : PRINT
120  PRINT "1- ENTREES/MANIPULATIONS DE FRACTIONS": PRINT

130  PRINT "2- CALCUL DE F1(X0)": PRINT
140  PRINT "3- MULTIPLICATION DE F1 PAR F2": PRINT
150  PRINT "4- ADDITION DE F1 A F2": PRINT
160  PRINT "5- FRACTION SUBSTITUEE  F=F1(F2)": PRINT
170  PRINT "6- DERIVEES SUCCESSIVES DE F1": PRINT
180  PRINT "7- DECOMPOSITIONS EN ELEMENTS SIMPLES": PRINT

190  PRINT "8- FIN": PRINT : PRINT
200  INPUT "VOTRE CHOIX ";E
210  ON E GOTO 1000,2000,3000,4000,5000,6000,7000,8000
220  GOTO 200
250  IF A < B THEN Y = B:B = A:A = Y
```

```
260 C = A - B *  INT (A / B)
270 A = B:B = C
280  IF C <  > 0 GOTO 260
290  RETURN
300  IF X = 0 THEN NN = 0:DD = 1: GOTO 390
310 AX =  ABS (X):AB = AX:AC =  INT (AB):NN = AC
320 DD = 1:AD = 1:AF = 0:DG = 0: GOTO 370
330 AB = 1 / (AB - AC):AC =  INT (AB)
340 AY = AC * NN + AD:AD = NN:NN = AY
350 AY = AC * DD + AF:AF = DD:DD = AY
360  IF (DD > 1E5 OR DD * NN > 1E9) AND NN > 1 THEN DG =
    1:NN = AD / AF:DD = 1: GOTO 380
370  IF 1E6 *  ABS (AX * DD - NN) > DD * AX GOTO 330
380 NN = NN *  SGN (X)
390  RETURN
400 DP(K) = DP(I) + DP(J)
410  FOR M = 0 TO DP(K)
420 P(K,M) = 0
430  FOR H = 0 TO DP(I)
440 L = M - H
450  IF L < 0 OR L > DP(J) GOTO 470
460 P(K,M) = P(K,M) + P(I,H) * P(J,L)
470  NEXT H
480  NEXT M
490  RETURN
500 DP(6) = DP(4): FOR I = 0 TO DP(6):P(6,I) = P(4,I): NEXT
    I
510 DP(7) = DP(5): FOR I = 0 TO DP(7):P(7,I) = P(5,I): NEXT
    I
520  IF DP(7) > DP(6) THEN  FOR I = 0 TO DP(7):P(8,I) =
    P(6,I):P(6,I) = P(7,I):P(7,I) = P(8,I): NEXT I:DP(8)
     = DP(6):DP(6) = DP(7):DP(7) = DP(8)
530 I = 6:J = 7:K = 8:L = I: GOSUB 900
540  IF DP(6) = 0 GOTO 570
550  FOR I = 0 TO DP(6) - 1:P(6,I) = P(6,I) / P(6,DP(6))
    : NEXT I:P(6,DP(6)) = 1
560  FOR I = 0 TO DP(7):P(8,I) = P(6,I):P(6,I) = P(7,I):
    P(7,I) = P(8,I): NEXT I:DP(8) = DP(6):DP(6) = DP(7):
    DP(7) = DP(8): GOTO 530
570  IF  ABS (P(6,0)) > 1E - 6 THEN DP(7) = 0
580  IF DP(7) = 0 THEN P(7,0) = 1
590 PP = 1:PG = 0
600  IF RE = 0 THEN  FOR V = 0 TO DP(7):X = P(7,V): GOSUB
    300:P(7,V) = NN / DD: NEXT V
610  FOR V = 0 TO 1
```

```
620 G = 4 + V:T = 8 + V
630 I = G:J = 7:K = T:L = I: GOSUB 900
640  IF RE = 1 GOTO 710
650  FOR I = 0 TO DP(T)
660 X = P(T,I): GOSUB 300
670 A = PP:B =  ABS (DD): GOSUB 250:PP = PP *  ABS (DD) /
    A
680  IF PG = 0 AND DG = 0 THEN PG =  ABS (NN)
690  IF NN <  > 0 AND DG = 0 THEN A = PG:B =  ABS (NN): GOSUB
    250:PG = A
700  NEXT I
710  NEXT V
720  IF PG <  > 0 THEN PP = PP / PG
730  FOR V = 0 TO 1
740 G = 4 + V:T = 8 + V
750 DP(G) = DP(T)
760  FOR J = 0 TO DP(G)
770 P(G,J) = P(T,J)
780  IF RE = 0 THEN X = P(G,J) * PP: GOSUB 300:P(G,J) =
    NN / DD
790  NEXT J
800  NEXT V
810  RETURN
900 DP(L) = DP(I): FOR H = 0 TO L:P(L,H) = P(I,H): NEXT
    H
910 DP(K) = DP(L) - DP(J): IF DP(K) < 0 THEN DP(K) = 0:P
    (K,0) = 0: RETURN
920  FOR H = DP(K) TO 0 STEP  - 1
930 P(K,H) = P(L,DP(L)) / P(J,DP(J))
940  FOR M = H TO DP(L):P(L,M) = P(L,M) - P(K,H) * P(J,M
     - H): NEXT M
950  IF DP(L) > 0 THEN DP(L) = DP(L) - 1
960  NEXT H
970  IF DP(L) = 0 THEN  RETURN
980  IF  ABS (P(L,DP(L))) < 1E - 6 THEN DP(L) = DP(L) -
    1: GOTO 970
990  RETURN
1000 CLS
1010  PRINT  TAB( 3);"ENTREES/MANIPULATIONS DE FRACTIONS
    ": PRINT : PRINT
1020  PRINT "1- ENTREE DE F1"
1030  PRINT "2- ENTREE DE F2": PRINT
1040  PRINT "3- RAPPEL DE F1"
1050  PRINT "4- RAPPEL DE F2"
1060  PRINT "5- RAPPEL DE F": PRINT
```

```
1070  PRINT "6- TRANSFERT DE F1 DANS F "
1080  PRINT "7- TRANSFERT DE F2 DANS F "
1090  PRINT "8- TRANSFERT DE F  DANS F1"
1100  PRINT "9- TRANSFERT DE F  DANS F2": PRINT
1110  PRINT "10- REMPLACEMENT DE F PAR 1/F": PRINT
1120  PRINT "11- PASSAGE EN MODE REEL"
1130  PRINT "12- PASSAGE EN MODE RATIONNEL": PRINT
1140  PRINT "13- RETOUR AU MENU": PRINT
1150  INPUT "VOTRE CHOIX ";Z$:E =  VAL (Z$)
1160  IF E < 1 GOTO 1000
1170 CLS
1180  IF E > 12 GOTO 100
1190  IF E = 12 THEN RE = 0: GOTO 1000
1200  IF E = 11 THEN RE = 1: GOTO 1000
1210  IF E = 10 GOTO 1570
1220  IF E = 9 THEN I = 4:J = 2: GOTO 1620
1230  IF E = 8 THEN I = 4:J = 0: GOTO 1620
1240  IF E = 7 THEN I = 2:J = 4: GOTO 1620
1250  IF E = 6 THEN I = 0:J = 4: GOTO 1620
1260  ON E GOTO 1270,1290,1480,1530,1550
1270  PRINT  TAB( 5);"ENTREE DE F1": PRINT : PRINT
1280 I = 0: GOTO 1310
1290  PRINT  TAB( 5);"ENTREE DE F2": PRINT : PRINT
1300 I = 2
1310  FOR L = I TO I + 1
1320  IF L = I THEN  PRINT "ENTREZ LE NUMERATEUR:"
1330  IF L = I + 1 THEN  PRINT "ENTREZ LE DENOMINATEUR:"

1340  PRINT : INPUT "DEGRE ";DP(L)
1350  IF DP(L) < 0 OR DP(L) > 10 OR DP(L) <  >  INT (DP(
    L)) GOTO 1340
1360  FOR J = 0 TO DP(L)
1370  PRINT "  COEFFICIENT DE X";J;
1380  INPUT " ";Z$
1390  GOSUB 1900
1400  IF ER = 1 GOTO 1370
1410 P(L,J) = N
1420  IF J = DP(L) AND P(L,J) = 0 AND (L = I + 1 OR DP(L
    ) > 0) THEN  PRINT "IMPOSSIBLE...": GOTO 1370
1430  NEXT J
1440  PRINT : PRINT
1450  IF DP(I) = 0 AND P(I,0) = 0 THEN L = I + 2:DP(I +
    1) = 0:P(I + 1,0) = 1
1460  NEXT L
1470  GOTO 1000
```

```
1480  PRINT  TAB( 5);"RAPPEL DE F1": PRINT : PRINT
1490 I = 0
1500  GOSUB 1670
1510  PRINT : PRINT : INPUT "APPUYEZ SUR RETURN POUR REV
    ENIR AU MENU";Z$
1520  GOTO 1000
1530  PRINT  TAB( 5);"RAPPEL DE F2": PRINT : PRINT
1540 I = 2: GOTO 1500
1550  PRINT  TAB( 5);"RAPPEL DE F": PRINT : PRINT
1560 I = 4: GOTO 1500
1570  IF (DP(5) = 0 AND P(5,0) = 0) OR (DP(4) = 0 AND P(
    4,0) = 0) THEN  PRINT "IMPOSSIBLE": GOTO 1650
1580  FOR K = 0 TO DP(4):P(8,K) = P(4,K): NEXT K:DP(8) =
    DP(4)
1590  FOR K = 0 TO DP(5):P(4,K) = P(5,K): NEXT K:DP(4) =
    DP(5)
1600  FOR K = 0 TO DP(8):P(5,K) = P(8,K): NEXT K:DP(5) =
    DP(8)
1610  GOTO 1640
1620  FOR K = 0 TO DP(I):P(J,K) = P(I,K): NEXT K:DP(J) =
    DP(I)
1630  FOR K = 0 TO DP(I + 1):P(J + 1,K) = P(I + 1,K): NEXT
    K:DP(J + 1) = DP(I + 1)
1640  PRINT "TRANSFERT EFFECTUE"
1650  PRINT : INPUT "APPUYEZ SUR RETURN ";Z$
1660  GOTO 1000
1670  IF DP(I + 1) = 0 AND P(I + 1,0) = 0 THEN  PRINT "L
    A FRACTION N'EST PAS DEFINIE": PRINT : PRINT : GOTO
    1890
1680  IF DP(I) = 0 AND P(I,0) = 0 THEN  PRINT "LA FRACTI
    ON EST NULLE": PRINT : PRINT : GOTO 1890
1690 L = I: PRINT "NUMERATEUR:": GOSUB 1800
1700 L = I + 1: PRINT "DENOMINATEUR:"
1800  PRINT
1810  FOR J = 0 TO DP(L)
1820 X = P(L,J): GOSUB 300
1830  IF X = 0 GOTO 1880
1840  PRINT "COEFF. DE X";J;":";X;
1850  IF DG = 1 THEN NN = AD *  SGN (X):DD = AF
1860  IF DD > 1 THEN  PRINT  TAB( 30);NN;"/";DD;
1870  PRINT
1880  NEXT J: PRINT : PRINT
1890  RETURN
1900 ER = 0:LZ =  LEN (Z$):FD = 0
1910  IF LZ = 0 THEN ER = 1: GOTO 1990
```

```
1920  FOR K = 1 TO LZ
1930  IF  MID$ (Z$,K,1) = "/" THEN FD = K:K = LZ
1940  NEXT K
1950  IF FD = 0 THEN N =  VAL (Z$): GOTO 1990
1960 B =  VAL ( MID$ (Z$,FD + 1,LZ - FD))
1970  IF B = 0 THEN ER = 1: GOTO 1990
1980 N =  VAL ( MID$ (Z$,1,FD - 1)) / B
1990  RETURN
8000  END
```

L'ensemble nécessite environ 15 Koctets de mémoire, dont 5 Koctets pour le stockage des variables.

2. *Entrées et manipulations de fractions*

Tous les programmes opèrent sur deux fractions F1 et F2, et placent le résultat dans F.

Cette partie permet d'entrer les fractions F1 et F2, de rappeler les fractions F1, F2 et F, d'effectuer des transferts entre F1, F2 et F, et enfin de remplacer F par son inverse $\frac{1}{F}$.

a. Entrée d'une fraction

L'entrée d'une fraction consiste à introduire successivement son numérateur et son dénominateur, ce qui se fait de la même manière que dans le programme sur les polynômes.

En particulier, il est toujours possible d'introduire un rationnel en utilisant la barre oblique.
Entrons par exemple la fraction:

$$F1\ (x) = \frac{2x^3 - \frac{5}{3}x + 1}{x^2 + \frac{1}{2}x}$$

```
    ENTREE DE F1


ENTREZ LE NUMERATEUR:

DEGRE 3
  COEFFICIENT DE X0 1
  COEFFICIENT DE X1 -5/3
  COEFFICIENT DE X2 0
  COEFFICIENT DE X3 2


ENTREZ LE DENOMINATEUR:

DEGRE 2
  COEFFICIENT DE X0 0
  COEFFICIENT DE X1 1/2
  COEFFICIENT DE X2 1
```

b. Rappel d'une fraction

Le rappel d'une fraction élimine les termes à coefficients nuls et indique les approximations rationnelles correspondantes.

Rappelons que les approximations rationnelles ne sont que des approximations et peuvent être différentes des nombres rationnels si ceux-ci sont compliqués.

```
    RAPPEL DE F1


NUMERATEUR:

COEFF. DE X0:1
COEFF. DE X1:-1.66666667     -5/3
COEFF. DE X3:2


DENOMINATEUR:

COEFF. DE X1:.5              1/2
COEFF. DE X2:1
```

c. Transfert de fraction

Les opérations de transfert sont évidentes : elles recopient la fraction de départ (ou fraction source) dans la fraction d'arrivée (ou fraction destination).

d. Inversion de F

L'option "remplacement de F par 1/F" effectue l'inversion de la fraction F.

Si on veut inverser F1 ou F2, il suffit de transférer la fraction dans F, d'inverser F et de réintégrer F dans F1 ou F2.

e. Mode réel et mode rationnel

Ces modes de calcul sont les mêmes que dans le chapitre précédent. Le mode de calcul par défaut est le mode réel.

Le "mode rationnel" améliore la présentation du résultat à la double condition suivante : la fraction ne doit pas contenir de coefficients réels, et les degrés des fractions ne doivent pas être trop élevés (supérieurs à 5).

Une particularité intéressante du programme est la présentation des résultats sous forme de fractions irréductibles ; pour cela, l'ordinateur calcule le PGCD du numérateur et du dénominateur, de la même manière que dans le chapitre précédent.

3. *Calcul de F1 (x_0)*

a. Principe

Le programme permet de calculer la valeur de la fraction F1 en un point x_0.

Le numérateur et le dénominateur sont calculés par la méthode de Horner exposée dans le chapitre précédent.

Si le dénominateur est nul, la fraction n'est pas définie en x_0 et le programme l'indique.

Dans le cas contraire, l'ordinateur affiche le numérateur, le dénominateur et leur rapport.

b. Programme

On peut introduire la valeur x_0 avec la barre oblique.

```
2000 CLS
2010  PRINT  TAB( 5);"CALCUL DE F1(X0)": PRINT : PRINT
2020  INPUT "X0=";Z$: GOSUB 1900
2030  IF ER = 1 GOTO 2020
2040  PRINT : PRINT :X = N:N = 0
2050  FOR I = 0 TO DP(0):N = N * X + P(0,DP(0) - I): NEXT
    I
2060 D = 0
2070  FOR I = 0 TO DP(1):D = D * X + P(1,DP(1) - I): NEXT
    I
2080  IF D = 0 THEN  PRINT "F1 N'EST PAS DEFINIE EN ";Z$
    : GOTO 2110
2090  PRINT "F1(";Z$;") = ";N;" / ";D
2100  PRINT : PRINT  TAB( 5);" = ";N / D
2110  PRINT : PRINT : INPUT "VOULEZ-VOUS REUTILISER LE P
    ROGRAMME ";Z$
2120  IF Z$ = "O" GOTO 2000
2130  GOTO 100
```

c. Exemple commenté

```
    ENTREE DE F1


ENTREZ LE NUMERATEUR:

DEGRE 3
  COEFFICIENT DE X0 2
  COEFFICIENT DE X1 -5
  COEFFICIENT DE X2 0
  COEFFICIENT DE X3 1


ENTREZ LE DENOMINATEUR:

DEGRE 2
  COEFFICIENT DE X0 -1
  COEFFICIENT DE X1 0
  COEFFICIENT DE X2 1
```

```
    CALCUL DE F1(X0)

X0=1
F1 N'EST PAS DEFINIE EN 1

VOULEZ-VOUS REUTILISER LE PROGRAMME O
    CALCUL DE F1(X0)

X0=2
F1(2) = 0 / 3
     = 0

VOULEZ-VOUS REUTILISER LE PROGRAMME O
    CALCUL DE F1(X0)

X0=-2
F1(-2) = 4 / 3
     = 1.33333333

VOULEZ-VOUS REUTILISER LE PROGRAMME O
    CALCUL DE F1(X0)

X0=1/2
F1(1/2) = -.375 / -.75
     = .5

VOULEZ-VOUS REUTILISER LE PROGRAMME N
```

Le programme donne les valeurs de la fraction :

$$F1\,(x) = \frac{x^3 - 5x + 2}{x^2 - 1}$$

aux points -2, $\frac{1}{2}$ et 2.

Remarquons que le numérateur et le dénominateur ne sont pas toujours entiers ; il ne s'agit pas ici d'approximations rationnelles.

4. Multiplication de F1 par F2

a. Principe

Le programme calcule le produit des fractions F1 et F2, et place le résultat dans F :

Si :

$$F1 = \frac{P_1}{Q_1} \text{ et } F2 = \frac{P_2}{Q_2}$$

alors :

$$F1.F2 = \frac{P_1.P_2}{Q_1.Q_2}$$

b. Programme

Le programme calcule les produits $P_1.P_2$ et $Q_1.Q_2$ et fait ensuite appel au sous-programme chargé d'obtenir un représentant irréductible de la fraction F.

```
3000 CLS
3010  PRINT  TAB( 2);"MULTIPLICATION DE F1 PAR F2 (F=F1*
    F2)": PRINT : PRINT
3020  IF DP(1) = 0 AND P(1,0) = 0 THEN  PRINT "IMPOSSIBL
    E ... F1 N'EST PAS DEFINIE": PRINT : GOTO 3080
```

```
3030  IF DP(3) = 0 AND P(3,0) = 0 THEN  PRINT "IMPOSSIBL
    E ... F2 N'EST PAS DEFINIE": PRINT : GOTO 3080
3040 I = 0:J = 2:K = 4: GOSUB 400
3050 I = 1:J = 3:K = 5: GOSUB 400
3060  GOSUB 500
3070 I = 4: GOSUB 1670
3080  PRINT : INPUT "APPUYEZ SUR RETURN POUR REVENIR AU
    MENU";Z$
3090  GOTO 100
```

c. Exemples commentés

• ***Exemple 1*** :

Soient les deux fractions:

$$F1\,(x) = \frac{x-1}{x^2+5x-7} \qquad F2\,(x) = \frac{x+3}{x^2-1}$$

Le produit F1.F2 vaut:

$$F(x) = F1\,(x).F2\,(x) = \frac{(x-1)\,(x+3)}{(x^2+5x-7)\,(x^2-1)}$$

Le programme simplifie la fraction par $(x-1)$ et fournit le résultat:

$$F(x) = \frac{x+3}{(x^2+5x-7)\,(x+1)} = \frac{x+3}{x^3+6x^2-2x-7}$$

```
    ENTREE DE F1


ENTREZ LE NUMERATEUR:

DEGRE 1
  COEFFICIENT DE X0 -1
  COEFFICIENT DE X1 1


ENTREZ LE DENOMINATEUR:

DEGRE 2
  COEFFICIENT DE X0 -7
  COEFFICIENT DE X1 5
  COEFFICIENT DE X2 1
```

```
    ENTREE DE F2


ENTREZ LE NUMERATEUR:

DEGRE 1
  COEFFICIENT DE X0 3
  COEFFICIENT DE X1 1


ENTREZ LE DENOMINATEUR:

DEGRE 2
  COEFFICIENT DE X0 -1
  COEFFICIENT DE X1 0
  COEFFICIENT DE X2 1



 MULTIPLICATION DE F1 PAR F2 (F=F1*F2)


NUMERATEUR:

COEFF. DE X0:3
COEFF. DE X1:1


DENOMINATEUR:

COEFF. DE X0:-7
COEFF. DE X1:-2
COEFF. DE X2:6
COEFF. DE X3:1
```

• ***Exemple 2*** :

Considérons la fraction suivante :

$$F1\,(x) = \frac{2x^3 - 7x^2 + 12x - 7}{6x^4 - 6x^3 - 5x^2 + 3x + 2}$$

Cette fraction est simplifiable par $(x - 1)$.

Pour obtenir la fraction simplifiée, on peut multiplier F1 par la fraction unité, soit $F2 = 1$ (à entrer sous la forme 1/1).

```
    ENTREE DE F1


ENTREZ LE NUMERATEUR:

DEGRE 3
  COEFFICIENT DE X0 -7
  COEFFICIENT DE X1 12
  COEFFICIENT DE X2 -7
  COEFFICIENT DE X3 2


ENTREZ LE DENOMINATEUR:

DEGRE 4
  COEFFICIENT DE X0 2
  COEFFICIENT DE X1 3
  COEFFICIENT DE X2 -5
  COEFFICIENT DE X3 -6
  COEFFICIENT DE X4 6


    ENTREE DE F2


ENTREZ LE NUMERATEUR:

DEGRE 0
  COEFFICIENT DE X0 1


ENTREZ LE DENOMINATEUR:

DEGRE 0
  COEFFICIENT DE X0 1


 MULTIPLICATION DE F1 PAR F2 (F=F1*F2)


NUMERATEUR:

COEFF. DE X0:7
COEFF. DE X1:-5
COEFF. DE X2:2
```

```
DENOMINATEUR:

COEFF. DE X0:-2
COEFF. DE X1:-5
COEFF. DE X3:6
```

Le programme fournit ainsi la fraction F1 simplifiée :

$$F1\ (x) = \frac{2x^2 - 5x + 7}{6x^3 - 5x - 2}$$

5. *Addition de F1 à F2*

a. Principe

Soient deux fractions $F1 = \frac{P_1}{Q_1}$ et $F2 = \frac{P_2}{Q_2}$.

Prenons $Q_1 . Q_2$ comme dénominateur commun :

$$F = F1 + F2 = \frac{P_1Q_2}{Q_1Q_2} + \frac{P_2Q_1}{Q_1Q_2} = \frac{P_1Q_2 + P_2Q_1}{Q_1Q_2}$$

Le programme calcule donc les trois produits de polynômes P_1Q_2, P_2Q_1 et Q_1Q_2, effectue l'addition $P_1Q_2 + P_2Q_1$, et fait enfin appel au sous-programme de simplification de la fraction.

b. Programme

```
4000 CLS
4010  PRINT  TAB( 5);"ADDITION DE F1 A F2 (F=F1+F2)": PRINT
    : PRINT
4020  IF DP(1) = 0 AND P(1,0) = 0 THEN  PRINT "IMPOSSIBL
    E ... F1 N'EST PAS DEFINIE": PRINT : GOTO 4130
4030  IF DP(3) = 0 AND P(3,0) = 0 THEN  PRINT "IMPOSSIBL
    E ... F2 N'EST PAS DEFINIE": PRINT : GOTO 4130
4040 I = 1:J = 3:K = 5: GOSUB 400
```

```
4050 I = 0:J = 3:K = 6: GOSUB 400
4060 I = 1:J = 2:K = 4: GOSUB 400
4070  IF DP(6) > DP(4) THEN  FOR I = DP(4) + 1 TO DP(6):
    P(4,I) = 0: NEXT I:DP(4) = DP(6)
4080  FOR I = 0 TO DP(4):P(4,I) = P(4,I) + P(6,I): NEXT
    I
4090  IF P(4,DP(4)) <  > 0 OR DP(4) = 0 GOTO 4110
4100 DP(4) = DP(4) - 1: GOTO 4090
4110  GOSUB 500
4120 I = 4: GOSUB 1670
4130  PRINT : INPUT "APPUYEZ SUR RETURN POUR REVENIR AU
    MENU";Z$
4140  GOTO 100
```

c. Exemples commentés

• ***Exemple 1*** :

Considérons les deux fractions :

$$\begin{cases} F1\ (x) = \dfrac{2x + 3}{x - 1} \\[2ex] F2\ (x) = \dfrac{-5x + 1}{x^2 - 1} \end{cases}$$

$$F(x) = F1(x) + F2(x) = \frac{(2x + 3)\ (x + 1) - 5x + 1}{x^2 - 1} = \frac{2x^2 + 4}{x^2 - 1}$$

Le programme confirme ce résultat :

```
    ENTREE DE F1


ENTREZ LE NUMERATEUR:

DEGRE 1
  COEFFICIENT DE X0 3
  COEFFICIENT DE X1 2


ENTREZ LE DENOMINATEUR:

DEGRE 1
  COEFFICIENT DE X0 -1
  COEFFICIENT DE X1 1
```

```
    ENTREE DE F2


ENTREZ LE NUMERATEUR:

DEGRE 1
  COEFFICIENT DE X0 1
  COEFFICIENT DE X1 -5


ENTREZ LE DENOMINATEUR:

DEGRE 2
  COEFFICIENT DE X0 -1
  COEFFICIENT DE X1 0
  COEFFICIENT DE X2 1


    ADDITION DE F1 A F2 (F=F1+F2)


NUMERATEUR:

COEFF. DE X0:4
COEFF. DE X2:2


DENOMINATEUR:

COEFF. DE X0:-1
COEFF. DE X2:1
```

● ***Exemple 2*** :

Nous avons vu qu'il était possible d'obtenir la simplification d'une fraction rationnelle en la multipliant par la fraction unité $\frac{1}{1}$.

On peut encore lui ajouter la fraction nulle $\frac{0}{1}$ pour obtenir le même résultat.

L'exemple suivant montre la simplification de la fraction à coefficients rationnels :

$$F1\,(x) = \frac{\frac{5}{3}x^2 + \frac{73}{18}x + \frac{7}{6}}{\frac{10}{3}x^2 + \frac{59}{12}x - \frac{35}{8}}$$

```
    ENTREE DE F1


ENTREZ LE NUMERATEUR:

DEGRE 2
  COEFFICIENT DE X0 7/6
  COEFFICIENT DE X1 73/18
  COEFFICIENT DE X2 5/3


ENTREZ LE DENOMINATEUR:

DEGRE 2
  COEFFICIENT DE X0 -35/8
  COEFFICIENT DE X1 59/12
  COEFFICIENT DE X2 10/3



    ENTREE DE F2


ENTREZ LE NUMERATEUR:

DEGRE 0
  COEFFICIENT DE X0 0



    ADDITION DE F1 A F2 (F=F1+F2)


NUMERATEUR:

COEFF. DE X0:4
COEFF. DE X1:12


DENOMINATEUR:

COEFF. DE X0:-15
COEFF. DE X1:24
```

La fraction simplifiée vaut :

$$F1\,(x) = \frac{12x + 4}{24x - 15}$$

Remarquons sur cet exemple le double rôle du programme de simplification : il divise numérateur et dénominateur par un représentant de leur PGCD afin d'obtenir une fraction irréductible ; il multiplie numérateur et dénominateur par un nombre adéquat rendant tous les coefficients entiers (valable uniquement en mode rationnel).

6. *Fraction substituée*

a. Principe

Nous avons vu lors de la substitution de polynômes que la notation $R(x) = P[Q(x)]$ indiquait que l'on remplaçait x par Q(x) dans l'expression de P.

De la même manière, nous allons définir une fraction substituée.

Soient $F1 = \frac{P_1}{Q_1}$ et $F2 = \frac{P_2}{Q_2}$ deux fractions rationnelles ; la notation $F = F1\,[F2(x)]$ indique que l'on remplace x par F2(x) dans l'expression de F1.

Explicitons F en fonction des polynômes P_1, Q_1, P_2 et Q_2 :

$$F(x) = \frac{P_1\,[\frac{P_2}{Q_2}]}{Q_1\,[\frac{P_2}{Q_2}]} = \frac{a_0 + a_1\,[\frac{P_2}{Q_2}] + \dots + a_{p_1}\,[\frac{P_2}{Q_2}]^{p1}}{b_0 + b_1\,[\frac{P_2}{Q_2}] + \dots + b_{q_1}\,[\frac{P_2}{Q_2}]^{q1}}$$

où les (a_i) sont les coefficients du polynôme P_1 de degré p_1, et (b_i) les coefficients du polynôme Q_1 de degré q_1.

Cette expression peut également s'écrire :

$$F(x) = \frac{a_0 Q_2^{p1} + a_1 P_2 Q_2^{p1-1} + a_2 P_2^2 Q_2^{p1-2} + \dots + a_{p_1} P_2^{p1}}{b_0 Q_2^{q1} + b_1 P_2 Q_2^{q1-1} + b_2 P_2^2 Q_2^{q1-2} + \dots + b_{q_1} P_2^{q1}} \cdot \frac{Q_2^{q1}}{Q_2^{p1}}$$

Le numérateur et le dénominateur sont en fait des polynômes en P_2 à coefficients polynômiaux.

Ils sont calculés dans le programme par une généralisation de la méthode de Hörner.

b. Programme

```
5000 CLS
5010  PRINT  TAB( 5);"FRACTION SUBSTITUEE  F=F1(F2)": PRINT
    : PRINT
5020  IF DP(1) = 0 AND P(1,0) = 0 THEN  PRINT "IMPOSSIBL
    E ... F1 N'EST PAS DEFINIE": PRINT : GOTO 5320
5030  IF DP(3) = 0 AND P(3,0) = 0 THEN  PRINT "IMPOSSIBL
    E ... F2 N'EST PAS DEFINIE": PRINT : GOTO 5320
5040  FOR T = 0 TO 1
5050 G = 4 + T
5060 DP(G) = 0:P(G,0) = P(T,DP(T))
5070 DP(6) = 0:P(6,0) = 1
5080  IF DP(T) = 0 GOTO 5180
5090  FOR N = 1 TO DP(T)
5100 I = 6:J = 3:K = 7: GOSUB 400
5110 DP(6) = DP(7): FOR V = 0 TO DP(6):P(6,V) = P(7,V): NEXT
    V
5120 I = G:J = 2:K = 7: GOSUB 400
5130 DP(G) = DP(7): FOR V = 0 TO DP(G):P(G,V) = P(7,V): NEXT
    V
5140  IF DP(6) > DP(G) THEN  FOR V = DP(G) + 1 TO DP(6):
    P(G,V) = 0: NEXT V:DP(G) = DP(6)
5160  FOR V = 0 TO DP(G):P(G,V) = P(G,V) + P(T,DP(T) - N
    ) * P(6,V): NEXT V
5170  NEXT N
5180  IF DP(G) > 0 AND P(G,DP(G)) = 0 THEN DP(G) = DP(G)
     - 1: GOTO 5180
5190  NEXT T
5200  IF DP(1) = DP(0) GOTO 5300
5210  IF DP(1) > DP(0) THEN G = 4
5220  IF DP(1) < DP(0) THEN G = 5
5230 DP(6) = 0:P(6,0) = 1
5240  FOR N = 1 TO  ABS (DP(1) - DP(0))
5250 I = 6:J = 3:K = 7: GOSUB 400
5260 DP(6) = DP(7): FOR V = 0 TO DP(6):P(6,V) = P(7,V): NEXT
    V
5270  NEXT N
5280 I = G:J = 6:K = 7: GOSUB 400
5290 DP(G) = DP(7): FOR V = 0 TO DP(G):P(G,V) = P(7,V): NEXT
    V
```

```
5300  GOSUB 500
5310 I = 4: GOSUB 1670
5320  PRINT : INPUT "APPUYEZ SUR RETURN POUR REVENIR AU
    MENU";Z$
5330  GOTO 100
```

c. Exemples commentés

● ***Exemple 1*** :

Soient les deux fractions rationnelles :

$$\begin{cases} F1\ (X) = \dfrac{X-2}{4X+3} \\ \\ F2\ (x) = \dfrac{x-1}{x+1} \end{cases}$$

on obtient alors :

$$F(x) = F1[F2(x)] = \frac{\dfrac{x-1}{x+1} - 2}{4\left(\dfrac{x-1}{x+1}\right) + 3} = \frac{-x-3}{7x-1}$$

```
    ENTREE DE F1


ENTREZ LE NUMERATEUR:

DEGRE 1
  COEFFICIENT DE X0 -2
  COEFFICIENT DE X1 1


ENTREZ LE DENOMINATEUR:

DEGRE 1
  COEFFICIENT DE X0 3
  COEFFICIENT DE X1 4
```

```
    ENTREE DE F2


ENTREZ LE NUMERATEUR:

DEGRE 1
  COEFFICIENT DE X0 -1
  COEFFICIENT DE X1 1


ENTREZ LE DENOMINATEUR:

DEGRE 1
  COEFFICIENT DE X0 1
  COEFFICIENT DE X1 1


    FRACTION SUBSTITUEE  F=F1(F2)


NUMERATEUR:

COEFF. DE X0:-3
COEFF. DE X1:-1


DENOMINATEUR:

COEFF. DE X0:-1
COEFF. DE X1:7
```

• ***Exemple 2*** :

Les remarques faites à propos de la substitution de polynômes restent valables ; il est en particulier possible d'élever la fraction F2 à la puissance n.

Il faut pour cela prendre $F1(X) = X^n$.

Alors :

$$F1[F2(x)] = [F2(x)]^n$$

Calculons par exemple :

$$F(x) = \left[\frac{\frac{1}{2}x}{x^2 - 5}\right]^3$$

```
    ENTREE DE F1


ENTREZ LE NUMERATEUR:

DEGRE 3
  COEFFICIENT DE X0 0
  COEFFICIENT DE X1 0
  COEFFICIENT DE X2 0
  COEFFICIENT DE X3 1


ENTREZ LE DENOMINATEUR:

DEGRE 0
  COEFFICIENT DE X0 1




    ENTREE DE F2


ENTREZ LE NUMERATEUR:

DEGRE 1
  COEFFICIENT DE X0 0
  COEFFICIENT DE X1 1/2


ENTREZ LE DENOMINATEUR:

DEGRE 2
  COEFFICIENT DE X0 -5
  COEFFICIENT DE X1 0
  COEFFICIENT DE X2 1




    FRACTION SUBSTITUEE  F=F1(F2)


NUMERATEUR:

COEFF. DE X3:1
```

```
DENOMINATEUR:

COEFF. DE X0:-1000
COEFF. DE X2:600
COEFF. DE X4:-120
COEFF. DE X6:8



    FRACTION SUBSTITUEE  F=F1(F2)

NUMERATEUR:

COEFF. DE X3:.125            1/8


DENOMINATEUR:

COEFF. DE X0:-125
COEFF. DE X2:75
COEFF. DE X4:-15
COEFF. DE X6:1
```

Le premier résultat est obtenu en mode rationnel, le second en mode réel.

7. *Dérivées successives d'une fraction*

a. Principe

Notre but est d'obtenir l'expression de la dérivée n-ième de la fraction rationnelle

$$F1 = \frac{P_0}{Q_0}$$

La dérivée première de F1 vaut:

$$F1' = \frac{P_0' Q_0 - P_0 Q_0'}{Q_0^2} = \frac{P_1}{Q_1}$$

Pour obtenir la dérivée seconde, on dérive F1':

$$F1'' = (F1')' = \frac{P_1'Q_1 - P_1Q_1'}{Q_1^2} = \frac{P_2}{Q_2}$$

De manière générale, si $F1^{(n)} = \frac{P_n}{Q_n}$, on obtient $F1^{(n+1)}$ par:

$$F1^{(n+1)} = \frac{P_n'Q_n - P_nQ_n'}{Q_n^2}$$

b. Programme

Après avoir introduit l'ordre n, le programme calcule la dérivée n-ième de F1 en appliquant les formules précédentes et transforme le résultat en une fraction irréductible.

La fraction initiale F1 n'est pas modifiée et le résultat est stocké dans F, ce qui permet d'effectuer le calcul pour différentes valeurs de n.

```
6000 CLS
6010  PRINT  TAB( 5);"F = DERIVEE N-IEME DE F1": PRINT :
     PRINT
6020  IF DP(1) = 0 AND P(1,0) = 0 THEN  PRINT "IMPOSSIBL
    E ... F1 N'EST PAS DEFINIE": PRINT : GOTO 6290
6030 DP(10) = DP(0): FOR I = 0 TO DP(0):P(10,I) = P(0,I)
    : NEXT I
6040 DP(11) = DP(1): FOR I = 0 TO DP(1):P(11,I) = P(1,I)
    : NEXT I
6050  INPUT "ORDRE DE LA DERIVEE ";N
6060  IF N < 1 GOTO 6290
6070  FOR TT = 1 TO N
6080 I = 1:J = 1:K = 5: GOSUB 400
6090  IF DP(0) = 0 THEN DP(8) = 0:P(8,0) = 0:DP(4) = 0:P
    (4,0) = 0: GOTO 6130
6100 DP(8) = DP(0) - 1
6110  FOR I = 0 TO DP(8):P(8,I) = P(0,I + 1) * (I + 1): NEXT
    I
6120 I = 8:J = 1:K = 4: GOSUB 400
6130  IF DP(1) = 0 THEN DP(9) = 0:P(9,0) = 0:DP(7) = 0:P
    (7,0) = 0: GOTO 6170
6140 DP(9) = DP(1) - 1
6150  FOR I = 0 TO DP(9):P(9,I) = P(1,I + 1) * (I + 1): NEXT
    I
```

```
6160 I = 0:J = 9:K = 7: GOSUB 400
6170  IF DP(7) > DP(4) THEN  FOR I = DP(4) + 1 TO DP(7):
    P(4,I) = 0: NEXT I:DP(4) = DP(7)
6180  FOR I = 0 TO DP(4):P(4,I) = P(4,I) - P(7,I): NEXT
    I
6190  IF P(4,DP(4)) <  > 0 OR DP(4) = 0 GOTO 6210
6200 DP(4) = DP(4) - 1: GOTO 6190
6210 RF = RE: IF TT < N THEN RE = 1
6220  GOSUB 500:RE = RF
6230 DP(0) = DP(4): FOR I = 0 TO DP(0):P(0,I) = P(4,I): NEXT
    I
6240 DP(1) = DP(5): FOR I = 0 TO DP(1):P(1,I) = P(5,I): NEXT
    I
6250  NEXT TT
6260 DP(0) = DP(10): FOR I = 0 TO DP(0):P(0,I) = P(10,I)
    : NEXT I
6270 DP(1) = DP(11): FOR I = 0 TO DP(1):P(1,I) = P(11,I)
    : NEXT I
6280 I = 4: GOSUB 1670
6290  PRINT : INPUT "VOULEZ-VOUS REUTILISER LE PROGRAMME
     ";Z$
6300  IF Z$ = "O" GOTO 6000
6310  GOTO 100
```

REMARQUE : les calculs étant nombreux, l'exécution peut atteindre plusieurs minutes dans certains cas.

c. Exemples commentés

● ***Exemple 1*** :

La dérivée de la fonction $F1\,(x) = \dfrac{3x^2 + 2x + 1}{4x^2 + 5x + 56}$ est égale à :

$$F1'\,(x) = \frac{7x^2 + 328x + 107}{16x^4 + 40x^3 + 473x^2 + 560x + 3136}$$

Le programme confirme ce résultat:

```
    ENTREE DE F1


ENTREZ LE NUMERATEUR:

DEGRE 2
  COEFFICIENT DE X0 1
  COEFFICIENT DE X1 2
  COEFFICIENT DE X2 3


ENTREZ LE DENOMINATEUR:

DEGRE 2
  COEFFICIENT DE X0 56
  COEFFICIENT DE X1 5
  COEFFICIENT DE X2 4



    F = DERIVEE N-IEME DE F1


ORDRE DE LA DERIVEE 1
NUMERATEUR:

COEFF. DE X0:107
COEFF. DE X1:328
COEFF. DE X2:7


DENOMINATEUR:

COEFF. DE X0:3136
COEFF. DE X1:560
COEFF. DE X2:473
COEFF. DE X3:40
COEFF. DE X4:16
```

• ***Exemple 2*** :

Voici un exemple de calcul de dérivée cinquième en mode rationnel :

```
    ENTREE DE F1


ENTREZ LE NUMERATEUR:

DEGRE 1
  COEFFICIENT DE X0 -1
  COEFFICIENT DE X1 1


ENTREZ LE DENOMINATEUR:

DEGRE 1
  COEFFICIENT DE X0 1
  COEFFICIENT DE X1 1



    F = DERIVEE N-IEME DE F1


ORDRE DE LA DERIVEE 5
NUMERATEUR:

COEFF. DE X0:240


DENOMINATEUR:

COEFF. DE X0:1
COEFF. DE X1:6
COEFF. DE X2:15
COEFF. DE X3:20
COEFF. DE X4:15
COEFF. DE X5:6
COEFF. DE X6:1
```

Un représentant de la dérivée cinquième de $f(x) = \dfrac{x-1}{x+1}$ est donc :

$$f^{(5)}(x) = \frac{240}{x^6 + 6x^5 + 15x^4 + 20x^3 + 15x^2 + 6x + 1}$$

d. Phénomènes rencontrés lors de la recherche de dérivées d'ordre important

Lors de chaque dérivation, le programme fait appel au sous-programme de simplification de fraction, ce qui introduit des erreurs numériques. Ces erreurs s'amplifient à chaque dérivation supplémentaire, et on assiste parfois aux deux phénomènes suivants.

● La précision sur les coefficients correspondant à de faibles puissances de x devient moins bonne, ce qui empêche un calcul en mode rationnel.

Calculons par exemple en mode réel la dérivée d'ordre 7 de $\dfrac{x-1}{x+1}$:

```
    F = DERIVEE N-IEME DE F1


ORDRE DE LA DERIVEE 7
NUMERATEUR:

COEFF. DE X0:1


DENOMINATEUR:

COEFF. DE X0:9.92011467E-05  2/20161
COEFF. DE X1:7.93653772E-04  1/1260
COEFF. DE X2:2.7777773E-03   1/360
COEFF. DE X3:5.55555553E-03  1/180
COEFF. DE X4:6.94444449E-03  1/144
COEFF. DE X5:5.55555554E-03  1/180
COEFF. DE X6:2.77777779E-03  1/360
COEFF. DE X7:7.93650794E-04  1/1260
COEFF. DE X8:9.92063492E-05  1/10080
```

Le coefficient de x^0 n'est pas $\dfrac{2}{20161}$ mais $\dfrac{1}{10080}$.

Le sous-programme de recherche d'approximations rationnelles n'est pas à mettre en cause, la précision de ce coefficient était insuffisante.

Ainsi, si l'on tentait le calcul en mode rationnel, le résultat fourni par le programme serait inexploitable.

- Le second phénomène est plus gênant.

Mettons le en évidence en effectuant la recherche des dérivées d'ordre 4 et 5 de

$$f(x) = \frac{x-2}{x-3}.$$

```
    ENTREE DE F1


ENTREZ LE NUMERATEUR:

DEGRE 1
  COEFFICIENT DE X0 -2
  COEFFICIENT DE X1 1


ENTREZ LE DENOMINATEUR:

DEGRE 1
  COEFFICIENT DE X0 -3
  COEFFICIENT DE X1 1




    F = DERIVEE N-IEME DE F1


ORDRE DE LA DERIVEE 4
NUMERATEUR:

COEFF. DE X0:1


DENOMINATEUR:

COEFF. DE X0:-10.125          -81/8
COEFF. DE X1:16.875           135/8
```

```
COEFF. DE X2:-11.25          -45/4
COEFF. DE X3:3.75            15/4
COEFF. DE X4:-.625           -5/8
COEFF. DE X5:.0416666667     1/24



    F = DERIVEE N-IEME DE F1


ORDRE DE LA DERIVEE 5
NUMERATEUR:

COEFF. DE X0:-16.875         -135/8
COEFF. DE X1:22.5            45/2
COEFF. DE X2:-11.25          -45/4
COEFF. DE X3:2.5             5/2
COEFF. DE X4:-.208333333     -5/24


DENOMINATEUR:

COEFF. DE X0:102.515625      6561/64
COEFF. DE X1:-341.71875      -10935/32
COEFF. DE X2:512.578125      23066/45
COEFF. DE X3:-455.625        -3645/8
COEFF. DE X4:265.78125       8505/32
COEFF. DE X5:-106.3125       -1701/16
COEFF. DE X6:29.53125        945/32
COEFF. DE X7:-5.625          -45/8
COEFF. DE X8:.703125         45/64
COEFF. DE X9:-.0520833333    -5/96
COEFF. DE X10:1.73611111E-03 1/576
```

La dérivée d'ordre 4 est exacte.

Par contre, le numérateur de la dérivée d'ordre 5, qui devrait être une constante, est un polynôme de degré 4 ! Le résultat n'est pas faux, mais le sous-programme de réduction n'a pas fonctionné correctement.

En effet, pour parvenir à un représentant irréductible, le programme calcule d'abord le PGCD du numérateur et du dénominateur de la fraction. Nous avons vu dans le chapitre précédent que le calcul de PGCD de polynômes ne fournissait pas de résultats satisfaisants lorsque les degrés des polynômes mis en jeu étaient importants.

Ainsi, le résultat obtenu est bien un représentant de $(\frac{x-2}{x-3})^{(5)}$, mais il est encore réductible.

Ce phénomène ne se rencontre que rarement lors du calcul de dérivée première sur les fractions rationnelles courantes.

8. Décomposition en éléments simples

a. Rappels théoriques et principe

Considérons une fraction rationnelle $F = \frac{P}{Q}$ dont le dénominateur est écrit sous la forme d'un produit de polynômes irréductibles.

Rappelons qu'un polynôme à coefficients réels irréductible est soit de degré 1, soit de degré 2 mais ne possédant aucune racine.

Le polynôme Q(x) doit donc se présenter sous la forme :

$$Q(x) = [H_1(x)]^{\alpha_1} . [H_2(x)]^{\alpha_2} \dots [H_f(x)]^{\alpha_f}$$

où les polynômes $H_f(x)$ sont irréductibles.

On peut alors démontrer qu'il existe une décomposition unique de la fraction F, de la forme :

$$F(x) = E(x) + B_1(x) + B_2(x) + \dots + B_f(x)$$

où E(x) est un polynôme appelé partie entière de F(x) et les termes $B_i(x)$ sont de la forme :

$$B_i(x) = \sum_{j=1}^{\alpha_i} \frac{L_{ij}(x)}{[H_i(x)]^j}$$

où le polynôme $L_{ij}(x)$ est de degré strictement inférieur à celui de H_i.

Ainsi, pour les éléments de "première espèce", c'est à dire les polynômes de degré 1, $L_{ij}(x)$ est forcément un scalaire a_{ij}.

Pour les éléments de "deuxième espèce", L_{ij} est un polynôme de degré inférieur ou égal à 1, et s'exprime donc sous la forme $a_{ij} + b_{ij}x$.

Considérons par exemple la fraction :

$$F(x) = \frac{2x^9 + 1}{x^3 (x^2 + x + 1)^2}$$

Les polynômes irréductibles sont :

$$H_1 (x) = x \qquad (\alpha_1 = 3)$$
$$H_2 (x) = x^2 + x + 1 \qquad (\alpha_2 = 2)$$

La théorie nous permet donc d'affirmer que la fraction peut s'écrire de manière unique sous la forme :

$$F(x) = E(x) + \underbrace{\frac{a_{11}}{x} + \frac{a_{12}}{x^2} + \frac{a_{13}}{x^3}}_{B_1 (x)} + \underbrace{\frac{a_{21} + b_{21} x}{x^2 + x + 1} + \frac{a_{22} + b_{22} x}{(x^2 + x + 1)^2}}_{B_2 (x)}$$

Décomposer la fraction F(x) en éléments simples, c'est trouver la partie entière E(x) ainsi que l'ensemble des coefficients a_{ij} et b_{ij} correspondant aux différents facteurs.

b. Programme

Ce programme, contrairement aux autres, n'opère pas sur une des fractions F1 ou F2, mais demande directement la fraction à décomposer, qui doit être introduite de la manière suivante.

Le numérateur est entré de la manière habituelle : d'abord le degré, puis l'ensemble des coefficients.

Le dénominateur doit par contre déjà être décomposé en produit de facteurs irréductibles. Chacun des facteurs doit être entré séparément, en donnant au programme le degré du facteur (1 ou 2), les coefficients (2 ou 3 selon l'ordre), la puissance du facteur.

Lorsque tous les facteurs ont été introduits, il suffit de répondre Ø quand le programme demande le degré du facteur suivant.

L'ordinateur commence alors la recherche : après quelques vérifications sur la validité des données introduites (en particulier, irréductibilité de la fraction), il identifie la fraction initiale à la somme d'éléments simples devant être obtenus pour plusieurs valeurs de x.

Il obtient ainsi un système linéaire de plusieurs équations, résolu par la méthode de Gauss-Jordan qui est présentée dans le chapitre "Matrices et vecteurs"

Lorsque les calculs sont terminés, le programme affiche la partie entière ainsi que l'ensemble des coefficients a_{ij} et b_{ij}, accompagnés de leur approximation rationnelle.

```
7000 CLS
7010  PRINT  TAB( 3);"DECOMPOSITION EN ELEMENTS SIMPLES"

7020 I = 1: PRINT : PRINT :NN = 0
7030  PRINT "ENTREZ LE NUMERATEUR:"
7040  PRINT : INPUT "DEGRE ";DP(9)
7050  IF DP(9) < 0 OR DP(9) > 10 OR DP(9) <  >  INT (DP(
    9)) GOTO 7040
7060  FOR J = 0 TO DP(9)
7070  PRINT "  COEFFICIENT DE X";J;
7080  INPUT Z$: GOSUB 1900
7090  IF ER = 1 GOTO 7070
7100 P(9,J) = N
7110  IF J = DP(9) AND P(9,J) = 0 THEN  PRINT "IMPOSSIBL
    E...": GOTO 7070
7120  NEXT J
7130  PRINT : PRINT "FACTEUR ";I;":"
7140  PRINT " DEGRE DU FACTEUR ";I;: INPUT PR(I)
7150  IF PR(I) > 2 OR PR(I) < 0 GOTO 7140
7160  IF PR(I) = 0 GOTO 7340
7170  FOR J = 0 TO PR(I)
7180  PRINT "  COEFFICIENT DE X";J;: INPUT Z$
7190  GOSUB 1900: IF ER = 1 GOTO 7180
7200 DN(I,J) = N
7210  NEXT J
7220  IF DN(I,PR(I)) = 0 GOTO 7140
7230  IF PR(I) = 2 AND DN(I,1) * DN(I,1) >  = 4 * DN(I,0
    ) * DN(I,2) THEN  PRINT "POLYNOME REDUCTIBLE...": GOTO
    7130
7240 J = 1
7250  IF I = J GOTO 7280
7260  IF DN(I,0) = DN(J,0) AND DN(I,1) = DN(J,1) AND DN(
    I,2) = DN(J,2) AND PR(I) = PR(J) THEN  PRINT "FACTEU
    R DEJA ENTRE...": GOTO 7130
7270 J = J + 1: GOTO 7250
7280  PRINT " PUISSANCE DU FACTEUR ";I;: INPUT PN(I)
7290  IF PN(I) <  = 0 GOTO 7280
7300 PN(I) =  INT (PN(I)):U = NN + PR(I) * PN(I)
7310  IF U >  = MX THEN  PRINT "DENOMINATEUR TROP VOLUMI
    NEUX...": GOTO 7130
7320 NN = U
```

```
7330  IF I < 10 THEN I = I + 1: GOTO 7130
7340 FF = I - 1:N = NN:GG = 0
7350  IF FF = 0 GOTO 7960
7360  FOR G = 1 TO FF
7370 DP(8) = PR(G)
7380  FOR V = 0 TO DP(8):P(8,V) = DN(G,V): NEXT V
7390 I = 9:J = 8:K = 10:L = 11: GOSUB 900
7400  IF DP(11) = 0 AND  ABS (P(11,0)) < 1E - 7 THEN  PRINT
    : PRINT "LA FRACTION EST REDUCTIBLE...":GG = G
7410  NEXT G
7420  IF GG > 0 GOTO 7960
7430 DP(8) = 0:P(8,0) = 1
7440  FOR G = 1 TO FF
7450  FOR U = 1 TO PN(G)
7460 DP(11) = PR(G): FOR K = 0 TO DP(11):P(11,K) = DN(G,
    K): NEXT K
7470 I = 8:J = 11:K = 10: GOSUB 400
7480 DP(8) = DP(10): FOR K = 0 TO DP(8):P(8,K) = P(10,K)
    : NEXT K
7490  NEXT U
7500  NEXT G
7510 I = 9:J = 8:K = 10:L = 11
7520  GOSUB 900: PRINT
7530  IF DP(10) = 0 AND P(10,0) = 0 THEN  PRINT : PRINT
    "LA PARTIE ENTIERE EST NULLE": GOTO 7560
7540  PRINT "PARTIE ENTIERE:"
7550 L = 10: GOSUB 1800
7560 HX = .5:X =  - N * HX / 2:I = 0
7570 K = 0:J = 0
7580 K = K + 1
7590 U = DN(K,1) * X + DN(K,0)
7600  IF U = 0 AND PR(K) = 1 THEN X = X + HX: GOTO 7570
7610  IF K < FF GOTO 7580
7620 I = I + 1
7630  FOR K = 1 TO FF
7640 U = 0:V = 1:W = 0
7650  FOR H = DP(8) TO 0 STEP  - 1:W = W * X + P(8,H): NEXT
    H
7660  FOR H = PR(K) TO 0 STEP  - 1:U = U * X + DN(K,H): NEXT
    H
7670  FOR H = 1 TO PN(K)
7680 V = V * U
7690 J = J + 1:F(I,J) = W / V
7700  IF PR(K) = 2 THEN J = J + 1:F(I,J) = X * W / V
7710  NEXT H
```

```
7720  NEXT K
7730 U = 0
7740  FOR H = DP(11) TO 0 STEP  - 1:U = U * X + P(11,H):
     NEXT H
7750 F(I,N + 1) = U
7760  IF I < N THEN X = X + HX: GOTO 7570
7770  GOSUB 9000
7780  IF CR <  > 0 THEN  PRINT "SYSTEME IRRESOLVABLE..."
    : GOTO 7960
7790 I = 0: PRINT
7800  FOR G = 1 TO FF
7810  PRINT : PRINT "TERMES RELATIFS AU FACTEUR ";G
7820  FOR K = 1 TO PN(G)
7830 I = I + 1:X = F(I,N + 1): IF  ABS (X) < 1E - 6 THEN
    X = 0
7840  GOSUB 300: IF DG = 1 THEN NN = AD *  SGN (X):DD =
    AF
7850  PRINT " A(";G;",";K;")=";NN / DD;
7860  IF DD > 1 AND NN <  > 0 THEN  PRINT  TAB( 27);NN;"
    /";DD;
7870  PRINT
7880  IF PR(G) = 1 GOTO 7940
7890 I = I + 1:X = F(I,N + 1): IF  ABS (X) < 1E - 6 THEN
    X = 0
7900  GOSUB 300: IF DG = 1 THEN NN = AD *  SGN (X):DD =
    AF
7910  PRINT " B(";G;",";K;")=";NN / DD;
7920  IF DD > 1 AND NN <  > 0 THEN  PRINT  TAB( 27);NN;"
    /";DD;
7930  PRINT : PRINT
7940  NEXT K
7950  NEXT G
7960  PRINT : INPUT "VOULEZ-VOUS REUTILISER LE PROGRAMME
     ";Z$
7970  IF Z$ = "O" GOTO 7000
7980  GOTO 100
9000 CR = 0
9010  FOR K = 1 TO N
9020 I = K
9030  IF I > N THEN CR = 2:K = N: GOTO 9100
9040  IF  ABS (F(I,K)) < 1E - 8 THEN I = I + 1: GOTO 903
    0
9050  IF I > K THEN  FOR J = 1 TO N + 1:M = F(I,J):F(I,J
    ) = F(K,J):F(K,J) = M: NEXT J
9060  FOR I = K + 1 TO N + 1:F(K,I) = F(K,I) / F(K,K): NEXT
    I:F(K,K) = 1
```

```
9070  FOR I = 1 TO N
9080  IF I <  > K THEN  FOR J = K + 1 TO N + 1:F(I,J) =
    F(I,J) - F(I,K) * F(K,J): NEXT J:F(I,K) = 0
9090  NEXT I
9100  NEXT K
9110  RETURN
```

c. Exemples commentés

• ***Exemple 1*** :

Considérons la décomposition :

$$\frac{1}{(x-1)(x+1)} = \frac{(\frac{1}{2})}{x-1} + \frac{(-\frac{1}{2})}{x+1}$$

La partie entière est nulle dans cet exemple.

```
  DECOMPOSITION EN ELEMENTS SIMPLES


ENTREZ LE NUMERATEUR:

DEGRE 0
  COEFFICIENT DE X0?1

FACTEUR 1:
 DEGRE DU FACTEUR 1?1
  COEFFICIENT DE X0?-1
  COEFFICIENT DE X1?1
 PUISSANCE DU FACTEUR 1?1

FACTEUR 2:
 DEGRE DU FACTEUR 2?1
  COEFFICIENT DE X0?1
  COEFFICIENT DE X1?1
 PUISSANCE DU FACTEUR 2?1

FACTEUR 3:
 DEGRE DU FACTEUR 3?0
```

```
LA PARTIE ENTIERE EST NULLE


TERMES RELATIFS AU FACTEUR 1
 A(1,1)=.5                    1/2

TERMES RELATIFS AU FACTEUR 2
 A(2,1)=-.5                   -1/2

VOULEZ-VOUS REUTILISER LE PROGRAMME O
```

- ***Exemple 2*** :

Reprenons la fraction présentée dans le paragraphe précédent :

$$F(x) = \frac{1 + 2x^9}{x^3\,(x^2 + x + 1)^2}$$

```
  DECOMPOSITION EN ELEMENTS SIMPLES


ENTREZ LE NUMERATEUR:

DEGRE 9
  COEFFICIENT DE X0?1
  COEFFICIENT DE X1?0
  COEFFICIENT DE X2?0
  COEFFICIENT DE X3?0
  COEFFICIENT DE X4?0
  COEFFICIENT DE X5?0
  COEFFICIENT DE X6?0
  COEFFICIENT DE X7?0
  COEFFICIENT DE X8?0
  COEFFICIENT DE X9?2

FACTEUR 1:
 DEGRE DU FACTEUR 1?1
  COEFFICIENT DE X0?0
  COEFFICIENT DE X1?1
 PUISSANCE DU FACTEUR 1?3

FACTEUR 2:
 DEGRE DU FACTEUR 2?2
  COEFFICIENT DE X0?1
```

```
  COEFFICIENT DE X1?1
  COEFFICIENT DE X2?1
 PUISSANCE DU FACTEUR 2?2

FACTEUR 3:
 DEGRE DU FACTEUR 3?0

PARTIE ENTIERE:

COEFF. DE X0:2
COEFF. DE X1:-4
COEFF. DE X2:2



TERMES RELATIFS AU FACTEUR 1
 A(1,1)=1
 A(1,2)=-2
 A(1,3)=1

TERMES RELATIFS AU FACTEUR 2
 A(2,1)=-3
 B(2,1)=3

 A(2,2)=3
 B(2,2)=0


VOULEZ-VOUS REUTILISER LE PROGRAMME O
```

La décomposition de F(x) est donc la suivante :

$$F(x) = \underbrace{(2x^2 - 4x + 2)}_{\text{partie entière}} + \underbrace{\left(\frac{1}{x} + \frac{(-2)}{x^2} + \frac{1}{x^3}\right)}_{\substack{\text{partie relative}\\ \text{au premier facteur}}} + \underbrace{\left(\frac{(-3) + 3x}{x^2 + x + 1} + \frac{3 + 0x}{(x^2 + x + 1)^2}\right)}_{\substack{\text{partie relative}\\ \text{au deuxième facteur}}}$$

• ***Exemple 3*** :

L'exemple suivant montre que les approximations rationnelles sont bien utiles lorsque le nombre réel correspondant n'est pas immédiatement identifiable.

```
  DECOMPOSITION EN ELEMENTS SIMPLES


ENTREZ LE NUMERATEUR:

DEGRE 4
  COEFFICIENT DE X0?3
  COEFFICIENT DE X1?-4
  COEFFICIENT DE X2?-4
  COEFFICIENT DE X3?-4
  COEFFICIENT DE X4?2

FACTEUR 1:
 DEGRE DU FACTEUR 1?1
  COEFFICIENT DE X0?-3
  COEFFICIENT DE X1?1
 PUISSANCE DU FACTEUR 1?1

FACTEUR 2:
 DEGRE DU FACTEUR 2?2
  COEFFICIENT DE X0?1
  COEFFICIENT DE X1?1
  COEFFICIENT DE X2?1
 PUISSANCE DU FACTEUR 2?1

FACTEUR 3:
 DEGRE DU FACTEUR 3?0

PARTIE ENTIERE:

COEFF. DE X1:2




TERMES RELATIFS AU FACTEUR 1
 A(1,1)=.692307692        9/13

TERMES RELATIFS AU FACTEUR 2
 A(2,1)=-.769230769       -10/13
 B(2,1)=-.692307692       -9/13


VOULEZ-VOUS REUTILISER LE PROGRAMME O
```

La fraction s'écrit donc :

$$\frac{3-4x-4x^2-4x^3+2x^4}{(x-3)(x^2+x+1)} = 2x + \frac{\frac{9}{13}}{x-3} + \frac{(-\frac{10}{13})+(-\frac{9}{13})x}{x^2+x+1}$$

Les coefficients sont exacts dans les trois exemples précédents. En fait, la précision des coefficients est meilleure que 10^{-6} et dans ce cas, le polynôme arrondit les coefficients afin d'améliorer la présentation.

Les ordres respectifs des systèmes linéaires résolus par le programme sont 2, 7 et 3. En effet, l'ordre du système est égal à la somme des puissances de x les plus élevées du dénominateur. Tant que celui-ci reste inférieur à 10, la précision est généralement très bonne.

Examinons l'exemple suivant, pour lequel l'ordre du système est 14 :

$$F(x) = \frac{x+2}{(x^2+x+1)^7}$$

Les coefficients théoriques sont tous nuls sauf $a_{1,7} = 2$ et $b_{1,7} = 1$. Voici les résultats fournis par le programme.

```
  DECOMPOSITION EN ELEMENTS SIMPLES


ENTREZ LE NUMERATEUR:

DEGRE 1
  COEFFICIENT DE X0?2
  COEFFICIENT DE X1?1

FACTEUR 1:
 DEGRE DU FACTEUR 1?2
  COEFFICIENT DE X0?1
  COEFFICIENT DE X1?1
  COEFFICIENT DE X2?1
 PUISSANCE DU FACTEUR 1?7

FACTEUR 2:
 DEGRE DU FACTEUR 2?0


LA PARTIE ENTIERE EST NULLE
```

```
TERMES RELATIFS AU FACTEUR 1
 A(1,1)=0
 B(1,1)=0

 A(1,2)=-2.05337966E-06    -1/487002
 B(1,2)=8.48226782E-06     1/117893

 A(1,3)=1.00261784E-06     1/997389
 B(1,3)=-9.44666179E-05    -4/42343

 A(1,4)=8.16486495E-05     5/61238
 B(1,4)=4.99906267E-04     8/16003

 A(1,5)=-3.67937641E-04    -27/73382
 B(1,5)=-1.29871817E-03    -100/76999

 A(1,6)=5.62551387E-04     13/23109
 B(1,6)=1.55925156E-03     3/1924

 A(1,7)=1.99972467         7263/3632
 B(1,7)=.999325691         1482/1483


VOULEZ-VOUS REUTILISER LE PROGRAMME N
```

La précision moyenne est de l'ordre de 10^{-4}, et naturellement, les approximations rationnelles n'ont ici aucune signification, puisque les coefficients exacts sont nuls.

Nous avons donc limité l'ordre du système à résoudre à 15, de façon à conserver une précision et un temps de calcul acceptables. Ainsi, si l'on essaie d'introduire une fraction trop compliquée, le programme la refuse en affichant "dénominateur trop volumineux".

Il reste cependant possible de modifier l'ordre maximal du système à résoudre à la ligne 30; il faut alors être très prudent quant à l'interprétation des résultats.

On peut améliorer la précision de manière sensible lors de la décomposition de fractions donnant lieu à des systèmes d'ordre élevé, en ajustant la valeur HX à la ligne 7560; généralement, une diminution de HX améliore la précision (HX = 0,25 ou 0,2, voire 0,1 dans certains cas).

9. Conclusion

Nous avons présenté dans ce chapitre un large éventail d'outils permettant le calcul, et surtout la vérification, des opérations les plus courantes que l'on peut être amené à effectuer sur les fractions rationnelles.

Pour certains exemples mettant en œuvre des fractions de degré élevé, les calculs sont relativement lents à cause du nombre important d'opérations effectuées (les temps d'exécution peuvent atteindre plusieurs minutes). Généralement, l'exécution reste tout de même beaucoup plus rapide et beaucoup plus sûre qu'à la main !

La principale difficulté est la nécessité de réduire la fraction obtenue à un représentant irréductible. Nous avons vu que cette opération impliquait un calcul de PGCD de polynômes, et que le calcul de PGCD était relativement imprécis.

La précision des coefficients eux-mêmes reste par contre excellente dans la plupart des cas.

6 Matrices et vecteurs

1. Introduction

Les calculs sur les matrices et les vecteurs sont extrêmement fréquents dans tous les domaines de la physique. De nombreux problèmes d'optique, de mécanique ou d'électromagnétisme conduisent à des systèmes linéaires ou à des recherches de valeurs propres, dont le résolution est le plus souvent effectuée numériquement (citons par exemple la méthode des éléments finis, très utilisée en aéronautique pour le calcul des écoulements et des surpressions).

Les programmes présentés dans ce chapitre permettront d'effectuer des opérations élémentaires sur les vecteurs (combinaisons linéaires, produits scalaire, mixte et vectoriel) et sur les matrices (transposition, combinaisons linéaires, produit) mais également des opérations plus complexes telles que le calcul du déterminant, du rang, des valeurs propres et de l'inverse de la matrice A ainsi que l'élévation de A à la puissance n.

On trouvera également la résolution des systèmes linéaires ainsi que le calcul de l'image d'un vecteur par une application linéaire, définie par sa matrice.

La présentation générale du chapitre est semblable à celle du chapitre précédent concernant les fractions rationnelles; en particulier, deux parties intitulées "Manipulations sur les vecteurs" et "Manipulations sur les matrices" sont destinées à faciliter la saisie et la visualisation des données et résultats.

Les programmes sont indépendants et peuvent être entrés dans un ordre quelconque, à la condition d'avoir introduit le menu principal ci-dessous et les deux programmes d'entrée et de manipulation de matrices et de vecteurs.

Le listing ci-dessous contient également des sous-programmes d'usage général, et qui doivent donc être introduits avant tout autre programme.

```
10 MX = 15
20  DIM A(MX,MX),B(MX,MX),C(MX,MX),D(MX,2 * MX),VE(3,MX)

30  DIM DM(4,2),MM$(3),DV(3),VV$(3)
40 MM$(1) = "A":MM$(2) = "B":MM$(3) = "C"
50 VV$(1) = "U":VV$(2) = "V":VV$(3) = "W"
60 PR = 1E - 8
100 CLS
110  PRINT  TAB( 5);"MATRICES ET VECTEURS": PRINT : PRINT

120  PRINT "1- MANIPULATIONS SUR LES VECTEURS": PRINT
130  PRINT "2- MANIPULATIONS SUR LES MATRICES": PRINT
140  PRINT "3- CALCUL VECTORIEL": PRINT
150  PRINT "4- CALCUL MATRICIEL": PRINT
160  PRINT "5- CALCUL DE V=AU": PRINT
170  PRINT "6- RESOLUTION DU SYSTEME AX=U": PRINT
180  PRINT "7- FIN": PRINT
190  INPUT "VOTRE CHOIX ";E
200  ON E GOTO 1000,2000,3000,3500,4000,8000,9000
210  GOTO 190
300  IF X = 0 THEN NN = 0:DD = 1: GOTO 390
310 AX =  ABS (X):AB = AX:AC =  INT (AB):NN = AC
320 DD = 1:AD = 1:AF = 0: GOTO 370
330 AB = 1 / (AB - AC):AC =  INT (AB)
340 AY = AC * NN + AD:AD = NN:NN = AY
350 AY = AC * DD + AF:AF = DD:DD = AY
360  IF (DD > 1E5 OR DD * NN > 1E9) AND NN > 1 THEN NN =
    AD:DD = AF: GOTO 380
370  IF 1E6 *  ABS (AX * DD - NN) > DD * AX GOTO 330
380 NN = NN *  SGN (X)
390  RETURN
500 DR = 1:RG = N
510  FOR K = 1 TO N
520 I = K
```

```
530  IF I > N THEN DR = 0:RG = RG - 1: GOTO 610
540  IF  ABS (D(I,K)) < PR THEN I = I + 1:DR =  - DR: GOTO
    530
550  IF I > K THEN  FOR J = 1 TO N + P:U = D(I,J):D(I,J)
     = D(K,J):D(K,J) = U: NEXT J
560  FOR I = K + 1 TO N + P:D(K,I) = D(K,I) / D(K,K): NEXT
    I
570 DR = DR * D(K,K):D(K,K) = 1
580  FOR I = 1 TO N
590  IF I <  > K THEN  FOR J = K + 1 TO N + P:D(I,J) = D
    (I,J) - D(I,K) * D(K,J): NEXT J:D(I,K) = 0
600  NEXT I
610  NEXT K
620  RETURN
800 ER = 0:LZ =  LEN (Z$):FD = 0
810  IF LZ = 0 THEN ER = 1: GOTO 890
820  FOR K = 1 TO LZ
830  IF  MID$ (Z$,K,1) = "/" THEN FD = K:K = LZ
840  NEXT K
850  IF FD = 0 THEN NN =  VAL (Z$): GOTO 890
860 BB =  VAL ( MID$ (Z$,FD + 1,LZ - FD))
870  IF BB = 0 THEN ER = 1: GOTO 890
880 NN =  VAL ( MID$ (Z$,1,FD - 1)) / BB
890  RETURN
900  IF L = 1 THEN U = A(I,J)
910  IF L = 2 THEN U = B(I,J)
920  IF L = 3 THEN U = C(I,J)
930  IF L = 4 THEN U = D(I,J)
940  RETURN
950  IF M = 1 THEN A(I,J) = U
960  IF M = 2 THEN B(I,J) = U
970  IF M = 3 THEN C(I,J) = U
980  IF M = 4 THEN D(I,J) = U
990  RETURN
9000  END
```

Le programme complet nécessite environ 18 Koctets de mémoire, dont 7 Koctets pour les variables.

2. Manipulations sur les vecteurs

a. Principe

Les programmes effectuant des calculs vectoriels utilisent trois vecteurs U, V et W.

Le programme "Manipulations sur les vecteurs" permet l'introduction, le rappel de U, V et W, ainsi que tous les transferts possibles de l'un vers l'autre.

b. Programme

Il y a donc trois types de fonctions:

- *Entrée d'un vecteur*:

Le programme demande tout d'abord la dimension du vecteur, qui doit être inférieure à 15, puis la suite de ses coordonnées.

- *Rappel d'un vecteur*:

Le programme affiche les coordonnées du vecteur choisi.

- *Transfert d'un vecteur dans un autre*:

Les coordonnées du vecteur d'arrivée sont celles du vecteur de départ, celui-ci n'étant pas affecté par le transfert.

```
1000 CLS
1010  PRINT  TAB( 5);"MANIPULATIONS SUR LES VECTEURS": PRINT

1020  PRINT "1- ENTREE DE U"
1030  PRINT "2- ENTREE DE V"
1040  PRINT "3- ENTREE DE W": PRINT
1050  PRINT "4- RAPPEL DE U"
1060  PRINT "5- RAPPEL DE V"
1070  PRINT "6- RAPPEL DE W": PRINT
1080  PRINT "7- TRANSFERT DE U DANS V"
1090  PRINT "8- TRANSFERT DE V DANS U": PRINT
1100  PRINT "9- TRANSFERT DE U DANS W"
1110  PRINT "10- TRANSFERT DE W DANS U": PRINT
1120  PRINT "11- TRANSFERT DE V DANS W"
1130  PRINT "12- TRANSFERT DE W DANS V": PRINT
1140  PRINT "13- RETOUR AU MENU PRINCIPAL": PRINT
```

```
1150  INPUT "VOTRE CHOIX ";E
1160 CLS
1170  IF E < 1 OR E > 12 GOTO 100
1180  ON  INT ((E - 1) / 3 + 1) GOSUB 1300,1500,1600,160
   0
1190  PRINT : INPUT "APPUYEZ SUR RETURN POUR REVENIR AU
   MENU";Z$
1200  GOTO 1000
1300 NV = E
1310  PRINT  TAB( 5);"ENTREE DE ";VV$(NV): PRINT
1320  PRINT "DIMENSION DU VECTEUR ";VV$(NV);: INPUT " ";
   E
1330  IF E < 1 OR E > MX GOTO 1420
1340 DV(NV) = E: PRINT
1350  PRINT "COORDONNEES DU VECTEUR ";VV$(NV);" :"
1360  FOR I = 1 TO DV(NV)
1370  PRINT  TAB( 5);VV$(NV);"(";I;") ";: INPUT Z$
1380  GOSUB 800
1390  IF ER = 1 GOTO 1370
1400 VE(NV,I) = NN
1410  NEXT I
1420  RETURN
1500  PRINT  TAB( 10);"RAPPEL DE ";VV$(E - 3): PRINT
1510 NV = E - 3
1520  IF DV(NV) = 0 THEN  PRINT "LE VECTEUR ";VV$(NV);"
   N'EST PAS DEFINI...": RETURN
1530  FOR I = 1 TO DV(NV)
1540 X = VE(NV,I): GOSUB 300
1550  PRINT VV$(NV);" (";I;") =";X;
1560  IF DD > 1 THEN  PRINT  TAB( 26);NN;"/";DD;
1570  PRINT
1580  NEXT I
1590  RETURN
1600  IF E = 7 THEN VD = 1:VA = 2
1610  IF E = 8 THEN VD = 2:VA = 1
1620  IF E = 9 THEN VD = 1:VA = 3
1630  IF E = 10 THEN VD = 3:VA = 1
1640  IF E = 11 THEN VD = 2:VA = 3
1650  IF E = 12 THEN VD = 3:VA = 2
1660 DV(VA) = DV(VD)
1670  FOR I = 1 TO DV(VD)
1680 VE(VA,I) = VE(VD,I)
1690  NEXT I
1700 E = 3 + VA
1710  GOSUB 1500
1720  RETURN
```

3. *Manipulations sur les matrices*

a. Principe

De la même manière que pour les vecteurs, les programmes effectuant des calculs matriciels travaillent sur trois matrices A, B et C.

Les opérandes doivent être introduits dans A et éventuellement B, et le résultat est placé dans C.

Le programme permet l'introduction et le rappel des matrices, ainsi que les différents transferts possibles entre les matrices, ceci afin de permettre le chaînage des opérations. De plus, les matrices peuvent être transposées (les lignes de la matrice transposée sont les colonnes de la matrice de départ).

b. Programme

Le programme réalise donc quatre fonctions différentes :

- **Entrée d'une matrice** :

Il faut tout d'abord préciser les deux dimensions de la matrice (nombre de lignes et nombre de colonnes) ; ces deux nombres doivent être inférieurs ou égaux à 15.

Le programme demande ensuite le "Numéro de la colonne à entrer".

Ceci permet l'introduction de la matrice colonne par colonne, de manière à faciliter la correction d'éventuelles erreurs sans avoir à réintroduire l'intégralité des coefficients.

Après avoir donné le numéro J de la colonne sélectionnée, il faut entrer tous les coefficients de la colonne en commençant par le haut de la matrice, c'est-à-dire par le terme A (1,J).

Il est ici encore possible d'introduire les rationnels en utilisant la barre oblique ; $\frac{2}{3}$ peut être entré 0.666 666 666 mais aussi 2/3.

Lorsque toutes les colonnes ont été introduites, il faut répondre Ø à la question "numéro de la colonne à entrer" pour indiquer au programme que la saisie est terminée.

Il est possible d'introduire la matrice sans préciser les numéros des colonnes ; à la question "numéro de la colonne à entrer", la réponse "T" pour toutes indique au programme que tous les coefficients seront introduits en une seule fois, mais toujours colonne par colonne.

Dans le cas où la matrice était déjà définie avant la nouvelle introduction, il ne faut pas oublier que les anciennes valeurs ne sont pas effacées.

Supposons par exemple que la matrice A soit :

$$A = \begin{pmatrix} 1 & 2 & 6 \\ 0 & 4 & -2 \\ 3 & 5 & 8 \end{pmatrix}$$

et que l'on désire la remplacer par :

$$A' = \begin{pmatrix} 0 & 0 & 0 \\ 0 & 3 & 0 \\ 0 & 4 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

Si l'on introduit uniquement la colonne 2, la matrice en mémoire sera en fait :

$$A'' = \begin{pmatrix} 1 & 0 & 6 \\ 0 & 3 & -2 \\ 3 & 4 & 8 \\ 0 & 0 & 0 \end{pmatrix}$$

Nous avons donc prévu la possibilité de remettre la matrice à 0 en une opération ; à la question "numéro de la colonne à entrer", il faut répondre ZÉRO (en toutes lettres).

L'introduction de la matrice A' ci-dessus se fera donc en deux étapes ; mise à zéro de la matrice A, puis introduction de la deuxième colonne.

Le fait de ne pas remettre à zéro la matrice à chaque introduction permet de compléter une matrice précédemment entrée. Par exemple, pour remplacer la matrice :

$$B = \begin{pmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{pmatrix}$$

par la matrice :

$$B' = \begin{pmatrix} 1 & 2 & 3 & 88 \\ 4 & 5 & 6 & 99 \end{pmatrix}$$

il suffira d'introduire la quatrième colonne, les trois premières ayant conservé leurs valeurs;

Récapitulons les quatre réponses possibles à la question "Numéro de la colonne à entrer":

* Un nombre j strictement positif, qui indique au programme que la colonne j de la matrice va être introduite.
* Le nombre Ø, qui provoque le retour au menu.
* La lettre T, qui indique au programme que la totalité de la matrice va être entrée (colonne par colonne).
* Le mot ZÉRO, qui provoque la remise à zéro de la matrice.

- **Rappel d'une matrice** :

Le programme commence par afficher les dimensions de la matrice, puis pose la question "Numéro de la colonne à afficher?", à laquelle il faut répondre par:

* un nombre $j > 0$, pour afficher la colonne j,
* la lettre T, pour afficher toute la matrice,
* le chiffre Ø, pour revenir au menu.

- **Transposition d'une matrice** :

Considérons la matrice $M = (m_{ij})$ à l lignes et c colonnes.

La matrice ${}^tM$ est alors la matrice à c lignes et l colonnes dont le terme général est défini par:

$${}^tm_{ij} = m_{ji}$$

Par exemple, la matrice:

$$M = \begin{pmatrix} 5 & 2 \\ 3 & -1 \\ 8 & 6 \end{pmatrix}$$

admet comme transposée:

$${}^tM = \begin{pmatrix} 5 & 3 & 8 \\ 2 & -1 & 6 \end{pmatrix}$$

Le programme peut transposer l'une quelconque des trois matrices A, B et C.

```
2000 CLS
2010  PRINT  TAB( 5);"MANIPULATIONS SUR LES MATRICES": PRINT
    : PRINT
2020  PRINT "1- ENTREE DE A"
2030  PRINT "2- ENTREE DE B": PRINT
```

```
2040  PRINT "3- RAPPEL DE A"
2050  PRINT "4- RAPPEL DE B"
2060  PRINT "5- RAPPEL DE C": PRINT
2070  PRINT "6- TRANSFERT DE A DANS C"
2080  PRINT "7- TRANSFERT DE C DANS A": PRINT
2090  PRINT "8- TRANSFERT DE B DANS C"
2100  PRINT "9- TRANSFERT DE C DANS B": PRINT
2110  PRINT "10- TRANSFERT DE A DANS B"
2120  PRINT "11- TRANSFERT DE B DANS A": PRINT
2130  PRINT "12- TRANSPOSITION": PRINT
2140  PRINT "13- RETOUR AU MENU PRINCIPAL": PRINT
2150  INPUT "VOTRE CHOIX ";E:CLS
2160  IF E < 1 OR E > 12 GOTO 100
2170  IF E = 12 GOTO 2890
2180  IF E > 5 GOTO 2700
2190  IF E > 2 GOTO 2500
2200  PRINT  TAB( 5);"ENTREE DE ";MM$(E): PRINT : PRINT

2210  INPUT "NOMBRE DE LIGNES ";NL
2220  IF NL > MX OR NL < 0 GOTO 2210
2230  INPUT "NOMBRE DE COLONNES ";NC
2240  IF NC > MX OR NC < 0 GOTO 2230
2250  IF NC <  > 0 GOTO 2280
2260  IF NL <  > 0 GOTO 100
2270 DM(E,1) = 0:DM(E,2) = 0: GOTO 2000
2280  IF NL = 0 GOTO 100
2290 DM(E,1) = NL:DM(E,2) = NC:M = E
2300  PRINT : INPUT "NUMERO DE LA COLONNE A ENTRER ";Z$
2310  IF Z$ = "T" THEN  FOR J = 1 TO NC: GOSUB 2400: PRINT
    : NEXT J: GOTO 2300
2320  IF Z$ = "ZERO" THEN  PRINT : PRINT "PATIENTEZ,JE R
    EMETS LA MATRICE A ZERO": FOR I = 1 TO NL: FOR J = 1
     TO NC:U = 0: GOSUB 950: NEXT J: NEXT I: PRINT : GOTO
    2300
2330 J =  VAL (Z$): IF J <  = 0 GOTO 2000
2340  IF J <  = NC THEN  GOSUB 2400
2350  GOTO 2300
2400  FOR I = 1 TO NL
2410  PRINT  TAB( 5);MM$(E);"(";I;",";J;")";
2420  INPUT Z$: GOSUB 800
2430  IF ER = 1 GOTO 2410
2440 U = NN: GOSUB 950
2450  NEXT I
2460  RETURN
2500  PRINT  TAB( 5);"RAPPEL DE ";MM$(E - 2): PRINT : PRINT
```

```
2510 L = E - 2
2520  IF DM(L,1) = 0 THEN  PRINT "LA MATRICE ";MM$(L);"
   N'EST PAS DEFINIE...": PRINT : GOTO 2790
2530  PRINT "NOMBRE DE LIGNES:";DM(L,1)
2540  PRINT "NOMBRE DE COLONNES:";DM(L,2): PRINT
2550  INPUT "NUMERO DE LA COLONNE A AFFICHER ";Z$
2560  IF Z$ = "T" THEN  GOSUB 2670: GOTO 2550
2570 J =  VAL (Z$): IF J <  = 0 GOTO 2000
2580  IF J <  = DM(L,2) THEN  GOSUB 2600
2590  GOTO 2550
2600  FOR I = 1 TO DM(L,1)
2610  GOSUB 900:X = U: GOSUB 300
2620  PRINT MM$(L);"(";I;",";J;")=";X;
2630  IF DD > 1 THEN  PRINT  TAB( 28);NN;"/";DD;
2640  PRINT
2650  NEXT I
2660  RETURN
2670  FOR J = 1 TO DM(L,2): GOSUB 2600: PRINT : NEXT J: RETURN

2700  IF E = 6 THEN L = 1:M = 3
2710  IF E = 7 THEN L = 3:M = 1
2720  IF E = 8 THEN L = 2:M = 3
2730  IF E = 9 THEN L = 3:M = 2
2740  IF E = 10 THEN L = 1:M = 2
2750  IF E = 11 THEN L = 2:M = 1
2760  PRINT  TAB( 5);"TRANSFERT DE ";MM$(L);" DANS ";MM$
   (M)
2770  GOSUB 2810
2780  PRINT : PRINT "TRANSFERT EFFECTUE": PRINT
2790  INPUT "APPUYEZ SUR RETURN POUR REVENIR AU MENU ";Z
   $
2800  GOTO 2000
2810 DM(M,1) = DM(L,1):DM(M,2) = DM(L,2)
2820  FOR I = 1 TO DM(M,1)
2830  FOR J = 1 TO DM(M,2)
2840  GOSUB 900: GOSUB 950
2850  NEXT J
2860  NEXT I
2870  RETURN
2890  PRINT  TAB( 10);"TRANSPOSITION": PRINT
2900  INPUT "QUELLE MATRICE?";Z$
2910  IF Z$ = "A" THEN M = 1: GOTO 2950
2920  IF Z$ = "B" THEN M = 2: GOTO 2950
2930  IF Z$ = "C" THEN M = 3: GOTO 2950
2940  GOTO 2900
```

```
2950 DM(4,1) = DM(M,2):DM(4,2) = DM(M,1):L = M: PRINT
2960  IF DM(4,1) = 0 GOTO 2990
2970  FOR J = 1 TO DM(4,1): FOR I = 1 TO DM(4,2): GOSUB
    900:D(J,I) = U: NEXT I: NEXT J
2980 L = 4: GOSUB 2810
2990  PRINT "TRANSPOSITION EFFECTUEE": PRINT : GOTO 2790
```

4. *Calcul vectoriel*

a. Rappels théoriques

Soit un espace vectoriel E de dimension n, U et V deux vecteurs de cet espace vectoriel.

● Soients K1 et K2 deux réels ; le vecteur W défini par :

$W = K1 . U + K2 . V$

est appelé combinaison linéaire de U et V.

Dans le cas où U et V ne sont pas colinéaires, W appartient au plan vectoriel engendré par U et V.

● On appelle produit scalaire une forme bilinéaire symétrique définie positive de E × E dans R, c'est-à-dire une application f de E × E dans R vérifiant les propriétés suivantes :

- $f(U,V) = U.V \in R$
- $f(\lambda U + \mu U',V) = \lambda f(U,V) + \mu f(U',V)$
  $f(U,\lambda' V + \mu' V') = \lambda' f(U,V) + \mu' f(U,V')$
  ceci $\forall\ \lambda \in R,\ \forall\ \lambda' \in R,\ \forall\ \mu \in R,\ \forall\ \mu' \in R,\ \forall\ U \in E,\ \forall\ U' \in E,\ \forall\ V \in E,\ \forall\ V' \in E$
  (propriétés de bilinéarité).
- $f(U,V) = f(V,U) \quad \forall\ U \in E,\ \forall\ V \in E$
- $f(U,U) = 0 \Rightarrow U = 0$
- $\forall\ U \neq 0 \qquad f(U,U) > 0$

Par abus de langage, on désigne par produit scalaire de U et V le réel U.V.

Lorsque le produit scalaire U.V est nul, on dit que les vecteurs U et V sont orthogonaux.

Par définition, on appelle norme d'un vecteur U le réel $\|U\|$ donné par la relation :

$\|U\| = \sqrt{U.U}$

On appelle base orthonormée de l'espace vectoriel E de dimension n, une base de E dont les vecteurs $(a_i)_{i \in [1,n]}$ vérifient les relations :

$$\forall\, i, j \in [1,n] \text{ et } i \neq j \qquad a_i \,.\, a_j = 0$$
$$\forall\, i \in [1,n] \qquad a_i \,.\, a_i = 1$$

ce qui revient à dire que les vecteurs de la base sont orthogonaux deux à deux et ont tous une norme égale à 1

Cherchons alors à exprimer le produit scalaire des vecteurs U et V dans une base orthonormée $(a_i)_{i \in [1,n]}$ de l'espace vectoriel E.

Les vecteurs U et V peuvent s'écrire en fonction des vecteurs de la base :

$$U = \sum_{i=1}^{n} u_i\, a_i \qquad u_i \in R$$

$$V = \sum_{j=1}^{n} v_j\, a_j \qquad v_i \in R$$

Le produit scalaire s'écrit alors :

$$U \,.\, V = \left(\sum_{i=1}^{n} u_i\, a_i\right) . \left(\sum_{j=1}^{n} v_j\, a_j\right)$$

$$U \,.\, V = \sum_{i=1}^{n} \sum_{j=1}^{n} (u_i\, a_i) \,.\, (v_j\, a_j)$$

$$U \,.\, V = \sum_{i=1}^{n} \sum_{j=1}^{n} u_i\, v_j\, (a_i \,.\, a_j)$$

Les vecteurs a_i étant normés et orthogonaux deux à deux, on obtient finalement :

$$U \,.\, V = \sum_{i=1}^{n} u_i\, v_i$$

Par exemple, dans un espace de dimension 3, si les vecteurs U et V ont pour coordonnées :

$$U \begin{pmatrix} x \\ y \\ z \end{pmatrix} \qquad V \begin{pmatrix} x' \\ y' \\ z' \end{pmatrix}$$

dans une base orthonormée, le produit scalaire vaut alors :

$$U \,.\, V = xx' + yy' + zz'$$

● Plaçons nous cette fois dans un espace vectoriel euclidien (c'est-à-dire muni d'un produit scalaire) orienté de dimension 3 rapporté à une base orthonormée directe.

On appelle produit vectoriel des deux vecteurs U et V le vecteur $W = U \wedge V$ défini par :

* si U et V sont colinéaires, $W = 0$,
* si U et V ne sont pas colinéaires ;
  W est orthogonal au plan engendré par U et V
  (U,V,W) forment une base directe,
  la norme de W est donnée par :

  $$\|W\| = \|U\| \, . \, \|V\| \, . \, |\sin (U,V)|$$

Si les vecteurs U et V ont pour coordonnées :

$$U \begin{pmatrix} x \\ y \\ z \end{pmatrix} \qquad V \begin{pmatrix} x' \\ y' \\ z' \end{pmatrix}$$

le vecteur W est alors donné par :

$$W \begin{pmatrix} yz' - zy' \\ zx' - xz' \\ xy' - yx' \end{pmatrix}$$

De nombreuses lois physiques font intervenir le produit vectoriel : citons la force de Laplace, s'exerçant sur toute charge q en mouvement dans un champ magnétique :

$$\vec{f} = q \, \vec{v} \wedge \vec{B}$$

où $\vec{B}$ est le champ magnétique et $\vec{v}$ la vitesse de la charge.

● On appelle produit mixte de trois vecteurs U, V, W le déterminant de ces vecteurs :

$$PM = \det (U,V,W)$$

Le produit mixte est lié aux produits scalaire et vectoriel par les relations :

$$\det (U,V,W) = (U \wedge V) \, . \, W = U \, . \, (V \wedge W)$$

b. Programme

Le programme permet le calcul de combinaisons linéaires, de produits scalaire, vectoriel et mixte des vecteurs U, V (et W pour le produit mixte).

● *Combinaison linéaire* :

Après avoir fourni au programme les valeurs des coefficients K1 et K2, celui-ci effectue la combinaison linéaire K1 . U + K2 . V et place le résultat dans W.

● *Produit scalaire* :

Les vecteurs U et V ayant été définis, le calcul est automatique et le programme affiche le résultat.

● *Produit vectoriel* :

Le programme calcule le produit U ∧ V et place le résultat dans W.

● *Produit mixte* :

Le programme calcule et affiche automatiquement le produit mixte des vecteurs U, V et W.

```
3000 CLS
3010  PRINT  TAB( 10);"CALCUL VECTORIEL": PRINT  TAB( 5)
    ;"(BASE ORTHONORMEE DIRECTE)": PRINT : PRINT
3020  PRINT "1- CALCUL DE W=K1*U+K2*V": PRINT
3030  PRINT "2- CALCUL DU PRODUIT SCALAIRE PS=U.V": PRINT

3040  PRINT "3- CALCUL DU PRODUIT VECTORIEL W=U^V": PRINT

3050  PRINT "4- CALCUL DU PRODUIT MIXTE PM=DET(U,V,W)"
3060  PRINT "5- RETOUR AU MENU PRINCIPAL": PRINT
3070  INPUT "VOTRE CHOIX ";E: IF E < 1 OR E > 4 GOTO 100

3080 CLS: ON E GOSUB 3100,3200,3300,3400
3090  PRINT : INPUT "APPUYEZ SUR RETURN POUR REVENIR AU
    MENU";Z$: GOTO 3000
3100  PRINT  TAB( 5);"CALCUL DE W=K1*U+K2*V": PRINT : PRINT

3110  IF DV(1) <  > DV(2) THEN  PRINT "DIMENSIONS DE U E
    T V INCOMPATIBLES": GOTO 3190
3120  INPUT "VALEUR DE K1 ";K1
3130  INPUT "VALEUR DE K2 ";K2
3140 DV(3) = DV(1)
3150  FOR I = 1 TO DV(1)
3160 VE(3,I) = K1 * VE(1,I) + K2 * VE(2,I)
3170  NEXT I
3180 E = 6: GOSUB 1500
3190  RETURN
3200  PRINT  TAB( 5);"CALCUL DU PRODUIT SCALAIRE PS=U.V"
    : PRINT
```

```
3210  IF DV(1) <   > DV(2) THEN  PRINT "DIMENSIONS DE U E
   T V INCOMPATIBLES": GOTO 3290
3220 X = 0
3230  FOR I = 1 TO DV(1):X = X + VE(1,I) * VE(2,I): NEXT
   I
3240  PRINT "LE PRODUIT SCALAIRE VAUT :"
3250  GOSUB 300
3260  PRINT " U.V = ";X;
3270  IF DD > 1 THEN  PRINT  TAB( 26);NN;"/";DD;
3280  PRINT
3290  RETURN
3300  PRINT  TAB( 5);"CALCUL DU PRODUIT VECTORIEL W=U^V"
   : PRINT
3310  IF DV(1) <   > 3 OR DV(2) <   > 3 THEN  PRINT "U ET
   V DOIVENT ETRE DE DIMENSION 3": GOTO 3370
3320 DV(3) = 3
3330 VE(3,1) = VE(1,2) * VE(2,3) - VE(2,2) * VE(1,3)
3340 VE(3,2) = VE(1,3) * VE(2,1) - VE(2,3) * VE(1,1)
3350 VE(3,3) = VE(1,1) * VE(2,2) - VE(2,1) * VE(1,2)
3360 E = 6: GOSUB 1500
3370  RETURN
3400  PRINT "  CALCUL DU PRODUIT MIXTE PM=DET(U,V,W)": PRINT

3410  IF DV(1) <   > 3 OR DV(2) <   > 3 OR DV(3) <   > 3 THEN
     PRINT "U,V,W DOIVENT ETRE DE DIMENSION 3": GOTO 349
   0
3420 P = 1:N = 3
3430  FOR I = 1 TO 3: FOR J = 1 TO 3:D(I,J) = VE(I,J): NEXT
   J: NEXT I
3440  GOSUB 500:X = DR: GOSUB 300
3450  PRINT "LE PRODUIT MIXTE VAUT :"
3460  PRINT " DET(U,V,W)=";DR;
3470  IF DD > 1 THEN  PRINT  TAB( 25);NN;"/";DD;
3480  PRINT
3490  RETURN
```

c. Exemples commentés

• ***Exemple 1*** :

Considérons les deux vecteurs :

$$U \begin{pmatrix} 2 \\ -5 \\ 7/2 \\ 0 \\ 6 \end{pmatrix} \qquad V \begin{pmatrix} 2 \\ -7 \\ 0 \\ 3/5 \\ 1 \end{pmatrix}$$

d'un espace de dimension 5.

Calculons la combinaison linéaire :

$$W = 2U - 4V$$

ainsi que le produit scalaire U.V.

```
     ENTREE DE U

DIMENSION DU VECTEUR U 5

COORDONNEES DU VECTEUR U :
    U(1) ?2
    U(2) ?-5
    U(3) ?7/2
    U(4) ?0
    U(5) ?6


    ENTREE DE V

DIMENSION DU VECTEUR V 5

COORDONNEES DU VECTEUR V :
    V(1) ?2
    V(2) ?-7
    V(3) ?0
    V(4) ?3/5
    V(5) ?1
```

```
    CALCUL DE W=K1*U+K2*V


VALEUR DE K1 2
VALEUR DE K2 -4
          RAPPEL DE W

W (1) =-4
W (2) =18
W (3) =7
W (4) =-2.4                    -12/5
W (5) =8



    CALCUL DU PRODUIT SCALAIRE PS=U.V

LE PRODUIT SCALAIRE VAUT :
 U.V = 45
```

Il est ici encore possible d'utiliser la forme rationnelle pour introduire les coordonnées des vecteurs ; de même, les résultats sont accompagnés des approximations rationnelles.

● ***Exemple 2*** :

Soients, dans un espace de dimension 3, les deux vecteurs :

$$U \quad \begin{pmatrix} 2 \\ -5 \\ 6 \end{pmatrix} \qquad\qquad V \quad \begin{pmatrix} 1 \\ 7 \\ 5/2 \end{pmatrix}$$

Calculons leur produit vectoriel et leur produit scalaire.

```
    ENTREE DE U

DIMENSION DU VECTEUR U 3

COORDONNEES DU VECTEUR U :
    U(1) ?2
    U(2) ?-5
    U(3) ?6
    ENTREE DE V
```

```
DIMENSION DU VECTEUR V 3

COORDONNEES DU VECTEUR V :
    V(1) ?1
    V(2) ?7
    V(3) ?5/2


    CALCUL DU PRODUIT VECTORIEL W=U^V

         RAPPEL DE W

W (1) =-54.5                 -109/2
W (2) =1
W (3) =19



    CALCUL DU PRODUIT SCALAIRE PS=U.V

LE PRODUIT SCALAIRE VAUT :
 U.V = -18
```

• ***Exemple 3*** :

Toujours dans un espace de dimension 3, cherchons le produit mixte des trois vecteurs :

$$U \begin{pmatrix} 2 \\ 5 \\ 0 \end{pmatrix} \qquad V \begin{pmatrix} -4 \\ 2/5 \\ 3/2 \end{pmatrix} \qquad W \begin{pmatrix} 4 \\ -1/5 \\ 3/2 \end{pmatrix}$$

```
    ENTREE DE U

DIMENSION DU VECTEUR U 3

COORDONNEES DU VECTEUR U :
    U(1) ?2
    U(2) ?5
    U(3) ?0
    ENTREE DE V
```

```
DIMENSION DU VECTEUR V 3

COORDONNEES DU VECTEUR V :
    V(1) ?-4
    V(2) ?2/5
    V(3) ?3/2
    ENTREE DE W

DIMENSION DU VECTEUR W 3

COORDONNEES DU VECTEUR W :
    W(1) ?4
    W(2) ?-1/5
    W(3) ?3/2




  CALCUL DU PRODUIT MIXTE PM=DET(U,V,W)

LE PRODUIT MIXTE VAUT :
 DET(U,V,W)=61.8         309/5
```

5. *Calcul matriciel*

Le choix de cette option du menu principal conduit à un menu secondaire, où les possibilités suivantes sont offertes :

- calcul de combinaisons linéaires de matrices,
- calcul du produit de deux matrices,
- élévation à la puissance N d'une matrice,
- calcul du déterminant d'une matrice,
- inversion d'une matrice,
- recherche des valeurs propres d'une matrice.

Il faudra donc introduire le menu suivant avant d'entrer l'un des six programmes mentionnés ci-dessus.

```
3500 CLS: PRINT  TAB( 5);"CALCUL MATRICIEL": PRINT : PRINT

3510  PRINT "1- CALCUL DE C = K1 * A + K2 * B": PRINT
3520  PRINT "2- CALCUL DU PRODUIT C = A.B": PRINT
3530  PRINT "3- ELEVATION A LA PUISSANCE N DE A": PRINT

3540  PRINT "4- CALCUL DU DETERMINANT DE A": PRINT
3550  PRINT "5- INVERSION DE LA MATRICE A": PRINT
3560  PRINT "6- RECHERCHE DES VALEURS PROPRES DE A": PRINT

3570  PRINT "7- RETOUR AU MENU PRINCIPAL": PRINT
3580  INPUT "VOTRE CHOIX ";E
3590  ON E GOTO 4500,5000,5500,6000,6500,7000,100
```

6. *Combinaison linéaire de deux matrices*

a. Principe

Le programme effectue le calcul suivant :

$$C = K1 . A + K2 . B$$

où K1 et K2 sont des réels, A et B deux matrices de mêmes dimensions.

Deux cas particuliers sont très fréquemment rencontrés en pratique :

- la multiplication d'une matrice par une constante : si $K2 = 0$, alors $C = K1 . A$.
- la somme des matrices A et B : il suffit de choisir $K1 = K2 = 1$ pour obtenir $C = A + B$.

b. Programme

Après avoir vérifié la compatibilité des dimensions des matrices A et B, le programme calcule et stocke le résultat de la combinaison linéaire dans la matrice C, puis affiche cette dernière, accompagnée des approximations rationnelles.

Si la matrice C comporte beaucoup de coefficients, tous ne pourront être affichés sur l'écran ; il reste cependant possible de rappeler la matrice colonne par colonne grâce à l'option "Rappel de C".

```
4500 CLS
4510  PRINT  TAB( 5);"CALCUL DE C = K1 * A + K2 * B": PRINT
    : PRINT
4520  INPUT "K1?";K1: PRINT
4530  INPUT "K2?";K2: PRINT
4540  IF K1 = 0 GOTO 4590
4550  IF (DM(1,1) <  > DM(2,1) OR DM(1,2) <  > DM(2,2)) AND
    K2 <  > 0 THEN  PRINT "LES DIMENSIONS SONT INCOMPATI
    BLES": GOTO 4680
4560  IF DM(1,1) = 0 THEN  PRINT "LA MATRICE A N'EST PAS
     DEFINIE...": GOTO 4680
4570 DM(3,1) = DM(1,1):DM(3,2) = DM(1,2)
4580  IF K2 = 0 GOTO 4620
4590  IF K2 = 0 GOTO 100
4600  IF DM(2,1) = 0 THEN  PRINT "LA MATRICE B N'EST PAS
     DEFINIE...": GOTO 4680
4610 DM(3,1) = DM(2,1):DM(3,2) = DM(2,2)
4620  FOR I = 1 TO DM(3,1)
4630  FOR J = 1 TO DM(3,2)
4640 C(I,J) = K1 * A(I,J) + K2 * B(I,J)
4650  NEXT J
4660  NEXT I
4670 L = 3: GOSUB 2670
4680  PRINT : INPUT "VOULEZ-VOUS REUTILISER LE PROGRAMME
     ";Z$
4690  IF Z$ = "O" GOTO 4500
4700  GOTO 3500
```

c. Exemple commenté

Soient les deux matrices :

$$A = \begin{pmatrix} 2 & -1 & 1/5 \\ 3/5 & 0 & 6 \end{pmatrix} \qquad B = \begin{pmatrix} 1 & 2 & 5 \\ -7 & 3 & 0 \end{pmatrix}$$

Calculons $C = 5A + B$:

```
          ENTREE DE A


NOMBRE DE LIGNES 2
NOMBRE DE COLONNES 3

NUMERO DE LA COLONNE A ENTRER T
A(1,1)?2
A(2,1)?3/5

A(1,2)?-1
A(2,2)?0

A(1,3)?1/5
A(2,3)?6


NUMERO DE LA COLONNE A ENTRER 0
    ENTREE DE B


NOMBRE DE LIGNES 2
NOMBRE DE COLONNES 3

NUMERO DE LA COLONNE A ENTRER T
B(1,1)?1
B(2,1)?-7

B(1,2)?2
B(2,2)?3

B(1,3)?5
B(2,3)?0


NUMERO DE LA COLONNE A ENTRER 0
    CALCUL DE C = K1 * A + K2 * B


K1?5

K2?1
```

```
C(1,1)=11
C(2,1)=-4

C(1,2)=-3
C(2,2)=3

C(1,3)=6
C(2,3)=30


VOULEZ-VOUS REUTILISER LE PROGRAMME N
```

Le résultat est donc :

$$C = \begin{pmatrix} 11 & -3 & 6 \\ -4 & 3 & 30 \end{pmatrix}$$

7. *Multiplication de deux matrices*

a. Principe

Soient deux matrices $A = (a_{ij})$ et $B = (b_{ij})$.

Nous allons calculer le point $C = A . B = (c_{ij})$.

Ce produit n'existe que si le nombre de colonnes de A est égal au nombre de lignes de B ; soit n ce nombre.

Les coefficients c_{ij} sont alors donnés par :

$$c_{ij} = \sum_{k=1}^{n} a_{ik} . b_{kj}$$

La matrice produit C a le même nombre de lignes que A, et le même nombre de colonnes que B.

Rappelons que la multiplication des matrices n'est pas commutative (les matrices AB et BA ne peuvent exister simultanément que si A et B sont des matrices carrées de même dimension).

b. Programme

Le programme vérifie que les dimensions des matrices A et B sont compatibles, calcule et stocke le résultat du produit dans la matrice C, puis affiche celle-ci.

```
5000 CLS
5010  PRINT  TAB( 5);"CALCUL DU PRODUIT C = A.B": PRINT
    : PRINT
5020  IF DM(1,1) = 0 THEN  PRINT "LA MATRICE A N'EST PAS
     DEFINIE...": GOTO 5150
5030  IF DM(2,1) = 0 THEN  PRINT "LA MATRICE B N'EST PAS
     DEFINIE...": GOTO 5150
5040  IF DM(1,2) <  > DM(2,1) THEN  PRINT "DIMENSIONS IN
    COMPATIBLES...": GOTO 5150
5050 DM(3,1) = DM(1,1):DM(3,2) = DM(2,2)
5060  FOR I = 1 TO DM(1,1)
5070  FOR J = 1 TO DM(2,2)
5080 C(I,J) = 0
5090  FOR K = 1 TO DM(1,2)
5100 C(I,J) = C(I,J) + A(I,K) * B(K,J)
5110  NEXT K
5120  NEXT J
5130  NEXT I
5140 L = 3: GOSUB 2670
5150  PRINT : INPUT "APPUYEZ SUR RETURN POUR REVENIR AU
    MENU ";Z$
5160  GOTO 3500
```

c. Exemple commenté

Soient les deux matrices :

$$A = \begin{pmatrix} 3 & 1 & -7 \\ 2 & 4 & 1 \end{pmatrix}$$

$$B = \begin{pmatrix} 1/4 & 2 \\ 6 & 0 \\ 1 & -3 \end{pmatrix}$$

Le produit $C = A.B$ est :

$$C = \begin{pmatrix} -1/4 & 27 \\ 51/2 & 1 \end{pmatrix}$$

```
    ENTREE DE A


NOMBRE DE LIGNES 2
NOMBRE DE COLONNES 3

NUMERO DE LA COLONNE A ENTRER T
A(1,1)?3
A(2,1)?2

A(1,2)?1
A(2,2)?4

A(1,3)?-7
A(2,3)?1


NUMERO DE LA COLONNE A ENTRER 0
    ENTREE DE B


NOMBRE DE LIGNES 3
NOMBRE DE COLONNES 2

NUMERO DE LA COLONNE A ENTRER T
B(1,1)?1/4
B(2,1)?6
B(3,1)?1

B(1,2)?2
B(2,2)?0
B(3,2)?-3


NUMERO DE LA COLONNE A ENTRER 0
    CALCUL DU PRODUIT C = A.B


C(1,1)=-.25                   -1/4
C(2,1)=25.5                   51/2

C(1,2)=27
C(2,2)=1
```

8. *Élévation à la puissance N*

a. Principe

Il serait possible, pour élever la matrice A (qui doit être carrée) à la puissance N, d'effectuer $N-1$ multiplications successives grâce au programme précédent.

Cette méthode présente deux inconvénients lorsque N est grand :

- un grand nombre d'opérations est nécessaire, ce qui affaiblit la précision des calculs ;
- le temps de calcul est proportionnel à $N-1$ et devient rapidement important.

La méthode de Legendre que nous allons utiliser conduit à un temps de calcul proportionnel à $\log_2 N$, et est donc nettement plus efficace lorsque N est grand.

Le principe de la méthode repose sur la constatation suivante : pour calculer A^{2p}, il suffit de connaître A^p et de la multiplier par elle-même.

Utilisons une matrice auxiliaire D, et initialisons les matrices C et D de la manière suivante :

$$C = I \qquad\qquad D = A$$

L'algorithme est alors le suivant :

- si N est pair, on divise n par 2 et on élève D au carré,
- si N est impair, on décrémente n d'une unité et on multiplie C par D.

L'opération est répétée jusqu'à ce que N soit nul ; C contient alors A^N.

Considérons l'exemple suivant, où $N = 14$: nous avons initialement :

$$\begin{cases} C = I \\ D = A \end{cases} \qquad N = 14$$

N est pair ; divisons le par 2 et élevons D au carré :

$$\begin{cases} C = I \\ D = A^2 \end{cases} \qquad N = 7$$

N est impair ; décrémentons N et multiplions C par D :

$$\begin{cases} C = I \,.\, A^2 = A^2 \\ D = A^2 \end{cases} \qquad N = 6$$

on obtient de même successivement :

$$\begin{cases} C = A^2 \\ D = A^4 \end{cases} \qquad N = 3$$

$$\begin{cases} C = A^2 . A^4 = A^6 \\ D = A^4 \end{cases} \qquad N = 2$$

$$\begin{cases} C = A^6 \\ D = A^8 \end{cases} \qquad N = 1$$

$$\begin{cases} C = A^6 . A^8 \\ D = A^8 \end{cases} \qquad N = 0$$

Nous avons effectué 6 multiplications, au lieu de 13 par la méthode des multiplications successives.

La différence est d'autant plus sensible que N est grand :

si N = 100, seules 9 multiplications sont nécessaires,

si N = 1000, seules 15 multiplications sont nécessaires.

b. Programme

Le programme demande la valeur de N et place le résultat du calcul dans la matrice C.

```
5500 CLS
5510  PRINT "ELEVATION A LA PUISSANCE N DE A (C=A^N)": PRINT
    : PRINT
5520  IF DM(1,1) = 0 THEN  PRINT "LA MATRICE A N'EST PAS
     DEFINIE...": GOTO 5800
5530  IF DM(1,1) <  > DM(1,2) THEN  PRINT "LA MATRICE A
    N'EST PAS CARREE...": GOTO 5800
5540  INPUT "N=";NN:NN =  INT ( ABS (NN))
5550 N = DM(1,1):DM(3,1) = N:DM(3,2) = N: PRINT
5560  FOR I = 1 TO N: FOR J = 1 TO N:C(I,J) = 0:D(I,J) =
    A(I,J): NEXT J:C(I,I) = 1: NEXT I
5570  IF NN = 0 GOTO 5790
5580 PE = 0:U = NN / 2
5590  IF U =  INT (U) THEN PE = 1:NN = U
```

```
5600  IF PE = 0 THEN NN = NN - 1
5610  FOR I = 1 TO N
5620  FOR J = 1 TO N
5630 K = J + N:D(I,K) = 0
5640  FOR L = 1 TO N
5650 U = C(I,L)
5660  IF PE = 1 THEN U = D(I,L)
5670 D(I,K) = D(I,K) + U * D(L,J)
5680  NEXT L
5690  NEXT J
5700  NEXT I
5710  FOR I = 1 TO N
5720  FOR J = 1 TO N
5730 U = D(I,J + N)
5740  IF PE = 0 THEN C(I,J) = U
5750  IF PE = 1 THEN D(I,J) = U
5760  NEXT J
5770  NEXT I
5780  IF NN > 0 GOTO 5580
5790 L = 3: GOSUB 2670
5800  PRINT : INPUT "VOULEZ-VOUS REUTILISER LE PROGRAMME
     ";Z$
5810  IF Z$ = "O" GOTO 5500
5820  GOTO 3500
```

c. Exemple commenté

```
    ENTREE DE A


NOMBRE DE LIGNES 4
NOMBRE DE COLONNES 4

NUMERO DE LA COLONNE A ENTRER T
A(1,1)?1
A(2,1)?0
A(3,1)?2
A(4,1)?0

A(1,2)?0
A(2,2)?-3
A(3,2)?-5
A(4,2)?9
```

```
A(1,3)?2
A(2,3)?0
A(3,3)?0
A(4,3)?0

A(1,4)?1
A(2,4)?-5
A(3,4)?2
A(4,4)?0


NUMERO DE LA COLONNE A ENTRER 0
ELEVATION A LA PUISSANCE N DE A (C=A^N)


N=5

C(1,1)=65
C(2,1)=0
C(3,1)=58
C(4,1)=0

C(1,2)=-1425
C(2,2)=-13608
C(3,2)=-157
C(4,2)=8019

C(1,3)=58
C(2,3)=0
C(3,3)=36
C(4,3)=0

C(1,4)=255
C(2,4)=-4455
C(3,4)=8993
C(4,4)=-10935


VOULEZ-VOUS REUTILISER LE PROGRAMME N
```

On obtient donc :

$$\begin{pmatrix} 1 & 0 & 2 & 1 \\ 0 & -3 & 0 & -5 \\ 2 & -5 & 0 & 2 \\ 0 & 9 & 0 & 0 \end{pmatrix}^5 = \begin{pmatrix} 65 & -1425 & 58 & 255 \\ 0 & -13608 & 0 & -4455 \\ 58 & -157 & 36 & 8993 \\ 0 & 8019 & 0 & -10935 \end{pmatrix}$$

9. Calcul du déterminant

a. Principe

Nous cherchons ici à calculer le déterminant d'une matrice A et, dans le cas où celui-ci est nul, à déterminer le rang de la matrice.

Ces deux résultats sont en fait deux sous-produits de la résolution d'un système linéaire par la méthode de Gauss-Jordan, que nous exposerons en détail dans le paragraphe 13.

L'algorithme consistant à ramener le calcul d'un déterminant d'ordre N à N déterminants d'ordre N − 1, puis N (N − 1) déterminants d'ordre N − 2... est déplorable puisque le temps de calcul est proportionnel à N !, alors que celui de l'algorithme de Gauss-Jordan est proportionnel à N^3.

b. Programme

Le programme calcule le déterminant et le rang de la matrice A, puis affiche le déterminant suivi de son approximation rationnelle et, dans le cas où le déterminant est nul, le rang de la matrice. (Si le déterminant est non nul, le rang est égal à la dimension de A).

```
6000 CLS
6010  PRINT  TAB( 5);"CALCUL DU DETERMINANT DE A": PRINT
    : PRINT
6020  IF DM(1,1) = 0 THEN  PRINT "LA MATRICE A N'EST PAS
     DEFINIE...": GOTO 6110
6030  IF DM(1,1) <  > DM(1,2) THEN  PRINT "LA MATRICE A
    N'EST PAS CARREE...": GOTO 6110
```

```
6040 L = 1:M = 4: GOSUB 2810
6050 P = 1:N = DM(1,1): GOSUB 500
6060  PRINT "DET(A)=";DR;
6070 X = DR: GOSUB 300
6080  IF DD > 1 THEN  PRINT  TAB( 25);NN;"/";DD;
6090  PRINT
6100  IF RG < DM(1,1) THEN  PRINT : PRINT "LE RANG DE LA
     MATRICE A EST ";RG
6110  PRINT : PRINT : INPUT "APPUYEZ SUR RETURN POUR REV
    ENIR AU MENU ";Z$
6120  GOTO 3500
```

c. Exemples commentés

● ***Exemple 1*** :

```
    ENTREE DE A


NOMBRE DE LIGNES 4
NOMBRE DE COLONNES 4

NUMERO DE LA COLONNE A ENTRER T
A(1,1)?2
A(2,1)?0
A(3,1)?3
A(4,1)?1

A(1,2)?1
A(2,2)?1
A(3,2)?1
A(4,2)?1

A(1,3)?1
A(2,3)?3
A(3,3)?2
A(4,3)?0

A(1,4)?4
A(2,4)?2
A(3,4)?1
A(4,4)?3
```

```
NUMERO DE LA COLONNE A ENTRER 0
    CALCUL DU DETERMINANT DE A


DET(A)=-16
```

On a donc:

$$\begin{vmatrix} 2 & 1 & 1 & 4 \\ 0 & 1 & 3 & 2 \\ 3 & 1 & 2 & 1 \\ 1 & 1 & 0 & 3 \end{vmatrix} = -16$$

- ***Exemple 2*** :

```
    ENTREE DE A


NOMBRE DE LIGNES 3
NOMBRE DE COLONNES 3

NUMERO DE LA COLONNE A ENTRER T
A(1,1)?-1
A(2,1)?2
A(3,1)?4

A(1,2)?2
A(2,2)?4
A(3,2)?0

A(1,3)?5
A(2,3)?3
A(3,3)?-7


NUMERO DE LA COLONNE A ENTRER 0
    CALCUL DU DETERMINANT DE A


DET(A)=0

LE RANG DE LA MATRICE A EST 2
```

Le déterminant $\begin{vmatrix} -1 & 2 & 5 \\ 2 & 4 & 3 \\ 4 & 0 & -7 \end{vmatrix}$ est nul ;
le programme affiche alors le rang de la matrice, soit 2.

10. *Calcul de l'inverse*

a. Principe

La matrice inverse A^{-1} de la matrice A vérifie l'égalité :

$$A.A^{-1} = I \qquad (1)$$

Nous allons calculer la matrice A^{-1} par résolution de systèmes linéaires, en calculant successivement toutes les colonnes de A^{-1}.

Pour cela, appelons X_j le vecteur colonne représentant la j-ième colonne de la matrice A^{-1}.

L'égalité (1) conduit alors à :

$$A.X_j = \begin{pmatrix} 0 \\ \vdots \\ 0 \\ 1 \\ 0 \\ \vdots \\ 0 \end{pmatrix}$$ (vecteur colonne comportant 1 à la j-ième coordonnée, 0 partout ailleurs)

Nous verrons dans le paragraphe traitant de la résolution des systèmes linéaires, que la méthode de Gauss-Jordan permet de résoudre plusieurs systèmes linéaires basés sur la même matrice A simultanément.

Ainsi, inverser la matrice A revient à chercher les n vecteurs solutions des n systèmes du type :

$$A.X_j = \begin{pmatrix} 0 \\ \vdots \\ 0 \\ 1 \\ 0 \\ \vdots \\ 0 \end{pmatrix} \qquad j \in [1,n]$$

Les vecteurs X_j obtenus sont alors les colonnes de la matrice A^{-1}.

b. Programme

Le programme commence par calculer le déterminant de la matrice A ; si celui-ci est non nul, il calcule alors l'inverse et place le résultat dans C.

Une option intéressante est proposée : au lieu d'afficher les coefficients de A^{-1}, le programme peut afficher ceux de det (A) . A^{-1}.

En effet, dans le cas où les coefficients de A sont entiers, ceux de A^{-1} ne le sont pas forcément, alors que la matrice det (A) . A^{-1} est, elle, à coefficients entiers.

Si l'on souhaite améliorer la présentation des résultats, il faut alors répondre par "O" à la question "Multiplication par le déterminant ?"

```
6500 CLS
6510  PRINT  TAB( 5);"CALCUL DE L'INVERSE DE A": PRINT :
     PRINT
6520  IF DM(1,1) = 0 THEN  PRINT "LA MATRICE A N'EST PAS
     DEFINIE...": GOTO 6670
6530  IF DM(1,1) <  > DM(1,2) THEN  PRINT "LA MATRICE A
    N'EST PAS CARREE...": GOTO 6670
6540  INPUT "MULTIPLICATION PAR LE DETERMINANT?";Z$: PRINT

6550 N = DM(1,1):DM(3,1) = N:DM(3,2) = N
6560  FOR I = 1 TO N
6570  FOR J = 1 TO N:D(I,J) = A(I,J): NEXT J
6580  FOR J = N + 1 TO N + N:D(I,J) = 0: NEXT J
6590 D(I,N + I) = 1
6600  NEXT I
6610 P = N: GOSUB 500
6620  IF DR = 0 THEN  PRINT "LA MATRICE A N'EST PAS INVE
    RSIBLE": GOTO 6670
6630  PRINT "DETERMINANT DE A:";DR: PRINT
6640  IF Z$ <  > "O" THEN DR = 1
6650  FOR I = 1 TO N: FOR J = 1 TO N:C(I,J) = DR * D(I,J
     + N): NEXT J: NEXT I
6660 L = 3: GOSUB 2670
6670  PRINT : INPUT "VOULEZ-VOUS REUTILISER LE PROGRAMME
     ";Z$
6680  IF Z$ = "O" GOTO 6500
6690  GOTO 3500
```

c. Exemple commenté

Soit la matrice :

$$A = \begin{pmatrix} 3 & -1 & 2 \\ 1 & 0 & 3 \\ 4 & 0 & 2 \end{pmatrix}$$

dont le déterminant vaut det (A) = − 10.

```
    ENTREE DE A


NOMBRE DE LIGNES 3
NOMBRE DE COLONNES 3

NUMERO DE LA COLONNE A ENTRER 1
A(1,1)?3
A(2,1)?1
A(3,1)?4

A(1,2)?-1
A(2,2)?0
A(3,2)?0

A(1,3)?2
A(2,3)?3
A(3,3)?2


NUMERO DE LA COLONNE A ENTRER 0
    CALCUL DE L'INVERSE DE A


MULTIPLICATION PAR LE DETERMINANT?N

DETERMINANT DE A:-10

C(1,1)=0
C(2,1)=-1
C(3,1)=0

C(1,2)=-.2                  -1/5
C(2,2)=.200000001           1/5
C(3,2)=.4                   2/5
```

```
C(1,3)=.3                     3/10
C(2,3)=.7                     7/10
C(3,3)=-.1                    -1/10

VOULEZ-VOUS REUTILISER LE PROGRAMME O


    CALCUL DE L'INVERSE DE A


MULTIPLICATION PAR LE DETERMINANT?O

DETERMINANT DE A:-10

C(1,1)=0
C(2,1)=10
C(3,1)=0

C(1,2)=2
C(2,2)=-2.00000001
C(3,2)=-4

C(1,3)=-3
C(2,3)=-7
C(3,3)=1


VOULEZ-VOUS REUTILISER LE PROGRAMME N
```

On obtient donc bien:

$$A^{-1} = \begin{pmatrix} 0 & -1/5 & 3/10 \\ -1 & 1/5 & 7/10 \\ 0 & 2/5 & -1/10 \end{pmatrix} = \frac{1}{(-10)} \begin{pmatrix} 0 & 2 & -3 \\ 10 & -2 & -7 \\ 0 & 4 & 1 \end{pmatrix}$$

11. *Recherche des valeurs propres*

a. Principe

Soit une matrice carrée A, de dimension n.

Un vecteur X est dit vecteur propre de A associé à la valeur propre λ si et seulement si la relation suivante est vérifiée :

$$A.X = \lambda X$$

Le vecteur X n'est pas unique, mais appartient à un sous espace vectoriel de R^n.

Cette relation peut encore s'écrire :

$$(A - \lambda I) . X = 0$$

où I est la matrice identité.

Ce système homogène admet des solutions pour X autres que la solution triviale si il est dégénérée, c'est-à-dire si :

$$\det (A - \lambda I) = 0$$

ce qui peut encore s'écrire :

$$\begin{vmatrix} a_{11} - \lambda & a_{12} & \cdots & a_{1n} \\ a_{21} & a_{22} - \lambda & & \vdots \\ \vdots & & \ddots & \vdots \\ a_{n1} & \cdots & \cdots & a_{nn} - \lambda \end{vmatrix} = 0$$

Ce déterminant est un polynôme en λ de degré n, appelé polynôme caractéristique de A.

Les valeurs propres de A sont les racines de ce polynôme. Elles sont donc au nombre de n (et peuvent être complexes, ou confondues).

Leur recherche se fera donc en deux étapes : détermination du polynôme caractéristique de A, puis recherche des racines complexes de ce polynôme.

- *Détermination du polynôme caractéristique* :

Nous allons utiliser la méthode de Souriau.

Appelons $(c_i)_{i \in [0,n]}$ les (n + 1) coefficients du polynôme caractéristique :

$$P(A) = c_0\lambda^n + c_1\lambda^{n-1} + \ldots + c_{n-1}\lambda + c_n$$

On définit alors une suite de matrices (A_k) par la relation de récurrence :

$$k \in [1,n] \qquad \begin{cases} A_0 = 0 \\ A_k = (A_{k-1} + c_{k-1}\, I) \,.\, A \end{cases} \qquad \begin{aligned} c_0 &= 1 \\ c_k &= -\frac{1}{k}\,\mathrm{tr}\,(A_k) \end{aligned}$$

où tr (A_k) désigne la trace de la matrice A_k, c'est-à-dire la somme de ses éléments diagonaux.

Les $(c_k)_{k \in [0,n]}$ sont alors obtenus de proche en proche.

● *Résolution de l'équation caractéristique* :

La recherche des n zéros du polynôme obtenu est effectuée selon la méthode de Bairstow, que nous avons déjà présentée dans le premier tome, lors de l'étude des méthodes de résolution d'équations.

Le programme a été légèrement modifié, de manière à fournir les racines complexes.

b. Programme

Le programme opère sur la matrice A.

Les nombres p et q qui doivent être introduits au début de l'exécution sont les paramètres exigés par la méthode de Bairstow ; le couple (0,0) conduit généralement au résultat. (On se reportera au tome 1 pour de plus amples renseignements sur la méthode de Bairstow).

REMARQUE : à la fin de l'exécution, la matrice C a été effacée par le programme, qui l'a utilisée pour stocker des résultats intermédiaires. Les matrices A et B ne sont par contre pas affectées.

```
7000 CLS
7010  PRINT "RECHERCHE DES VALEURS PROPRES DE A": PRINT
    : PRINT
7020  IF DM(1,1) <  > DM(1,2) THEN  PRINT "LA MATRICE A
   N'EST PAS CARREE...": GOTO 7630
7030  IF DM(1,1) < 2 GOTO 100
7040 N = DM(1,1):A(0,0) = 1:DM(4,1) = N:DM(4,2) = N:CT =
    1
7050  INPUT "P,Q ";P,Q: PRINT
7060  FOR I = 1 TO N: FOR J = 1 TO N:C(I,J) = 0: NEXT J:
     NEXT I
7070  FOR K = 1 TO N
7080  FOR I = 1 TO N:C(I,I) = C(I,I) + A(0,K - 1): NEXT
    I
```

```
7090  FOR I = 1 TO N
7100  FOR J = 1 TO N
7110 D(I,J) = 0
7120  FOR L = 1 TO N:D(I,J) = D(I,J) + C(I,L) * A(L,J): NEXT
    L
7130  NEXT J
7140  NEXT I
7150 L = 4:M = 3: GOSUB 2810:A(0,K) = 0
7160  FOR I = 1 TO N:A(0,K) = A(0,K) + C(I,I): NEXT I
7170 A(0,K) =  - A(0,K) / K
7180  NEXT K
7190  IF N = 2 THEN P = A(0,1):Q = A(0,2): GOTO 7400
7200 J = 0
7210 B(0,0) = A(0,0):B(0,1) = A(0,1) - P * B(0,0)
7220  FOR K = 2 TO N
7230 B(0,K) = A(0,K) - P * B(0,K - 1) - Q * B(0,K - 2)
7240  NEXT K
7250 C(0,0) = B(0,0):C(0,1) = B(0,1) - P * C(0,0)
7260  IF N = 3 GOTO 7300
7270  FOR K = 2 TO N - 2
7280 C(0,K) = B(0,K) - P * C(0,K - 1) - Q * C(0,K - 2)
7290  NEXT K
7300 C(0,N - 1) =  - P * C(0,N - 2) - Q * C(0,N - 3)
7310 DD = C(0,N - 2) * C(0,N - 2) - C(0,N - 1) * C(0,N -
    3)
7320 U = B(0,N - 1) * C(0,N - 2) - B(0,N) * C(0,N - 3)
7330 V = B(0,N) * C(0,N - 2) - B(0,N - 1) * C(0,N - 1)
7340  IF DD = 0 THEN  PRINT "VALEURS DE P ET Q INADAPTEE
    S...": GOTO 7630
7350 X = U / DD:Y = V / DD
7360 P = P + X:Q = Q + Y
7370  IF ( ABS (X) +  ABS (Y)) < PR GOTO 7400
7380  IF J < 100 THEN J = J + 1: GOTO 7210
7390  PRINT "LE CALCUL NECESSITE TROP D'ITERATIONS": GOTO
    7630
7400 T = P * P - 4 * Q
7410  IF T < 0 GOTO 7490
7420 U = ( - P +  SQR (T)) / 2
7430  IF  ABS (U) < PR THEN U = 0
7440  PRINT "L";CT;"=";U
7450 U = ( - P -  SQR (T)) / 2
7460  IF  ABS (U) < PR THEN U = 0
7470  PRINT "L";CT + 1;"=";U
7480  GOTO 7540
7490 U =  ABS ( SQR ( - T)) / 2
```

```
7500  IF  ABS (U) < PR THEN U = 0
7510  IF  ABS (P / 2) < PR THEN P = 0
7520  PRINT "L";CT;"="; - P / 2;"  +  ";U;"  I"
7530  PRINT "L";CT + 1;"="; - P / 2;"  -  ";U;"  I"
7540 N = N - 2:CT = CT + 2
7550  FOR I = 0 TO N:A(0,I) = B(0,I): NEXT I
7560  IF N > 2 GOTO 7200
7570  IF N = 2 THEN P = B(0,1):Q = B(0,2): GOTO 7400
7580  IF N < 1 GOTO 7620
7590 U =  - B(0,1) / B(0,0)
7600  IF  ABS (U) < PR THEN U = 0
7610  PRINT "L";CT;"=";U
7620 DM(3,1) = 0:DM(3,2) = 0
7630  PRINT : INPUT "VOULEZ-VOUS REUTILISER LE PROGRAMME
     ";Z$
7640  IF Z$ = "O" GOTO 7000
7650  GOTO 3500
```

c. Exemples commentés

● ***Exemple 1*** :

Soit la matrice A suivante :

$$A = \begin{pmatrix} 17 & 24 & 30 & 17 \\ 8 & 13 & 20 & 7 \\ 2 & 10 & 8 & 6 \\ -23 & -43 & -54 & -26 \end{pmatrix}$$

dont les quatre valeurs propres sont 7, 2, -1, 4.

```
    ENTREE DE A


NOMBRE DE LIGNES 4
NOMBRE DE COLONNES 4

NUMERO DE LA COLONNE A ENTRER T
    A(1,1)?17
    A(2,1)?8
    A(3,1)?2
    A(4,1)?-23
```

```
    A(1,2)?24
    A(2,2)?13
    A(3,2)?10
    A(4,2)?-43

    A(1,3)?30
    A(2,3)?20
    A(3,3)?8
    A(4,3)?-54

    A(1,4)?17
    A(2,4)?7
    A(3,4)?6
    A(4,4)?-26

NUMERO DE LA COLONNE A ENTRER 0


RECHERCHE DES VALEURS PROPRES DE A


P,Q 0,0

L1=2
L2=-1
L3=7
L4=4

VOULEZ-VOUS REUTILISER LE PROGRAMME N
```

● ***Exemple 2*** :

```
    ENTREE DE A


NOMBRE DE LIGNES 3
NOMBRE DE COLONNES 3

NUMERO DE LA COLONNE A ENTRER T
A(1,1)?0
A(2,1)?1
A(3,1)?3
```

```
A(1,2)?-73
A(2,2)?39
A(3,2)?103

A(1,3)?25
A(2,3)?-13
A(3,3)?-34


NUMERO DE LA COLONNE A ENTRER 0
RECHERCHE DES VALEURS PROPRES DE A


P,Q 0,0

L1=1   +  2  I
L2=1   -  2  I
L3=3

VOULEZ-VOUS REUTILISER LE PROGRAMME N
```

La matrice A possède ici deux valeurs propres complexes conjuguées, 1 + 2i et 1 − 2i.

● ***Exemple 3*** :

Les valeurs propres de la matrice :

$$A = \begin{pmatrix} 2 & -2 & 1 \\ 2 & -3 & 2 \\ -1 & 2 & 0 \end{pmatrix}$$

sont 1, 1 et − 3 : la valeur propre 1 est racine double du polynôme caractéristique.

```
    ENTREE DE A


NOMBRE DE LIGNES 3
NOMBRE DE COLONNES 3

NUMERO DE LA COLONNE A ENTRER T
A(1,1)?2
A(2,1)?2
A(3,1)?-1
```

```
A(1,2)?-2
A(2,2)?-3
A(3,2)?2

A(1,3)?1
A(2,3)?2
A(3,3)?0


NUMERO DE LA COLONNE A ENTRER 0
RECHERCHE DES VALEURS PROPRES DE A


P,Q 0,0

L1=1.00002643
L2=.999973571
L3=-3

VOULEZ-VOUS REUTILISER LE PROGRAMME N
```

On constate que la précision de la racine double est médiocre : on retrouve ici le phénomène rencontré lors de la recherche de racines multiples (voir Tome 1).

Il existe d'autres méthodes de recherche de valeurs propres, mais elles possèdent toutes la particularité de ne donner que des résultats approximatifs pour les racines multiples.

12.. Image d'un vecteur par une matrice

a. Principe

Soient une matrice A et un vecteur U, tels que le nombre de colonnes de A soit égal à la dimension de U.

On appelle alors image du vecteur U par la matrice A le vecteur V défini par la relation :

$$V = A.U$$

La dimension de V est égale au nombre de lignes de la matrice A.

b. Programme

Le programme opère sur la matrice A et le vecteur U, et place le résultat dans V.

```
4000 CLS
4010  PRINT  TAB( 5);"CALCUL DE V=AU": PRINT
4020  IF DM(1,1) = 0 THEN  PRINT "LA MATRICE A N'EST PAS
     DEFINIE": GOTO 4140
4030  IF DV(1) = 0 THEN  PRINT "LE VECTEUR U N'EST PAS D
    EFINI": GOTO 4140
4040  IF DM(1,2) <  > DV(1) THEN  PRINT "DIMENSIONS DE A
     ET U INCOMPATIBLES": GOTO 4140
4050 DV(2) = DM(1,1)
4060  FOR I = 1 TO DV(2)
4070 VE(2,I) = 0
4080  FOR J = 1 TO DV(1)
4090 VE(2,I) = VE(2,I) + A(I,J) * VE(1,J)
4100  NEXT J
4110  NEXT I
4120 E = 5
4130  GOSUB 1500
4140  INPUT "APPUYEZ SUR RETURN POUR REVENIR AU MENU";Z$

4150  GOTO 100
```

c. Exemple commenté

Considérons la matrice A et le vecteur U suivants:

$$A = \begin{pmatrix} 5 & -1/2 & 1 \\ -2 & 4 & 0 \\ 0 & 5/2 & 3 \\ -1 & 6 & 9 \end{pmatrix} \qquad U = \begin{pmatrix} 2 \\ -1 \\ 1/2 \end{pmatrix}$$

Le vecteur V obtenu est de dimension 4.

```
    ENTREE DE A


NOMBRE DE LIGNES 4
NOMBRE DE COLONNES 3

NUMERO DE LA COLONNE A ENTRER T
    A(1,1)?5
```

```
    A(2,1)?-2
    A(3,1)?0
    A(4,1)?-1

    A(1,2)?-1/2
    A(2,2)?4
    A(3,2)?2.5
    A(4,2)?6

    A(1,3)?1
    A(2,3)?0
    A(3,3)?3
    A(4,3)?9


NUMERO DE LA COLONNE A ENTRER 0
    ENTREE DE U

DIMENSION DU VECTEUR U 3

COORDONNEES DU VECTEUR U :
    U(1) ?2
    U(2) ?-1
    U(3) ?1/2



    CALCUL DE V=AU

         RAPPEL DE V

V (1) =11
V (2) =-8
V (3) =-1
V (4) =-3.5                -7/2
```

13. Résolution d'un système linéaire

a. Principe

Soient une matrice carrée A et un vecteur U de dimension n.

Résoudre le système linéaire $AX = U$, c'est trouver l'ensemble des vecteurs X vérifiant cette relation matricielle.

Lorsque la matrice A n'est pas inversible, le système est dégénéré ; l'ensemble des solutions est soit infini, soit vide.

Par contre, lorsque la matrice A est inversible, le vecteur solution X est unique : $X = A^{-1}.U$.

Le système linéaire peut s'écrire sous la forme :

$$\begin{pmatrix} a_{11} & \cdots & a_{1n} \\ a_{21} & \cdots & a_{2n} \\ \vdots & & \vdots \\ a_{n1} & \cdots & a_{nn} \end{pmatrix} \begin{pmatrix} x_1 \\ x_2 \\ \vdots \\ x_n \end{pmatrix} = \begin{pmatrix} u_1 \\ u_2 \\ \vdots \\ u_n \end{pmatrix}$$

Nous allons le résoudre par la méthode de Gauss-Jordan, dont l'algorithme repose sur les propriétés suivantes :

- on peut multiplier une ligne du système par un nombre quelconque non nul sans en changer les solutions ;
- de même, on peut ajouter une ligne à une autre ;
- enfin, on peut permuter deux lignes.

REMARQUE : une ligne du système comprend la ligne de la matrice ainsi que la composante u_i correspondante.

La méthode de Gauss-Jordan consiste à transformer le système initial en utilisant ces propriétés, jusqu'à ce que la matrice A soit remplacée par la matrice identité I ; le système se réduit alors à $I.X = U'$; le vecteur U' contient la solution du système initial.

Cet algorithme permet la résolution simultanée de plusieurs systèmes utilisant tous la matrice A.

Supposons en effet que nous ayons les p systèmes suivants à résoudre :

$$\begin{cases} AX = U_1 \\ AX = U_2 \\ \vdots \\ AX = U_p \end{cases}$$

En regroupant les vecteurs U_j, on obtient la forme condensée suivante :

$$\underset{A}{\begin{pmatrix} a_{11} & \ldots\ldots & a_{1n} \\ a_{21} & \ldots\ldots & a_{2n} \\ \vdots & & \vdots \\ a_{n1} & \ldots\ldots & a_{nn} \end{pmatrix}} \underset{X}{\begin{pmatrix} x_1 \\ x_2 \\ \vdots \\ x_n \end{pmatrix}} = \underset{U_1}{\begin{pmatrix} u_1^1 \\ u_2^1 \\ \vdots \\ u_n^1 \end{pmatrix}} \underset{U_2}{\begin{pmatrix} u_1^2 \\ u_2^2 \\ \vdots \\ u_n^2 \end{pmatrix}} \ldots\ldots \underset{U_p}{\begin{pmatrix} u_1^p \\ u_2^p \\ \vdots \\ u_n^p \end{pmatrix}}$$

La résolution de ces p systèmes se fait donc de la même manière qu'avec un seul système, en considérant cette fois comme ligne du système la ligne de la matrice A suivie de toutes les i-ièmes composantes des vecteurs U_j.

Nous avons utilisé la résolution multiple dans le programme d'inversion de matrice ; les n vecteurs U_j étaient alors :

$$U_1 \begin{pmatrix} 1 \\ 0 \\ \vdots \\ 0 \end{pmatrix} \qquad U_2 \begin{pmatrix} 0 \\ 1 \\ 0 \\ \vdots \\ 0 \end{pmatrix} \qquad \ldots\ldots\ldots \qquad U_n \begin{pmatrix} 0 \\ \vdots \\ 0 \\ 1 \end{pmatrix}$$

b. Algorithme de Gauss-Jordan

Considérons le terme a_{11} de la matrice. S'il est nul, intervertissons la ligne 1 avec une autre afin d'obtenir un terme non nul. On appelle ce terme pivot, ce qui explique le nom de "méthode du pivot" parfois donné à cet algorithme.

Divisons alors toute la première ligne par a_{11} ; on obtient alors :

$$\begin{pmatrix} 1 & \frac{a_{12}}{a_{11}} & \cdots & \frac{a_{1n}}{a_{11}} \\ a_{21} & a_{22} & \cdots & a_{2n} \\ \vdots & & & \vdots \\ a_{n1} & \cdots & \cdots & a_{nn} \end{pmatrix} \begin{pmatrix} x_1 \\ x_2 \\ \vdots \\ x_n \end{pmatrix} = \begin{pmatrix} \frac{u_1}{a_{11}} \\ u_2 \\ \vdots \\ u_n \end{pmatrix}$$

Soustrayons ensuite à chacune des lignes j ($j \neq 1$) la ligne 1 multipliée par a_{i1}; cette opération fait apparaître des zéros dans tout le reste de la première colonne.

Poursuivons le calcul avec le pivot a_{22}, puis a_{kk} ($k < n$).

Si à un certain moment il est impossible de trouver un pivot non nul, le système n'est pas inversible, et la ligne où se situe ce pivot fournit le rang de la matrice A, à une unité près.

c. Programme

Le programme opère sur la matrice A et le vecteur U. A l'issue du calcul, si le système n'est pas dégénéré, le programme affiche le vecteur solution.

```
8000 CLS
8010  PRINT "RESOLUTION DU SYSTEME LINEAIRE AX=U": PRINT
    : PRINT
8020  IF DM(1,1) = 0 THEN  PRINT "LA MATRICE A N'EST PAS
     DEFINIE...": GOTO 8180
8030  IF DV(1) = 0 THEN  PRINT "LE VECTEUR U N'EST PAS D
    EFINI": GOTO 8180
8040  IF DM(1,1) <  > DM(1,2) THEN  PRINT "LA MATRICE A
    N'EST PAS CARREE...": GOTO 8180
8050  IF DM(1,2) <  > DV(1) THEN  PRINT "DIMENSIONS DE A
     ET U INCOMPATIBLES": GOTO 8180
8060 N = DV(1):DV(3) = N
8070  FOR I = 1 TO N:D(I,N + 1) = VE(1,I): NEXT I
8080 L = 1:M = 4: GOSUB 2810
8090 P = 1: GOSUB 500
8100  IF DR = 0 THEN  PRINT : PRINT "LE SYSTEME EST DEGE
    NERE...": GOTO 8180
8110  PRINT : PRINT "SOLUTION DU SYSTEME:"
8120  FOR I = 1 TO N
8130 X = D(I,N + 1): GOSUB 300
8140  PRINT " X";I;"=";X;
```

```
8150  IF DD > 1 THEN  PRINT  TAB( 25);NN;"/";DD;
8160  PRINT :VE(3,I) = X
8170  NEXT I: PRINT
8180  PRINT : INPUT "VOULEZ-VOUS REUTILISER LE PROGRAMME
     ";Z$
8190  IF Z$ = "O" GOTO 8000
8200  GOTO 100
```

d. Exemple commenté

```
    ENTREE DE A


NOMBRE DE LIGNES 3
NOMBRE DE COLONNES 3

NUMERO DE LA COLONNE A ENTRER T
    A(1,1)?2
    A(2,1)?1
    A(3,1)?3

    A(1,2)?-1
    A(2,2)?5
    A(3,2)?-2

    A(1,3)?1
    A(2,3)?-2
    A(3,3)?3


NUMERO DE LA COLONNE A ENTRER 0
    ENTREE DE U

DIMENSION DU VECTEUR U 3

COORDONNEES DU VECTEUR U :
    U(1) ?5
    U(2) ?1
    U(3) ?3


RESOLUTION DU SYSTEME LINEAIRE AX=U
```

```
SOLUTION DU SYSTEME:
 X1=3.35714286              47/14
 X2=-1.92857143             -27/14
 X3=-3.64285715             -51/14


VOULEZ-VOUS REUTILISER LE PROGRAMME N
```

La solution du système :

$$\begin{pmatrix} 2 & -1 & 1 \\ 1 & 5 & -2 \\ 3 & -2 & 3 \end{pmatrix} \begin{pmatrix} x_1 \\ x_2 \\ x_3 \end{pmatrix} = \begin{pmatrix} 5 \\ 1 \\ 3 \end{pmatrix}$$

est :

$$X = \begin{pmatrix} 47/14 \\ -27/14 \\ -51/14 \end{pmatrix}$$

14. Rapidité et précision des calculs

Lorsque l'ordre des matrices ne dépasse pas 3 ou 4, les calculs sont généralement très précis et très rapides (quelques secondes).

La précision reste satisfaisante lorsque les dimensions augmentent avec les opérations simples (somme pondérée, multiplication, élévation à la puissance).

Par contre, les résultats d'inversion, de résolution de système ou de recherche de valeurs propres peuvent devenir médiocres.

En particulier, les erreurs numériques peuvent devenir très importantes si la matrice initiale est "mal conditionnée" (une matrice mal conditionnée se caractérise par la dispersion des ordres de grandeur de ses coefficients).

Les calculs sur les matrices carrées ont un temps d'exécution proportionnel à n^3 ; ainsi, lorsque l'ordre de la matrice est voisin de 10, les temps d'exécution deviennent de l'ordre de quelques minutes.

La dimension maximale des matrices a été fixée à 15 à la ligne 10 du programme. Il peut être augmenté, mais, en dehors du fait que la précision peut devenir mauvaise pour les grosses matrices, la capacité mémoire nécessaire croît rapidement : pour MX = 20, la mémoire utilisée est de l'ordre de 19 Koctets, pour MX = 30, 32 Koctets, pour MX = 40, 50 Koctets.

Annexe

Fonctions trigonométriques complexes

$e^z = e^{x+iy} = e^x (\cos y + i \sin y)$

$e^{-z} = e^{-x-iy} = e^{-x} (\cos y - i \sin y)$

$\mathrm{ch}\, z = \dfrac{e^z + e^{-z}}{2}$

$\mathrm{sh}\, z = \dfrac{e^z - e^{-z}}{2}$

$\mathrm{th}\, z = \dfrac{\mathrm{sh}\, z}{\mathrm{ch}\, z} = \dfrac{e^z - e^{-z}}{e^z + e^{-z}}$ (pour $\mathrm{ch}\, z \neq 0$)

$\sin z = -\, \mathrm{sh}\, iz$

$\cos z = \mathrm{ch}\, iz$

$\mathrm{tg}\, z = \dfrac{\sin z}{\cos z}$ (pour $\cos z \neq 0$)

$\mathrm{Arg\,sh}\, z = \ln (z + \sqrt{1 + z^2})$

$\mathrm{Arg\,ch}\, z = \ln (z + \sqrt{z^2 - 1})$

$\mathrm{Arg\,th}\, z = \dfrac{1}{2} \ln \left(\dfrac{1+z}{1-z} \right)$ (pour $z \neq 1$, $z \neq -1$)

$\mathrm{Arc\,sin}\, z = -i \ln (iz + \sqrt{1 - z^2})$

$\mathrm{Arc\,cos}\, z = -i \ln (z + i \sqrt{1 - z^2})$

$\mathrm{Arc\,tg}\, z = -\dfrac{i}{2} \ln \left(\dfrac{i-z}{i+z} \right)$ (pour $z \neq i$, $z \neq -i$)

Imprimé en France. — JOUVE, 18, rue Saint-Denis, 75001 PARIS
N° 14541. Dépôt légal : Juillet 1985
N° d'Editeur : 4294

Cet ouvrage fait suite au premier tome des "MATHÉMATIQUES SUR MICRO-ORDINATEUR" consacré à l'Analyse, et s'adresse à tous ceux qui désirent utiliser leur ordinateur dans un but scientifique.

Il permet l'initiation aux méthodes du calcul numérique et illustre par des exemples concrets les théories mathématiques.

Les bases théoriques indispensables sont rappelées sans être développées, et de nombreux exemples commentés démontrent l'utilisation des programmes proposés.

Ceux-ci, lorsqu'ils appartiennent à un même chapitre, peuvent être utilisés simultanément, ce qui permet les opérations chaînées, par exemple en calcul matriciel ou complexe.

Enfin, les sujets traités sont en étroit rapport avec les programmes de mathématiques des classes préparatoires aux grandes écoles. L'informatique devant être prochainement introduite dans ces classes, l'étude et l'utilisation des programmes proposés permettront aux étudiants de se familiariser à la programmation et aux méthodes numériques.

EYROLLES

8458

EYROLLES

MATHÉMATIQUES SUR MICRO-ORDINATEUR - 2

A. REVERCHON
M. DUCAMP


Document numérisé
avec amour par :
AMSTRAD
CPC
MÉMOIRE ÉCRITE
ACME
https://acpc.me/